
第 1

はじめに

1.1 なぜコンピュータアーキテクチャを勉強するのか？

コンピュータアーキテクチャの勉強とは、簡単に言うと、コンピュータの中身を知
り、それがどのように動いてどのように作られているかを理解することだ。アーキテ
クチャというのは「建築」という意味で、コンピュータアーキテクチャはコンピュー
タという建築物を、どのように作っていくかについての方法論である。

Webを作ったり、グラフィックを作ったりするのは格好いいので、コンピュータの
プログラムはするかもしれないけれど、コンピュータを設計するわけではないので、
コンピュータの中身には興味がない。
ケータイは便利だし、ゲーム機は凄い。iPodは格好いいし、地デジはきれい、こう

いう製品を作る所に行きたいけど、コンピュータは関係ない。大体マイクロプロセッ
サは、もう日本では作ってないし、Intelが作るのを買えばいい。コンピュータの構成
なんて、もう進歩が止まっている分野で大きな発明や面白いことは何もない。要する
に、コンピュータアーキテクチャは興味も関心もない。多くの電気、電子情報系の大
学では、基礎科目としてコンピュータアーキテクチャ、コンピュータ工学などが設け
られている。進級のために履修はするけど面倒なだけ、と思っている方々もおられる
かと思う。しかし、これは誤った考え方である。
まず第一に、コンピュータのプログラマは、それが何を対象にするのであれ、一通

りのコンピュータアーキテクチャの知識を持っている必要がある。コンピュータアー
キテクチャはプログラマのために、抽象化を行って、細かい所は気にしなくてもプロ
グラム書けるようにしている。したがって、プログラマを目指す方々は、コンピュー
タハードウェアの詳細を知る必要はない。しかし、コンピュータアーキテクチャに関
して一応の知識がないと、良いプログラムは書けないし、もちろんプロにはなれない。
次に IT機器の開発者を目指す人たちは、より詳しくコンピュータアーキテクチャ

に関して知っておく必要がある。ケータイ、ゲーム、情報家電、自動車に至るまで、
あらゆる IT機器にコンピュータは組み込まれている。これらは組み込みコンピュー

タと呼ばれて、多くの場合、その IT機器を制御する中心としての役割を果たす。現
在、日本の半導体業界は、組み込みコンピュータをその部品の一部とし、他にメモリ、
専用プロセッサ、インタフェースなどを一つのチップに入れるシステム LSIあるいは
SoC(System-on-a Chip)を主力製品としている。ケータイに入っているチップはこの
システム LSIの代表であり、最近はどの IT機器もこれが頭脳としての役割を果たす。
したがって、IT機器に関連する場合、コンピュータアーキテクチャを理解していない
と、製品開発をすることができないし、もちろんマネージャになることも、下手をす
ると営業をするにも支障が出る。
このように、今やコンピュータは、デスクトップ PC,ラップトップ PCあるいはサー

バなどだけではなく、IT産業の基礎となるパーツであり、IT産業に携わる者の基礎
的な知識として必要である。これが多くの大学の電気、電子、情報系の基礎科目にコ
ンピュータアーキテクチャが入っている理由である。
また、コンピュータアーキテクチャは、進歩が止まってしまった分野でも、重要性が

小さくなっていく分野でもない。確かにコンピュータの中心部、CPU(Central Processing
Unit:中央処理装置)の構成自体は、ほぼ固まっているし、その動作周波数もこれ以上
高くできなくなっている。しかし、このことによって新しいアーキテクチャの研究開
発はむしろ活気付いている。多数の CPUを使ったマルチコア、メニーコア、限られ
た処理のみを高速化するアクセラレータ、汎用化されたグラフィックプロセッサ GP-
GPU(General Perpuse Graphic Processing Unit)、柔かいハードウェアと呼ばれるリコン
フィギャラブルシステム、動的に構成を変更する動的リコンフィギャラブルプロセッ
サなど新しいアーキテクチャが次々に登場しており、実際の製品にも使われている。
つい最近まで、コンピュータの性能は圧倒的に半導体のプロセスによって支配され

た。アーキテクチャが同じでも、新しい半導体プロセスを使えば、性能は上がり、コ
ストと消費電力は下がった。このため、アーキテクチャなんてどうでもいいから新し
いプロセスで新しい製品を作れば、うまく行ったのである。しかし、最近、半導体の
微細化加工技術が限界に近づくにつれ、新しいプロセスの開発は難しくなり、プロセ
スの代替わりの間隔は開きつつある。また、プロセスが新しくなっても、特に性能面
では目立って改善されないようになった。一つのプロセスが長い時間使われる場合に
重要になるのは、それをいかに活用するか、すなわちコンピュータを含めた広い意味
でのアーキテクチャである。今後 IT製品においては、半導体プロセスを新しくするよ
りも、より良いアーキテクチャを使う方が良い製品につながっていく。すなわちアー
キテクチャの重要性は今後どんどん大きくなっていくのである。
今まで、コンピュータアーキテクチャを学ぶことの実利についてを紹介したが、コ

ンピュータアーキテクチャを学ぶもっとも正しい動機は、それが面白いからだ。解説
が進むに従って、原理的には非常に簡単なものが、(もちろんソフトウェアの力が大
きいとはいえ)、凄いことをやってのけることがご理解いただけると思う。本書を完
全にマスターすれば、自分自身でコンピュータを設計し、FPGA(Field Programmable
Gate Array)や実際のチップ上で動かすことができるのだ。また、建築物同様、良くで

きたコンピュータは美しい。互換性を守りながら性能を上げるために、理解不能なレ
ベルで複雑化した Intel系の CPUにすら、長年、増築と改築を重ねた歴史的建造物の
ような美を感じることができるかもしれない。

1.2 コンピュータの基本

コンピュータは大きく分けて三つの領域で主に使われる。

• デスクトップ PC、ラップトップ PC:いわゆるコンピュータっぽいコンピュータ。
Windows, Linuxが Operating Systemとして走り、Intel系の命令が走る CPU(中
央処理装置)を用いている。無線や LANでネットワークに接続して用いる。最
も発展が急速で、性能と値段が重要。

• サーバ: 大企業や大学のシステムを統括する大型システム。ネットワークを介
してデスクトップ PCやラップトップ PCと接続し、ファイルを管理し、デスク
トップなどで手に余る大規模な計算を行う。あるいは、トランザクション処理
と言って、データベースに接続されて、データ管理を行う。銀行の ATM、航空
券の予約システムはこの例である。Webを管理して、検索等を行うWebサーバ
も大規模なサーバの一つ。故障しないで長期間動くこと、故障からの回復が早
いこと等信頼性 (dependability)が重要。

• 組み込みシステム: 様々な IT機器のなかに組み込まれており、外からは見えな
いし、普通はプログラムもしない。洗濯機やクーラーを制御するのは 8bitのコ
ンピュータで可能だが、PS3などのゲーム機には、Cellという超強力なプロセッ
サが組み込まれているし、ケータイの CPUも特定用途には強力である。コスト
と消費電力と、特定分野に限った性能が重要。

1.2.1 コンピュータの基本形

コンピュータはそれがデスクトップ PCであれ、組み込み用であれ、基本的に図 1.1
に示すように、CPU(Central Processing Unit:中央処理装置)と命令やデータを記憶して
おくメモリ、外部とのやり取りを行う入出力 (I/O)の三つの部分から出来ている。CPU
は、メモリから命令を基本的には順番に取ってきて、その中に書いてあることを実行
することを繰り返す。これをプログラム格納型 (Stored Programming)と呼ぶ。データ
は入出力装置から取り込み、メモリ上に置き、CPUはこれを読み書きしながら処理
を進め、結果をメモリから出力装置に出力して表示する。

図 1.1: コンピュータの三つの要素

1.2.2 コンピュータの歴史と発展

最初のコンピュータは何で、誰が発明者なのか？という問題は、なかなか難しく、
これを検討するだけで一冊の本ができてしまう1 のだが、最初の電子計算機として最
も良く知られているのは、ペンシルバニア大学の ENIAC(1946年)である。ENIACは、
全ての回路を真空管で実装した（18000本）点で電子計算機であったが、プログラミン
グにはスイッチの設定やケーブルの結線を人手で変更する必要があった。この ENIAC
の経験を基に、命令をメモリ上に格納し、これを順に取ってきて実行するプログラム
格納方式の考え方が生まれ、ケンブリッジ大学の EDSAC(1949年)が誕生した。これ
がはじめてのプログラム格納型計算機であり、現在のコンピュータは基本的にこの方
式から変わっていない。
コンピュータの歴史についても様々な文献や著書があり、ここでは詳しく触れない

が、コンピュータについて歴史上最も重要な変革があったのは実はつい最近である。
すなわち、コンピュータの歴史は、その使われ方から以下のように分けられる。

1.2.3 計算機登場～1980年後半: 中央集権時代

メインフレームと呼ばれる大型計算機が計算センターにどんと置かれており、これ
を共同で利用した。初期は、一人の利用者がコンピュータを占有したが、その後オ
ペレーティングシステムが複数のジョブを管理することができるようになり、あらか
じめパンチカードから入力したプログラムをテープやディスクに入れておき、順に
実行するバッチ処理が可能になった。さらに端末をつないで対話的にジョブを実行す
る TSS(Time Sharing System)が登場した。この時代、計算機といえばメインフレーム
=IBMマシンであった。このメインフレームは、ハードウェアの素子技術によって以

1この問題に興味のある方は、星野力著「誰がどうやってコンピュータを創ったのか？」(共立出版) を
お勧めする。

下のように世代 (generation)分けされる2

• 第１世代: 真空管 (1946～1950年代中ごろ)

• 第２世代: トランジスタ (1950年代中ごろ～1960年代中ごろ)

• 第３世代: 集積回路 (1960年代中ごろ～)

IBMは、1964年に IBM360を発表し、これがメインフレームのアーキテクチャの方
向を決定づけた。キャッシュ、割り込み、仮想記憶、本格的オペレーティングシステ
ム（OS）、リアルタイムシステムやトランザクション処理等、現在の計算機の技術の
ほとんどはメインフレームによって発達を遂げた。一方で、用途に応じたコンピュー
タ自体の分化も進んだ。DEC社の PDP-11を代表するミニコンピュータは、メインフ
レームに比べて安価で気軽に使えるため、研究・教育施設に広く用いられた。また、
科学技術計算を専用に行うユーザのためには、強力な数値演算能力を持つ CRAY-Iに
代表されるスーパコンピュータが登場した。

1.2.4 1980年代後半～2003年：マイクロプロセッサ大成長時代

一人が一台の PC(Personal Computer) あるいは WS(Work Station) と呼ばれるコン
ピュータを持ち、これらがネットワークで接続されていて、互いに情報を交換しなが
ら処理を行なう、現在一般的になっている使い方が確立したのがこの時代である。
初めてのマイクロプロセッサと呼ばれる Intelの i4004が誕生したのは、1971年で

ある。マイクロプロセッサは、計算機の CPUを 1(数)チップに格納したもので、電
子機器制御、ゲーム、簡単なコンピュータに用いられ、8bit, 16bit, 32bit と bit 数を
増やしながら急速に発達を遂げた。さらに、1980年代前半に普及した RISC(Reduced
Instruction Set Computer)方式により、マイクロプロセッサの性能は年々飛躍的に増大
（年間 1.5倍）するようになった。
ここに以下の状況が加わった結果、1980年代後半に革命的な事態が起こった。

• ビットマップディスプレイによるウインドウ、マウス等WS/PCのユーザインタ
フェース技術が確立した。

• 高級言語（C言語）の普及により、アセンブラに依らずにシステムが全て記述
できるようになった。

• UNIX, MS-DOSなどの汎用 OSが普及し、OSの移植が容易になった。

2第３世代後に大規模集積回路 (LSI)で構成された第４世代を設定する考え方もあるが、LSIは集積回路
の単に大きいものに過ぎないので、どこで切るかが苦しくなっている。日本で 1982 年から 10 年間に渡っ
て行なわれた第 5 世代コンピュータプロジェクトが失敗したため、これ以降、第 5世代という言葉を使う
人は居なくなり、計算機を世代で分ける考え方は消滅した。

• ローカルコンピュータネットワークが、Ethernetによって標準化され、さらに、
インターネットの登場により、グローバルネットワークの基盤技術が整った。

これらの技術的基盤が整備されることによって、それまで広くもちいられてきたメイ
ンフレームに代わって、最新の半導体技術を用いた高性能マイクロプロセッサを用い
た、個人用の PCやWSが急速に普及した。これらの PC／WSは、Windows, Linux
の汎用 OSを搭載し、ネットワークに接続して用いる。この結果、コンピュータの主
流はメインフレームから PCやWSに移り、また PCとWSの区別は消滅した。さら
には、ラップトップ型の可搬型のコンピュータ、携帯端末が普及した。

1.2.5 2004年～：マルチコア時代

2003年まで、マイクロプロセッサの動作周波数は年々向上し、性能も年間 1.5倍の
伸び率を示した。しかし、以下の原因で、マイクロプロセッサ CPU単体の性能は壁
にぶつかった。

• 高い周波数を用いたことにより発熱が増大し、これ以上周波数を上げると放熱
することが困難になった。

• CPUの動作速度にメモリが付いていけなくなり、これ以上動作周波数を上げて
も性能があまり上がらなくなった。

• 単体 CPUの性能向上を支えてきた、命令レベルの並列処理が限界に達した。

このため、CPU単体の性能を上げるよりも複数の CPUを１チップに搭載した構成が
一般的に使われるようになった。2010年現在、高性能のデスクトップでは 1チップ
に 4つの CPUを搭載する構成が一般化している。また、100を越える数の CPUを搭
載するメニーコアも作られ始めている。

1.3 本書の目的

従来、コンピュータアーキテクチャのテキストは、コンピュータの専門家を対象と
した分 Patterson&Hennesyのテキスト [1]および Hennessy&Patterson [2]の厚い決定
版テキストを除いては、コンピュータのプログラマやユーザーが、自分達の利用して
いるコンピュータの構造を「知識」として理解するために書かれたものが多かった。
このようなテキストは、おおむねトップダウンに書かれており、コンピュータアーキ
テクチャの様々な技術を概観し、知識として理解するためには優れたものが数多く書
かれている。
しかし、システム LSI時代にコンピュータのアーキテクチャを学ぶ場合、その構造

と動作を概観するのではなく、具体的に構造を理解し、できれば簡単なプロセッサを、

自分で設計した経験を持つことが望ましい。本書は、このような時代の要求に基づ
き、コンピュータアーキテクチャの初学者を対象として、最終的にはパイプライン化
された RISCを目標として、ボトムアップに設計しながら、設計演習とプログラミン
グ演習を通じてコンピュータアーキテクチャの実際的な設計技術を修得するのが目的
である。設計には、Verilog-HDLを用い、シミュレーションにはフリーソフトで誰で
も使える Ikarus Verilogと、gtkwaveを利用している。PCをお持ちの方ならば自宅で
誰でも実際に設計演習が可能である。また、web上の教材を用いれば、実際の FPGA
のキット上で実習が可能である。3

本書の初版が出版されたのは 2001年である。今回改訂したのは、以下の点である。

• 初版本では、PARTHENON/SFLを用いた。これは、当時フリーソフトで利用で
きた唯一のハードウェア記述言語とそれに基づく設計環境であったこと、美しい
言語構造で教育用途に優れていた点である。今でも SFLがVerilog-HDLよりも優
れたハードウェア言語であるという考えには変わりがない。しかし、SFLを実際
の設計現場で利用する機会がほとんどなくなったこと、Verilog-HDLにもフリー
ソフトで優れたシミュレーション環境が普及したことから改訂に踏み切った。本
書の Verilog記述に対応する SFL記述は web上で公開するので、PARTHENON
の利用者はご面倒だがそちらを参照していただきたい。

• 初版本では、教育用マイクロプロセッサ PICO を用いた。PICO は、本格的な
RISCの骨格を持ちつつ、分かりやすい、ということを目的に設計したプロセッ
サであった。ところが、「これでもまだ難しい」という批判があり、より分かり
やすい POCOに移行した。PICOについての解説、SFL記述も web上で公開す
る予定である。

• 本書はコンピュータアーキテクチャの入門書ではあるが、それでもこの 10年間
にコンピュータは進歩を続けた。この内容を取り込む必要があった。

本書は、プログラミング言語 (特に C言語)に関する初歩的な知識とブール代数や
順序回路設計等、論理設計に関する初歩的な知識を前提としているが、これらは電気
情報系の大学あるいは高専の初年度に学習するごく簡単なレベルである。すなわち、
本書は大学以上、高専上級生レベルの、コンピュータアーキテクチャの初学者を対象
としている。しかし、これらの知識が全くない初学者も各章の例題や演習を実際に実
行しながら自習すれば、充分内容が理解できると思う。

3私自身翻訳に携わった [3] は本書と全く同じ目的意識に基づいた良書である。しかし、本書の初版本
が出版されたのは 2001 年で、[3] よりもずっと以前であり、本書がこれに影響されて書かれたわけではな
い。また、コンピュータアーキテクチャのテキストについてのサーベイは web を参照されたい。

データパス：コンピュータで演算をす
る所

2.1 数の表現

まずはコンピュータ内の数の表現を復習しておこう。ご存じのとおり、ディジタル
システムでは数を 1と 0の 2進数で表現する。これは 1と 0がHレベルと Lレベルの
二つのレベルで処理を行うディジタルシステムと本質的に良く適合するためである。
基本となるのは符号なし数、すなわちただの 2進数である。一番右の桁が 1の位

であり、これを LSB(Least Significant Bit) という。一番左が 2n の位であり、これを
MSB(Most Significant Bit)という。
コンピュータでは長いビット数を扱うことが多いため、通常の 2進表現では見にく

いことがある。このような場合は、4桁ずつ区切って、16進数として表現する。この
時、10進数との混乱を避けるため、C言語の書き方にならって頭に 0xを付けて表す
場合が多い。ちなみに、かつては 8進数や BCD(Binary Coded Decimal: 2進化 10進
数)も用いられていたが、現在はほとんど使われないため省略する。� �

10進 2進 　 16進 10進 2進 16進
0 0000 0 8 1000 8
1 0001 1 9 1000 9
2 0010 2 10 1000 A
3 0011 3 11 1000 B
4 0100 4 12 1000 C
5 0101 5 13 1000 D
6 0110 6 14 1000 E
7 0111 7 15 1000 F

� �
上記を符号無し数 (unsigned)と呼ぶ。ではマイナスの数を表現するためにはどのよ

うにすれば良いだろうか？単純なアイディアは符号ビット sと符号無し数 (絶対値)を

組み合わせることで、s=1で 1001ならば-9、s=0で 1001ならば 9とする方法である。
これは可能だが一般的ではない。
マイナスの数を表現するのに普通に使われる方法は、2の補数表現である。この方

法には以下のメリットがある。(1)一番上の桁で符号がすぐに判定できる。(2)マイナ
スの数を加算することで減算ができる。(3)+0と-0が生じない。(符号ビットと符号無
し数の組み合わせだと+0, -0ができてしまう)。このため、現在すべてのコンピュータ
で用いられている。

2の補数とは、足した結果、すべての桁が 0になり、桁上げが生じる数のことであ
る。具体的には、(1)ある数の 1と 0をひっくりかえす。(2) 1を足す。で、求めるこ
とができる。例えば、4桁の数があるとするならば、下に示すように、-8から 7まで
16個の数を表すことができる。� �

2進 　 2の補数 符号なし数 2進 　 2の補数 符号なし数
0000 0 0 1000 -8 8
0001 1 1 1001 -7 9
0010 2 2 1010 -6 10
0011 3 3 1011 -5 11
0100 4 4 1100 -4 12
0101 5 5 1101 -3 13
0110 6 6 1110 -2 14
0111 7 7 1111 -1 15

� �
上の表も見てもわかるように、2の補数を用いると一番上の桁が 1ならば、それは

マイナスの数になる。つまり、2の補数表現では、MSBを符号ビットと呼ぶ。
2の補数表現で m桁の数の桁数を増やす必要が生じた場合は、MSBが 0ならば 0

を、1ならば 1を補う。例えば、4桁の’6’=0110を 8桁で表す場合、上に 0 を 4 つ
補い 00000110とする。しかし、’-6’=1010 を 8 桁で表す場合に同じことをすると、
00001010=10になってしまう。1を 4つ補い 11111010とすれば、8桁にしてもちゃん
と-6になる。このように桁数を増やす方法を符号拡張 (sign extension)と呼ぶ。符号
拡張は後に命令コード中に頻繁に表れて、皆さんを悩ますことになる。

例題 2.1

下の bit列を 16bitに符号拡張し、2進数で示せ。

1. 0011

2. 10010010

3. 0x93

答

1. 0011: 頭に 0を付けて 0000000000000011

2. 10010010: 符号拡張して 1111111110010010

3. 0xa3: 10100011なので符号拡張して 1111111110100011

例題 2.2

8bitの符号付き数で、(1) -12, (2) -122を示せ。

答

1. 12は 8ビットの 2進数で 00001100となる。1の補数は 11110011であり、これ
に 1を足して 11110100となる。

2. 122は 8ビットの 2進数を求めることがまず面倒であるが、2で割った余りを順
に下の桁から並べて行けば良い。122/2=61..0, 61/2=30...1, 30/2=15...0, 15/2=7...1,
7/2=3...1, 3/2=1...1となるので、01111010である。後は同様に1000101+1=10000110
となる。

1より小さい桁が必要な場合、固定小数点 (Fixed Point)数が用いられる場合がある。
この方法では小数点の場所をあらかじめ決めておくだけで、数字自体の記述は、通常
の数の表現と同じである。
より桁数の大きな科学技術計算には浮動小数点 (Floating Point)数が用いられる。現

在、IEEEにより標準化されており、32ビットの単精度、64ビットの倍精度が存在す
る。図 2.1に示すように、それぞれ仮数部と指数部を持っており、(仮数部)　×　 (2
の指数部乗)で数を表す。これ自体は簡単だが、浮動小数演算でいつも問題となるの
は、仮数部に入らなくなった桁を省略して、指数部に繰り込む際に生じる「丸め」の
処理である。この処理は繁雑で、ハードウェアを実装する際に常に悩ましい問題と
なる。

expon mantissa
0222330

s

s expon. mantissa

051526263

31

図 2.1: 浮動小数

さらに、コンピュータで扱う 2進ビット列は、数を表すだけとは限らない。文字コー
ドなど、数以外のデータを表す場合も多い。文字コードとしては英数字を表すコード

として 8bitの ASCIIコードが、より多くの文字を表すのには 16bitの Unicodeが一般
的だが、他にも多種多様なコードが使われる。
さて、ここで覚えておいて欲しい重要な概念がある。コンピュータで扱うデータは、

それ自体は意味を持っていない。1/0の列によって表されたデータは、命令なのか、符
号無し数なのか、符号付き数なのか、区別を持っていない。それがどんな意味を持つ
かは、扱う側、つまりコンピュータのハードウェアやプログラマの解釈次第なのであ
る1。

演習 2-1

下の bit列を 16bitに符号拡張し、2進数で示せ。またこれを符号無し数と考えて 16
進数で表記するとどのようになるかを示せ。

1. 11000110

2. 0x3D

3. 0x80

演習 2.2

8bitの符号付き数で、(1) -17, (2) -115を示せ。

2.2 ALU(Arithmetic Logical Unit)
さて、コンピュータの中で行う演算とはどのようなものだろう。まず加減乗除の四則演

算が思い浮かぶだろう。その他代表的な演算は以下の通りである。例ではA=10110110
と B=11100010を入力とする。

• 論理演算: 各桁の論理演算を行う。

– 論理積 (AND): A,B各桁の論理積を出力する。例ではA AND B = 10100010
となる。

– 論理和 (OR): A,B各桁の論理和を出力する。例では A OR B = 11110110と
なる。

– 排他的論理和 (XOR): A,B各桁の排他的論理和を出力する。排他的論理和
とは 0 XOR 0 = 0, 0 XOR 1 = 1, 1 XOR 0=1, 1 XOR 1 =0、すなわちビット
が異なっていれば 1となる。例では A XOR B = 01010100となる。

1これが原因で、処理の信頼性が低くなると考えた一部の研究者は、データに意味を付与するタグ付き
アーキテクチャというのを考えたが、苦労した割にメリットが少ないため、今の所、成功していない。



2.2. ALU(Arithmetic Logical Unit) 13

– 反転 (NOT): A各桁の反転を出力する。例では 01001001となる。

• nビットシフト: 各桁を n桁分ずらして (シフトして)出力する。

– 左シフト (Shift Left:SL):左方向にずらす。n桁シフトすると 2n倍したこと
になる。空いた最下位 (最も右)の桁には 0を詰める。1bitシフトの場合、
01101100となる。

– 右シフト (Shift Right:SR):右方向にずらす。n桁シフトすると 1/2n 倍した
ことになる。ここで右シフトには 2種類ある。

∗ 論理シフト (SRL: Shift Right Logical)：空いた最上位 (最も左)の桁に
は 0を詰める。例では 1bitシフトの場合、01011011となる。

∗ 算術シフト (SRA:Shift Right Arithmetic)：空いた最上位 (最も左)の桁
に符号ビットを詰める。つまり符号を保持したままシフトする。例で
は 1bitシフトの場合、11011011となる。

上記の例を図 2.2に示す2。

10110110

101101100

Shift Left

10110110

010110110

Shift Right
Logical

10110110

110110110

Shift Right
Arithmetic

図 2.2: シフト操作

演習 2.3

8bitデータについて A=10100110, B=01100100とする。以下の演算の結果を求めよ。

1. A AND B

2. A OR B

3. A XOR B

4. SL A (1bit)

5. SRA A (1bit)

6. SRA B (1bit)

2SLA Shift Left Arithmetic も定義可能だが、あまり一般的ではない

コンピュータでは、上記の演算を行う装置を一まとめにして扱う。これを、ALU(Arithmetic
Logic Unit)と呼ぶ。乗算と除算は、やや複雑で 1クロックで終わらせるとが難しいた
め、基本的な ALUの中には入れない場合が多い。

ALUは図 2.3に示すように a, b二つのデータ入力とコマンド入力 s、出力として y
を持っている。コマンド入力 sの内容に応じて、a,bの入力に対して所定の演算を行っ
て、yに出力するのが ALUの役目である。

a b

com
 y

16bit

16bit

3bit

16bit

図 2.3: ALUのモデル

ある ALUで 8つの演算が可能な場合、3ビットの comの値によってその内容を切
り替える。例えばここでは以下のように決めよう。� �

s 出力 記号
000 a　 THA
001 b THB
010 a AND b AND
011 a OR b OR
100 aを 1ビット左シフト SL
101 aを 1ビット論理右シフト SR
110 a+b ADD
111 a-b SUB

� �
yに a,bをそのまま出すのは奇妙な感じがするが、実はこれがないと周辺回路が面倒

になるので、たいていのALUはこの機能を持っている。この命令をここでは THA(ス
ルー A)、THB(スルー B)と呼ぶ。
さて、もちろん ALUの中には、加算器や ANDのアレイが入っているのだが、最

初は中身については考えず、とにかく comの値によって決った演算をやってくれるこ
とにする。このような考え方をブラックボックス化と呼ぶ。中身を知らなければ納得
できない、という人は第 4章に解説があるのでそちらをご覧いただきたい。

2.3 計算をするということ

この ALUで X+Yを計算する場合、両方の入力に Xと Yを入れ、sに 110をセッ
トすれば良い。では、もう少し複雑な

X+Y-W+Z

などの計算を ALUでどのように行えば良いのだろう。図 2.4のように複数の ALUを
組み合わせればこの演算を実現することができる。しかし、このようなことをしてい
くと、式が長くなると、どんどん利用しなければならない ALU数が増えてしまう。

A B

com
Out

A B

com
Out

A B

com
Out

X Y W Z

111

110 111

X+Y-W+Z

W-ZX+Y

図 2.4: ALUの接続

そこで、数を記憶しておく場所を設ける。これをレジスタ (register)あるいは置数
器と呼ぶ。抵抗器の resiterではないので注意。店のレジと語源は同じである。これも
中身が気になる人は第 4章を参照されたい。ここでは、入力したクロックが Lから H
に変化した際に、入力データを記憶する装置だと考えてほしい。
図 2.5に示すように、ALUの片方の入力にレジスタの出力を入れ、レジスタの入

力に ALUの出力を接続する。すなわち、計算結果をレジスタに格納し、それを再び
ALUの入力として計算に使えるようにするわけである。

X+Y-W+Zをこのデータパスで実行するには以下のようにすれば良い。ここでは、
一行分計算を実行した所でクロックを変化させてレジスタに値をセットすると考える。

s b クロック変化後のレジスタ
001 X X
110 Y X+Y
111 W X+Y-W
110 Z X+Y-W+Z

a b

s
y

accumulator
clock

com

図 2.5: 最も簡単なデータパスの構成

このようにすると、ひとつの ALUを使い回すことが可能になる。計算結果はレジ
スタに置かれて、更新される。このレジスタは結果が積み重なる場所となり、アキュ
ムレータ (accumlator)と呼ばれる。電卓で計算をする時、我々は無意識にこのアキュ
ムレータの考え方を使っている。また、この一行め (com=001)は、最初に計算すべき
値をアキュムレータにセットする役割を果たす。このように外部からレジスタに値を
セットする操作をロード (load)操作と呼ぶ。また、コンピュータで演算を行い、その
途中結果を格納しておく部分をデータパス (datapath)と呼ぶ。

演習 2.4

図 2.5のデータパスで X-Y+Zを計算したい。comと bにどのような値を与えれば良
いか。

2.4 メモリを使った計算

これまでのデータパスでは、アキュムレータを使うことで、演算した結果に対して
次々に演算を施して行くことができた。しかし、以下のような演算はスムーズに行う
ことができない。

(SL X)+(SL Y)

上の式ではまず Xをアキュムレータに入れて 1ビット左シフトし、その結果をどこ
かにとっておいて、それから Yを左シフトして加えなければならない。つまり、結果
をとっておく場所が必要であり、これを可能にするのがメモリである。電卓でもメモ
リ付きになっているのが普通で、このような枠組みは計算には必須である。
ここでは、メモリは単純に図 2.6に示す表と考えて欲しい。この表は、各行に番号

（アドレス）が振られており、アドレスの値によって、対応する行の中身（データ）を
読み出すことができる。この場合、アドレスに 0を設定すれば 1001が、1を設定す
れば 1100が読み出され、Dataoutに出力される。

また、書き込みの場合、入力にデータを置き、アドレスを設定し、書き込み制御信
号 we(write enable)を Hにし、クロックを立ち上げると、そのアドレスにデータが書
き込まれる。
この図は簡単で、幅が 4ビットで深さが 24=16のメモリということになり、全容

量は 64bitである。深さが 16=24 であるため、番地を識別するアドレス線も 4本で良
い。もちろん、実際のメモリはもっとずっと大きく、アドレスが 32bit-48bit、データ
は 32bit-64bitで膨大な情報量を格納する。しかし原理は同じである。アドレスが m
bitあれば、2m 行分指定することができ、容量の大きいメモリほど、アドレスの本数
も多い。ここでは、幅が 16bit、深さが 256のメモリを考える。この場合、アドレス
の本数は 8本となる。

....

0 1001

1 1100

...

0xF

Address

Data

Data

we

図 2.6: メモリの構成

それでは、このメモリを前回のデータパスにつないでみよう。この様子を図 2.7に
示す。
この構成では、データパスの入力にはメモリの出力、アキュムレータ出力にはメモ

リの入力が接続されている。すなわち、データを外から与えてやる代わりにメモリ中
にあらかじめ格納してデータに対して演算を行い、途中結果もメモリに格納すること
ができる。

2.4.1 命令の基礎

さて、このデータパスに外部から指示を命令の形で与えて計算をやらせよう。
まず演算を行うためには、メモリ中に存在するデータをアキュムレータに格納する

必要がある。コンピュータでは、このための命令を、

a b

s
 y

accumulatorclk

com

Address

dataoutdatain
we

Memory

alu_y

図 2.7: メモリ付きデータパス

（持って来い）　 0番地のデータを
の形で表す。日本語とは語順が逆なのは、命令を最初に作った人が英語を元に考え

たからである。（持って来い）の部分、つまり操作を示す部分を命令コード (opcode,
オペコード、オプコード)と呼ぶ。0番地のデータをに対応する操作対象を示す部分
をオペランド (operand)と呼ぶ。
メモリからレジスタにデータを持って来いという意味は先に述べたように、英語で

Loadである。（これは習慣で決っている。getでも takeでもない。moveという場合も
ある。）命令としては Loadと直接英語で書かずに LDという記号で書く。これをニー
モニックと呼ぶ。本来、ニーモニックは命令セット毎に違うのだが、長年の蓄積で、
ほぼ共通のコードを使っている。したがって LDと言ったら、情報系の人ならば万国
共通で、メモリをレジスタに持ってくる命令だと分かる。オペランドの部分はここで
は、単にメモリの番地に相当する。したがって、0番地のデータを持ってくる命令は

LD 0

と書く。これは、人間に分かりやすい形であって、アセンブリ言語表記あるいはア
レンブラ表記と呼ぶ。コンピュータのハードウェアは、1/0のパターンに対応付けた
ものを実行する。この場合、LDに 0001、アドレスはそのまま 0000を割り当てると
命令は以下のようになる。

0001 00000000 // LD 0 0番地の内容を読み出してアキュムレータに格納
0001 XXXXXXXX // LD X X番地の内容を読み出してアキュムレータに格納

これが機械語である。LD命令が実行されると、オペランドがメモリのアドレスに
与えられ、読み出されたデータは ALUの B入力から Y入力に抜ける。このために
ALUのコマンドを 001にセットする。そして、アキュムレータに書き込んでやる。

LDの逆の操作、すなわち、アキュムレータからメモリにデータを書き込む操作を
Storeと呼ぶ。これもニーモニックでは STと書いてやる。ここでは、STには 1000を
割り当てることにしよう。オペランドは以前と同様でメモリのアドレスである。

1000 00000000 // ST 0 0番地にアキュムレータの内容を書き込む
1000 XXXXXXXX // ST X X番地にアキュムレータの内容を書き込む

ST命令実行時には、メモリのアドレスにはオペランドを与え、アキュムレータか
ら直接引かれているデータ線を用いて、データをメモリに書き込む。

2.4.2 演算命令

典型的な命令は演算命令である。これは、アキュムレータの内容とメモリの指定さ
れたアドレスのデータを演算して、答えをアキュムレータに格納する。ここで使う
ALUは、6つの演算操作ができるので、以下の 6命令が定義できる。命令コードはど
う決めてもいいのだが、ここでは下 3ビットを ALUの comと一致させ、最上位ビッ
トを 0として決めよう。オペランドは以前通りメモリの番地である。ここで、SL(左
シフト、shift left)と SR(右シフト、shift right)は、B入力は使われないため、アドレ
スには何を入れても良い。(ドントケア)

0010 XXXXXXXX // AND X (X番地とアキュムレータの内容を AND)
0011 XXXXXXXX // OR X (X番地とアキュムレータの内容を OR)
0100 -------- // SL (アキュムレータの内容を左 1ビットシフト)
0101 -------- // SR (アキュムレータの内容を右 1ビット論理シフト)
0110 XXXXXXXX // ADD X (X番地とアキュムレータの内容を加算)
0111 XXXXXXXX // SUB X (アキュムレータの内容から X番地の内容を減算)

実は LDはこれらの演算命令の一種であったことに気づいただろうか？それでは、今
まで定義した命令を使って、演算処理を実行してみよう。

例題 2.3

X,Y,Zがそれぞれ 0,1,2番地に格納されているとする。

Z=X+Y

は以下のような命令を順に実行すれば計算できる。

LD 0 0001 00000000

ADD 1 0110 00000001

ST 2 1000 00000010

では、以下はどうであろうか？

例題 2.4

Z= X + Y - Z

これは、A+Bの結果がアキュムレータに残っているのでそれがそのまま使えるの
で簡単である。これがアキュムレータの良い点である。

LD 0 0001 00000000

ADD 1 0110 00000001

SUB 2 0111 00000010

ST 2 1000 00000010

しかし、最初に示した演算では、アキュムレータの値を直接利用することはでき
ない。

(X<<1)+(Y<<1)

この場合 Xの 1ビットシフトを計算し、その結果をどこかにとっておく必要があ
る。この中間結果をとっておく場所をここでは 2番地としよう。

LD 0 0001 00000000

SL 0100 00000000

ST 2 1000 00000010

LD 1 0001 00000001

SL 0100 00000000

ADD 2 0110 00000010

ST 2 1000 00000010

2番地に一度中間結果を入れてから Yを LDし、シフトをした後、加算をする。入
れる。これは、ちょうどメモリ付きの電卓を使う際に中間結果をメモリにとっておく
のと似ている。

演習 2.5

0番地,1番地,2番地にそれぞれ A,B,Cが格納されている。(A-B) OR (B+C)の演算を
行い、3番地に答えを格納するプログラムをアセンブラ表記、機械語の両方で示せ。

2.5 Verilogでの記述
2.5.1 加算器の記述

Verilogでは、一つのまとまりをもった回路をモジュールと呼ぶ。Verilogの記述は、
まずモジュールの名前と入出力を定義することから始まる。

module 名前 (

input 入力 1,
input 入力 2,
output 出力 1,
output 出力 2 );

この定義の方法は、C言語の関数定義とちょっと似ているが、最後にセミコロン (;)
が付いて文の形になっている点が違っている。まず a, bの２入力を持ち、sの出力を
持つ 1bitの加算器の記述を示す。これは以下のように非常に簡単である。

/* 1 bit adder */

module adder (

input a, b, output s );

assign s = a + b ; // add a b

endmodule

コメントは C 言語同様２種類 (/* */と//)が可能である。さて、加算器は、後の章
に解説するようにリプルキャリアダーからプリフィックスアダーまで様々な構成があ
るが、Verilogでも VHDLでもここは抽象化して単に’+’で表す。どのような加算器
を使うかは、後に論理合成の段階で、必要なコストと性能を考えて、自動合成用の
CAD(Computer Aided Design)プログラムが選択する。設計者はこの選択時に必要な
指示を与えてやる。s=a+b;の前に assignが付いているのに注意されたい。assign文
は、ある回路の出力ある端子に出力する、あるいは接続することを意味する。ここで
は加算の結果を sに出力する、あるいは加算結果出力を sと接続することを示してい
る。最後にモジュール文は endmoduleで終わるが、ここは;は付かない。
この記述は adder.vというファイルに入れておく。C言語のプログラムの拡張子が.c

であるのと同様、Verilog-HDL記述の拡張子は.vである。
次にこれをシミュレーションしてみよう。第 1章に紹介したように、Verilogは元々

シミュレーション記述用の言語であり、シミュレーションに対する指示の方法は多様
で、理解するのが大変である。ここはまずは動かしてみるのが良いと思う。

下の test.vは上記の加算器のモジュールをテストするシミュレーション用の Verilog
記述である。この記述は、ハードウェアの回路を表すのではなく、テストするための
入力の与え方と、出力の表示の仕方を示す。このような記述をテストベンチと呼ぶ。

Verilogを習得する時にやっかいなのは、テストベンチ中には結構難しい文法が必要
とされるのに、これを書かないとテスト対象のモジュールがいかに簡単なものであっ
ても全くシミュレーションができない点である。これを避けるためにある種の CAD
では、パラメータを与えてテストベンチを自動生成する機能を持つ。
で、ここでは、テストベンチの解説は一応しておくが、重要な文法は後でまた解説

することにして、直感的に理解する程度で深く突っこまないで先に進みたい。実の所、
テストベンチの作り方は定型的なので、一定のパターンをマスターしてこれを場合に
応じて修正すれば、かなりの範囲で応用が効く。どうか細部にこだわらず、大体の意
味を掴んでほしい。

/* test bench */

‘timescale 1ns/1ps

module test;

parameter STEP = 10;

reg ina, inb;

wire outs;

adder adder_1(.a(ina), .b(inb), .s(outs));

initial begin

$dumpfile("adder.vcd");

$dumpvars(0,adder_1);

ina <= 1’b0;

inb <= 1’b0;

#STEP

$display("a:%b b:%b s:%b", ina, inb, outs);

ina <= 1’b0;

inb <= 1’b1;

#STEP

$display("a:%b b:%b s:%b", ina, inb, outs);

ina <= 1’b1;

inb <= 1’b0;

#STEP

$display("a:%b b:%b s:%b", ina, inb, outs);

ina <= 1’b1;

inb <= 1’b1;

#STEP

$display("a:%b b:%b s:%b", ina, inb, outs);

ina <= 1’b0;

inb <= 1’b0;

$finish;

end

endmodule

最初に出てくるタイムスケール文は、シミュレーションの基本時間、時刻刻みの最
小時間を定義する。

‘timescale 1ns/1ps

ここでは前者は 1ns(ナノ=10−9秒)、後者は 1ps(ピコ=10−12秒)とした。
次に、パラメータ文を使って今回シミュレーションを行う１ステップの時間を定義

する。

parameter STEP = 10;

このパラメータ文は実行時に外部から値を設定できるのがミソで、便利なので良く
用いられる。ここでは 10nsecを 1ステップとする。
次に加算器に外部から値を与えるための入力、値を取り出すための出力用の端子を

定義してやる。ここで、入力には、値を保持するためレジスタ (reg)として宣言し、出
力は単に値を取り出せば良いので端子名を宣言 (wire)してやる。

reg ina, inb;

wire outs;

次に、adderモジュールの実体を定義して入出力を接続する。先にモジュール文で
adder.v内に定義した加算器のモジュール名を最初に指定し、次に実体名 (ここでは
adder 1)を宣言して実体を生成する。一つのモジュールから多数の実体を別の名前で
生成することができる。括弧内で、入出力の接続を行う。.の後にモジュール内の入
出力名、()の中に接続する外部端子の名前を記述する。ここでは adder内で宣言され
た a入力に外部の inaを、b入力に外部の inbを、s出力に外部の outsを接続する。

adder adder_1(.a(ina), .b(inb), .s(outs));

これで宣言が終わりで、次は initial文で、シミュレーションの動作を記述する。initial
文は、開始後一回だけ beginから endまでの文が順に実行される。

$dumpfile("adder.vcd");

$dumpvars(0,adder_1);

最初にのついた二つの文が実行される。Verilog内では、シミュレーションを制御
する上で必要な特殊なタスク (操作)を示す。この dumpfileと dumpvarsでは波形ビュ
アーで見るためのファイル名 (adder.vcd)と、信号を記録する範囲が指定されている。
次にいよいよシミュレーションを開始する。

ina <= 1’b0;

inb <= 1’b0;

#STEP

$display("a:%b b:%b s:%b", ina, inb, outs);

ina <= 1’b0;

inb <= 1’b1;

#STEP

$display("a:%b b:%b s:%b", ina, inb, outs);

ina <= 1’b1;

inb <= 1’b0;

#STEP

$display("a:%b b:%b s:%b", ina, inb, outs);

ina <= 1’b1;

inb <= 1’b1;

#STEP

$display("a:%b b:%b s:%b", ina, inb, outs);

ina <= 1’b0;

inb <= 1’b0;

$finish;

ina, inbへの値の代入は、後に解説するノンブロック代入文 <=で行っている。こ
れで 0,1が順に設定される。#STEPは、10nsecの時間経過を示す。すなわち、以下の
文は、

ina <= 1’b0;

inb <= 1’b0;

#STEP

$display("a:%b b:%b s:%b", ina, inb, outs);

まず、ina, inbに 0を設定し、10nsec時間が経過したら、ina, inb, outsを表示する、
という意味である。さて、Verilogでは定数を以下のように表現する。

<ビット幅>’<基数><数値>

基数は

• b：2進数

• h：16進数

• o : 8進数

である。なにも書かないと 10進数になるので、数を直接書く際は、必ず桁数をはっ
きりさせて 2進数、あるいは 16進数で書くことをお勧めする。ここで 1’b0は 1bitの
0を、1’b1は 1bitの 1をそれぞれ示す。

$display("a:%b b:%b s:%b", ina, inb, outs);

display文はその名の通り、ina,inb,outsの値を表示するためのものである。display
文の記述は C言語の printfに似ているが、2進数を表示するための%bを持っている。
また、自動的に改行が入る。以下、入力を変えて、それぞれ 10nsec時間を経過させ
て結果を出力していく。最後は finish文でシミュレーションを終了する。

2.5.2 シミュレーションの実行

では、シミュレーションをしてみよう。第 1章で紹介した ikarus verilogをインス
トールしてある環境を想定する。

iverilog test.v adder.v

で、シミュレーションの実行形が生成される。シミュレーション自体は、

vvp a.out

で実行される。ここでは最初は a入力、次は b入力、最後に加算結果が出力される。
a.outの名前が気になる方は、

iverilog test.v adder.v -o adder

vvp adder

としても良い。

a:0 b:0 s:0

a:0 b:1 s:1

a:1 b:0 s:1

a:1 b:1 s:0

が出力される。次に波形を見てみよう。テストベンチで記述した波形データを入れ
ておくファイルを指定して立ち上げる。

図 2.8: gtkwaveによる波形表示

gtkwave adder.vcd

左側の SSTと書いてある所に testとテストベンチのモジュール名が表示されるので
そこをクリックすると adder 1という加算器の実体名が表示される。ここをクリック
すると信号名が Signalsウインドウに表示される。ここで、見たい信号をクリックし
てから Appendをクリックすると、Wavesウインドウに波形が表れる。gtkwaveを始
めて使う時に戸惑うのは、Wavesの設定が最小時刻刻みになっているため、最初のき
わめて一部しか表示されない点である。Zoomの所の虫眼鏡の-を連打して表示スケー
ルを調整すると、見たい範囲で波形が見られる。gtkwaveは優れた GUIを持っている
ため、勘で結構使える。ここではあまり解説しないが、色々機能を試して欲しい。ち
なみに終了の際は File→Quitで確認用の小ウインドウが出てくるので、ここで Yesを
押してやる。

演習 2.6

adderの記述を書き換えて、減算、AND、ORなどにして試してみよ (減算は実は加算
と同じになる)。

2.5.3 ALUの記述

次にやや難しい ALUを記述してみる。まず、1ビットはあんまりなのでビット数
を拡張することにしよう。ハードウェアでは配線を何本か束にして扱う場合が多い。
これをバス (Bus)と呼び、Verilogでは大括弧でくくって，MSB:LSBの形でその範囲
を表わす。本書では簡単のため LSBは常に 0としよう。
入出力 16ビット、コマンド 3ビットの ALUは次のように定義される。

module alu (

input [15:0] a, b,

input [2:0] com,

output [15:0] y );

次に中身を定義する。ALUの場合、入力が特定の条件ならば入力同士の演算の結
果を出力する。このような場合、Verilogでは以下のように記述する。

assign 出力 = 条件? 式 1　：　式 2;

条件が真であれば式 1の結果が出力され、そうでなければ式 2の結果が出力される。
条件を複数書くこともできる。これは case文と似ている。条件は前から順にチェック
されるので、優先順位は先に書いた方が高いことになる。下の例では、条件 1-3がど
れも真でなれば、コロンの後に書いた式 4の結果が出力される。

assign 出力 = 条件 1? 式 1 :
　　　　　　　条件 2? 式 2 :
　　　　　　　条件 3? 式 3 : 式 4;

今回の ALUは、以下のような機能を定義した。
com 出力 記号
000 a　 THA

　　　　　　　　　 001 b THB
010 a AND b AND
011 a OR b OR
100 aを 1ビット左シフト SL
101 aを 1ビット論理右シフト SR
110 a+b ADD
111 a-b SUB

これを Verilog記述すると、以下のようになる。

assign y = com==3’b000 ? a:

com==3’b001 ? b:

com==3’b010 ? a & b:

com==3’b011 ? a | b:

com==3’b100 ? a<<1:

com==3’b101 ? a>>1:

com==3’b110 ? a + b: a - b ;

Verilogは、表 2.1に示す演算を記述できる。ここでは AND, OR,左右のシフト、加
算、減算を使っている。AND, ORでは、16ビットの入力のそれぞれの桁同士で演算
が行われ、答えも 16ビットになる。
ここで、論理演算は、C言語で用いるのと同様に、条件の論理的な真偽に対する演

算を行い、結果は 1ビットの値となる。ここでは比較演算子==を使っている。比較
対象は 3ビットの 2進数なので、3’b000と書く。
演算子には優先順位があるので注意されたい。表 2.2で上位にあるもの程強い。

表 2.1: Verilogの基本的演算子
ビット演算 ˜ NOT

& AND
˜& NAND
| OR
˜| NOR
ˆ Ex-OR
˜ˆ Ex-NOR

シフト演算 << 左シフト
>> 右シフト

等号、関係演算 == 等しい
! = 等しくない
=== 等しい (x,zも比較)
! == 等しくない (x,zも比較)
< 小さい
<= 以下
> 大きい
>= 以上

算術演算 + 加算
− 減算
∗ 乗算
/ 除算
% 剰余算

論理演算 ! 否定
&& 論理積
|| 論理和

2.5.4 define文を使って格好を付ける

先に示した ALUの記述はきちんと動作し、合成もできるが、あまり格好良いと見
なされない。これは、コード中に 3’b000とか 15:0など定数が直に記述されているか
らである。これらのハードウェアの記述に必要な定数 (マジックナンバーと呼ばれる
こともある)は、直接コード内に書かない方が良い。これは、コードが読みにくくな
ることと、仕様変更があった際にコード中のあちこちを変更しなければならないため
である。そこで、Verilogでは C言語と同様 define文でこれらの定数をあらかじめ定
義しておく。ただし C言語と違って define文の前と定義された文字列を使う場合は
バックシングルコーテーション (‘)を使う。下記は４種類の演算のみこの変換を行っ
た例である。

表 2.2: 基本的演算子の優先順位
高い ˆ ˜ !&| + - (単項演算子)

∗/%
+ -

<<>><<<>>>

<<=>>=

==! ====! ==
& ˜& ˜ ˜ˆ
| ˜|

&&
||

低い

‘define DATA_W 16 // bit width

‘define SEL_W 3 //control width

‘define ALU_THA ‘SEL_W’b000

‘define ALU_THB ‘SEL_W’b001

‘define ALU_AND ‘SEL_W’b010

module alu (

input [‘DATA_W-1:0] a, b,

input [‘SEL_W-1:0] s,

output [‘DATA_W-1:0] y );

assign y = s==‘ALU_THA ? a:

s==‘ALU_THB ? b:

s==‘ALU_AND ? a & b: a + b ;

endmodule

演習 2.7

上記の記述を全ての演算について行え。
define文は parameter文よりも変更が少なく、基本的に決めたら滅多に変えないも

のを定義するのに使う。ただ両者の特徴は微妙で、これについては、webのうんちく
を参照されたい。

2.5.5 ALUのテストベンチ

下の test.vは ALUをテストするシミュレーション用のテストベンチである。基本
的なやり方は、加算器用と同じである。

/* test bench */

‘timescale 1ns/1ps

‘define DATA_W 16 // bit width

‘define SEL_W 3 //control width

‘define ALU_THA ‘SEL_W’b000

‘define ALU_THB ‘SEL_W’b001

‘define ALU_AND ‘SEL_W’b010

‘define ALU_ADD ‘SEL_W’b110

module test;

parameter STEP = 10;

reg [‘DATA_W-1:0] ina, inb;

reg [‘SEL_W-1:0] sel;

wire [‘DATA_W-1:0] outs;

alu alu_1(.a(ina), .b(inb), .s(sel), .y(outs));

initial begin

$dumpfile("alu.vcd");

$dumpvars(0,alu_1);

ina <= ‘DATA_W’h1111;

inb <= ‘DATA_W’h2222;

sel <= ‘ALU_THA;

#STEP

$display("a:%h b:%h s:%h y:%h", ina, inb, sel, outs);

sel <= ‘ALU_THB;

#STEP

$display("a:%h b:%h s:%h y:%h", ina, inb, sel, outs);

sel <= ‘ALU_AND;

#STEP

$display("a:%h b:%h s:%h y:%h", ina, inb, sel, outs);

sel <= ‘ALU_ADD;

#STEP

$display("a:%h b:%h s:%h y:%h", ina, inb, sel, outs);

$finish;

end

endmodule

今回は、桁数が 16bit になっており、a,b入力には１６進数を、comには２進数を
使って定義をしている点に注意されたい。16進表示はC言語の 0xではなく hを使う。

a <= 16’h1111;

b <= 16’h2222;

com <= 3’b000

同様に、display文も 16進表示は%xでなくて、%hである。では、加算器と同様に
シミュレーションをしてみよう。

iverilog test.v alu.v

で、シミュレーションの実行形が生成される。シミュレーション自体は、

vvp a.out

で実行される。ここでは最初は a入力、次は b入力、最後に加算結果が出力される。

a:1111 b:2222 com:0 y:1111

a:1111 b:2222 com:1 y:2222

a:1111 b:2222 com:6 y:3333

演習 2.7

AND,OR,減算のシミュレーションを行い、結果を確認せよ。

2.5.6 データパスの記述

図 2.5のデータパスを記述してみよう。このためには、レジスタを記述する必要が
ある。レジスタはクロックに同期してデータを格納する点でやや特殊な素子であり、
ALUでの論理演算とは雰囲気が違っている。

Verilogでレジスタを宣言する際はそのものずばり

reg [15:0] acum;

とすれば良い。ここでのアキュムレータは 16ビットのデータを記憶するので、入出
力同様バスの形にして宣言する。図 2.5を見ると、信号線を区別するため alu yなど
の名前が付いている。Verilogでは、wire文で信号の名前を宣言する。

wire [15:0] alu_y;

レジスタと違って、alu yは ALUの出力データそのものを示していて、データを記憶
することはできない。ALU同様、直接数字を書かないことにして、ここでは、

reg [‘DATA_W-1:0] accum;

wire [‘DATA_W-1:0] alu_y;

と記述する。ここで ALUの時と違って、この define文を 1ヶ所にまとめて def.hと
しておく。そして

‘include ‘‘def.h’’

とすれば、様々なファイルで同じ定義を共有することができる。この辺の発想は、
C言語と同じである。ただし、include文の先頭にはシングルバックコーテーション
(‘)が付く点に注意されたい。def.hの中身は以下の通りである。

‘define DATA_W 16

‘define SEL_W 3

‘define ADDR_W 8

‘define DEPTH 256

‘define ALU_THA ‘SEL_W’b000

‘define ALU_THB ‘SEL_W’b001

‘define ALU_AND ‘SEL_W’b010

‘define ALU_ADD ‘SEL_W’b110

‘define ALU_SUB ‘SEL_W’b111

‘define ENABLE 1’b1

‘define DISABLE 1’b0

‘define ENABLE_N 1’b0

‘define DISABLE_N 1’b1

では、全体の記述を見てみよう。

‘include "def.h"

module datapath(

input clk, input rst_n,

input [‘DATA_W-1:0] datain,

input [‘SEL_W-1:0] com,

output [‘DATA_W-1:0] accout);

reg [‘DATA_W-1:0] accum;

wire [‘DATA_W-1:0] alu_y;

assign accout = accum;

alu alu_1(.a(accum), .b(datain), .s(com), .y(alu_y));

always @(posedge clk or negedge rst_n)

begin

if(!rst_n) accum <= ‘DATA_W’b0;

else accum <= alu_y;

end

endmodule

このデータパスは、アキュムレータの中身をメモリに書き込むために外部に出さな
ければならない。このような場合、assign文を用いる

assign accout = accum;

この文は、accumのデータを dataoutに出力すると考えてもよいし、accumを出力
dataoutに接続すると考えても良い。さて、ALUは別モジュールとして宣言されてい
るので、これを部品として使う。これが以下の記述である。

alu alu_1(.a(accum), .b(datain), .com(com), .y(alu_y));

aluがモジュール名であり、alu 1が実体 (インスタンス)名である。同じモジュール
から実体を多数作ることも良く行われる。先に述べたように、Verilogでは、モジュー
ル内の入出力名を.(ピリオド)で表し、それに接続する外部信号名を括弧の中に書いて
対応を付ける。つまり、ALUの a入力には accumを、b入力には datainを、com入
力には comを接続し、yに alu yを接続して出力を取り出す。ここで、comは内部も
外部も同じ名前を使っているが、module aluと module datapathの両方で宣言されて
いれば問題ない。

always @(posedge clk)

accum <= alu_y;

最後の部分が、アキュムレータ accumに ALU出力 alu yを覚えさせる記述である。
この書き方がとっつきにくいのが、Verilogが入門者に嫌われる理由の一つになって
いるが、決った使われ方しかしないので、「おまじない」と思えば良い。
コンピュータでは、レジスタやメモリにデータを覚えさせるのは、一定の信号の

変化に合わせて行なう。この信号をシステムクロックあるいは単にクロックと呼ぶ。
3GHzで動作する CPUとか呼ぶが、これはクロックの周波数を示している。これが高
い程、CPUは高速で動くので宣伝文句として使われる。3

さて、
3しかしクロック周波数は性能と直接結びつかないので注意。これは後に取り上げる

always @(posedge clk)

は、always=いつも、@=atつまり..の時、(posedge clk)=クロックが LレベルからH
レベルに変化する (立ち上がり：positive edge)という意味である。つまり、「clkの立
ち上がりの時にはいつも」ということを示す。ちなみに立ち下がりで動作する際は、
negedge：negative edgeを使う。

accum <= alu_y;

<=は、ノンブロッキング代入文と呼び、レジスタにデータを覚えさせる時に使う。先
のテストベンチでも使っていた。ここには同じクロックの立ち上がりで動作する複数
の文を begin endで使って書くことができる。これは後に紹介する。

2.5.7 データパスのシミュレーション

ここまでのデータパスにはメモリが含まれていない。後に解説するがメモリは普通
は論理合成の対象としないため、テストベンチの中に入れて記述しよう。メモリはレ
ジスタ同様 reg文で記述するが、名前の後に

最初の番地 :最後の番地

の形で、記憶できる範囲を示す。

reg [15:0] dmem [0:15];

この場合、dmemは 0番地から 15番地まで記憶できるアドレスを 16持つ。という
ことはアドレスは 4ビットで表すことができる。

reg [15:0] dmem [0:1023];

と書けば、1024=1Kの範囲で、アドレスは 10本になる。原則として範囲は 2のべ
き乗で示す。メモリからの読み出しは、4ビットの daddrを宣言しておき、それを直
接括弧の中い入れれば良い。

dmem [daddr];

書き込みはレジスタ同様 always文を用いるが、メモリは、データを書き込む場合
と、何も書き込まず単に読み出しだけを行う場合があるので、we(write enable)端子
でこれを指示する。we=1とした時のクロックの立ち上がり時にデータが書き込まれ
る。そこで、記述は下のようになる。

always @(posedge clk)

if(we) dmem[daddr] <= ddataout;

if文は C言語など多くのプログラミング言語と同様、括弧内の条件が真になった時
に次に書く動作が実行される。最初は奇妙に思われるかもしれないがVerilogは always
文など限られた構造の中でのみ、if文の利用が許される。
さて、このメモリの記述を含むデータパスのテストベンチは以下のようになる。

/* test bench */

‘timescale 1ns/1ps

‘include "def.h"

module test;

parameter STEP = 10;

reg clk, rst_n;

reg [‘DATA_W-1:0] datain;

reg [‘SEL_W-1:0] com;

reg [‘ADDR_W-1:0] addr;

reg we;

wire [‘DATA_W-1:0] accout;

wire [‘DATA_W-1:0] dmem2;

reg [‘DATA_W-1:0] dmem [‘DEPTH-1:0];

always @(posedge clk)

begin

if(we) dmem[addr] <= accout;

end

always #(STEP/2) begin

clk <= ˜clk;

end

datapath datapath_1(.clk(clk), .rst_n(rst_n), .com(com),

.datain(dmem[addr]), .accout(accout));

initial begin

$dumpfile("datapath.vcd");

$dumpvars(0,test);

$readmemh("dmem.dat", dmem);

clk <= ‘DISABLE;

rst_n <= ‘ENABLE_N;

{we,com,addr} <= {‘DISABLE,‘ALU_THB,‘ADDR_W’h00}; // LD 0

#(STEP*1/4)

#STEP

rst_n <= ‘DISABLE_N;

$display("we:%b com:%h addr:%h accout:%h", we, com, addr, accout);

$display("dmem[0]:%h dmem[1]:%h dmem[2]:%h", dmem[0], dmem[1], dmem[2]);

#(STEP*1/2)

{we,com,addr} <= {‘DISABLE,‘ALU_ADD,‘ADDR_W’h01}; // ADD 1

#(STEP*1/2)

$display("we:%b com:%h addr:%h accout:%h", we, com, addr, accout);

$display("dmem[0]:%h dmem[1]:%h dmem[2]:%h", dmem[0], dmem[1], dmem[2]);

#(STEP*1/2)

{we,com,addr} <= {‘DISABLE,‘ALU_ADD,‘ADDR_W’h02}; // ADD 2

#(STEP*1/2)

$display("we:%b com:%h addr:%h accout:%h", we, com, addr, accout);

$display("dmem[0]:%h dmem[1]:%h dmem[2]:%h", dmem[0], dmem[1], dmem[2]);

#(STEP*1/2)

{we,com,addr} <= {‘ENABLE,‘ALU_THA,‘ADDR_W’h02}; // ST 2

#(STEP*1/2)

$display("we:%b com:%h addr:%h accout:%h", we, com, addr, accout);

$display("dmem[0]:%h dmem[1]:%h dmem[2]:%h", dmem[0], dmem[1], dmem[2]);

#(STEP*1/2)

{we,com,addr} <= {‘DISABLE,‘ALU_THA,‘ADDR_W’h02}; // NOP

#(STEP*1/2)

$display("we:%b com:%h addr:%h accout:%h", we, com, addr, accout);

$display("dmem[0]:%h dmem[1]:%h dmem[2]:%h", dmem[0], dmem[1], dmem[2]);

$finish;

end

endmodule

まず、クロック信号 clkが一定の間隔で L,H,L,Hを繰り返すようにしている。これ
が下記の文である。

always #(STEP/2) begin

clk <= ˜clk;

end

この場合、STEPの半分で反転するということは、２回反転すればもとのレベルに
戻ることから、クロック周期は STEPとなることがわかる。この記述でクロックを発
振させるには、clkを regで定義して、初期化する必要がある。

initial文以降は以前の記述と似ているが、データメモリに初期値を設定するための
文が必要である。

$readmemh("dmem.dat", dmem);



2.5. Verilogでの記述 37

この場合、データメモリの実体 dmemに対して dmem.datという名前のファイル中の
16 進データ (readmemh だから) が初期設定される。あらかじめ dmem.dat に以下の
フォーマットでデータを入れておく。

0002

0003

0004

0001

ちなみに、2進数のデータを読み込むためには、readmembを用いる。dmem.datは
シミュレーションを起動するディレクトリに置いておく必要がある。次に、以下の記
述に注目されたい。

{we,com,addr} <= {‘DISABLE,‘ALU_THB,‘ADDR_W’h00};

Verilog では中括弧は、信号をくっつけて一つにして扱うことを示す。すなわち、
we,com.addrはこの順番にくっついて一連の信号線として扱われる。ここでは、we=0,
com=001, addr=0000が設定される。この場合、データパスに対する命令を操作の部
分とアドレスの部分に分離している。ここでは、一定の時間間隔で、LD 0, ADD 1,
ADD 2, ST 2の命令が順に与えられる。このため、2+3+4=9が dmem[2]に入る。
それぞれの display文で、クロックが立ち上がる度に、アキュムレータの値とメモ

リの 0-2番地を表示する。最初に STEP*1/4としているのは、値を設定する時刻と表
示する時刻をクロックの立ち上がりとずらしてやるためである。シミュレーション上
完全に同じ時刻に行うという指定をすると、どのような順番でシミュレーションが実
行されるかわからなくなるため、困ったことになる。これを防ぐためである。
先ほどと同じように、

iverilog test.v datapath.v alu.v

vvp a.out

と打ち込みシミュレーションを行い結果が正しいことを確認せよ。

演習 2.8

Aを 0番地、Bを 1番地のデータとして、(SR A) OR (SR B)のデータを 2番地にしま
う命令の実行をテストベンチを改造してシミュレーションせよ。

演習 2.9

Aを 0番地、Bを 1番地、Cを 2番地のデータとして、(A+B) OR (A-B)の結果を 3
番地にしまう命令の実行をテストベンチを改造してシミュレーションせよ。

演習 2.10

このデータパスは、ST命令の実行時にも accum中のデータが壊されずに動作する。
これはなぜだろう？

アキュムレータマシン

3.1 アキュムレータマシンの構成

3.1.1 命令メモリと pc

2章のデータパスでは、計算するデータはデータメモリにあらかじめ入れておけば
良いのに、命令は外からクロックに合わせて与えてやらなければならず、これは大変
である。そこで、データ同様、命令もあらかじめメモリに入れておき、これを順番に
読み出してデータパスに与えるようにすれば便利である。この構成を図 3.1に示す。
ここでは、8ビット× 256エントリの命令メモリ (Instruction Memory)を想定し、こ

の中に命令が並んで格納されていると想定しよう。例えば前回の C=A+Bを計算する
プログラムが以下のように入っている。

0番地　 LD 0　　 0001 00000000
1番地　 ADD 1　 0110 00000001

2番地　 ST 2 1000 00000010

ここで、命令を順番に実行するためには、現在実行しているプログラムの番地を保
持しているレジスタが必要である。このレジスタをプログラムカウンタ PC(Program
Counter)と呼ぶ。今、命令メモリは 256エントリを持つので、PCは 8ビットあれば
良い。計算機は、以下の動作を繰り返す。

1. PCの指し示す命令を取ってくる（命令フェッチ）同時に PCに１を加える

2. 解読し（命令デコード）、

3. 前回定義したデータパス上でこれを実行する（命令実行）。その後１に戻る

前回の Verilogの実装では、ST命令実行時には com=000(THA:スルー A)とするこ
とで、アキュムレータに自分自身の値が書き込まれて、データが変化しなかった。し

かし、この方法では accumに値を書き込まない時は常に com=000にしなければなら
ず、命令数を増やすことができない。そこで、特定の命令でのみ accumに値を書き込
み、それ以外では書き込まないようにしてやる。このために、accumに accsetという
セット入力を与えてやり、これが 1の時のみ、結果を書き込むようにする。

a b

com
y

accum
clk

com

daddr

ddatain
we

Data
Memory

Instruction
Memory

pc

opcode

’1’

clk

iaddr
idatain

am.v

alu

accset

opcode=ST

’+’

図 3.1: アキュムレータマシンの構成

この構成が、プログラム格納型計算機の第一歩で、アキュムレータマシンと呼ば
れる。

3.1.2 アキュムレータマシンの動作

図 3.1の動作を追ってみよう。まず、命令メモリ (Instruction Memory)のアドレスに
は pcが繋がっており、この内容の示すアドレスの命令が読み出さる。
読み出された命令が ST命令以外ならば、Data memoryから命令中の下位 8ビットを

アドレスとしてデータが読み出される。つまり LD 1ならば 1がアドレスに載る。そし
て読み出されたデータは ddatainから aluの b入力に入る。aluの s入力には、opcode
の下位 3ビットが入っている。LD命令の場合、opcodeは 0001なので、comは 001
となり、THBが実行されて、読み出されて B入力に入ったデータが、クロックの立
ち上がリで、そのまま accumにセットされる。同時に pcには pc+1がセットされ、次
の番地の命令が指し示される。
読み出された命令が ADD 2ならば、opcodeは 0110なので、comには 110が与え

られ、aluは a入力と b入力の加算を行う。a入力には accumの値が常に入力され、
b入力は 2番地から読み出したデータがやってくる。加算結果は EX状態の終わりに
accumに書き込まれる。

図 3.2: ST命令の実行例

メモリへの書き込みを行う ST命令でのみ、データパスの動作はやや異なっている。
ST 3を実行する場合を図 3.2に示す。ST命令のアドレスは 00000011であるので、最
上位が 1ならば ST命令であるという判断ができる。ここで、Data memoryの入力ポー
トには常に accumの出力が dataoutを経由して接続されている。アドレスは他の命令
同様、operandの値が与えられるため、この場合は 3である。ここで weを Hにする
ことで、data memoryに accumの値を書き込む。
今の所、以下の命令が利用可能である。Xは番地の 2進数を示し、-は使われない

ので何でも良いこと（ドントケアと呼ぶ）を示す。

ニーモニック　　機械語　　　　　意味
NOP 　　　　　 0000 --------: No Operation: 何もしない
LD X 　　　　　 0001 XXXXXXXX: Load X: acc-m<- X番地中の値
AND X 　 　　　 0010 XXXXXXXX: AND X: acc-m <- acc-m AND X番地中の値
OR X 　　　 0011 XXXXXXXX: OR X: acc-m <- acc-m OR X番地中の値
SL 　　　 0100 --------: Shift Left: acc-m <- acc-m<<1
SR 　　　 0101 --------: Shift Right: acc-m <- acc-m>>1
ADD X 　　　 0110 XXXXXXXX: Add X: acc-m <- acc-m + X番地中の値
SUB X 　　　 0111 XXXXXXXX: S-b X: acc-m <- acc-m - X番地中の値
ST X 　　　 1000 XXXXXXXX: Store X: accum -> X番地中

ここで、NOPはアキュムレータの値が自分自身に書き込まれるため、何もしない
命令である。何もしない命令なんて意味があるんだろうか？と思われるかもしれない
が、実はこの命令は時間稼ぎのために重要で、全てのコンピュータで装備されている。

演習 3.1

0番地の内容から 1番地の内容を引き算し、2番地に格納する操作を実行するプログ
ラムを書け。

演習 3.2

0番地の内容から 1番地の内容を引き算した結果に、2番地の内容と 3番地の内容を
加算した結果と ORするプログラムを書け。

3.1.3 分岐命令

さて、このアキュムレータマシンは、命令メモリに格納された命令を順番に実行す
るに過ぎない。前の章の演習プログラムとて、実行が終わったら、次の番地の命令の
実行に移ってしまう。完全な猪突猛進型で、演算処理の終了後に、停止することすら
できない。

これではあらかじめ必要な処理に対応する命令を全て並べておかなければならず、
ちっとも便利ではない。コンピュータが便利なのは、膨大な繰り返し処理をやってく
れることと、結果に応じて判断して、実行する処理を変えてくれることである。これ
により、プログラムを作って複雑なアルゴリズムを実行させることができる。つまり、
「判断」して「処理を繰り返す」という命令が必要だ。
一定の処理を繰り返し実行させるためには、アキュムレータの内容を判断し、PC

の内容を変更する命令を設ければよい。これを分岐命令（Branch）と呼ぶ。ここでは
以下の命令を想定する。

BEZ X　アキュムレータの内容が0ならば、PCをXXXXXXXXにする。 1001 XXXXXXXX

BNZ X　アキュムレータの内容が0でなければ、PCをXXXXXXXXにする。1010 XXXXXXXX

例えば BNZ 0を実行すると、アキュムレータの内容が 0でなければ、PCが 0にな
り、次は 0番地の命令を実行することになる。これをコンピュータ屋は「0番地に飛
ぶ」と言う。この場合アキュムレータの内容が 0でなければ、分岐命令は何も行わず、
次の番地の命令が実行される。
以下の命令が順に実行される場合を想定しよう。

0　 LD 0 0001 00000000

1　 ADD 1 0110 00000001

2　 ST 0 1000 00000000

3　 LD 2 0001 00000010

4　 SUB 3 0111 00000011

5　 ST 2 1000 00000010

6　 BNZ 0 1010 00000000

7　 BEZ 7 1001 00000111

ここでデータメモリには、

0 0000

1 0002

2 0003

3 0001

が格納されているとする。何が起こるか考えて見よう。最初に 3命令で、0番地の
内容← 0番地の内容＋ 1番地の内容が実現できる。0番地の内容は 0に初期化してお
くことに注意。次の 3命令で、2番地の内容← 2番地の内容－ 3番地の内容が実行さ
れる。ここで 3番地は 1に初期化されているため、この 3命令は 2番地の内容から 1
を引いて書き戻すことになる。ここで、アキュムレータの内容が 0ならば 0番地に戻
ることになる。つまり、このプログラムは 2番地から 1つづつ引きながら 0になるま

で 1番地の内容を 0に足し続ける。2番地が 0になると、プログラムは 7番地に飛び
続けることで停止する (このような止め方をダイナミックストップと呼ぶ)。つまりこ
れは掛け算のプログラムであることがわかる。ここでは 1番地に 2が、2番地に 3が
格納されているので、2× 3が実行され、0番地が 6になる。アセンブラのプログラ
ムは単純で原始的なので、図 3.3に示すような流れ図を作って考えるとよい。

(0)<-(0)+m

(0)<-0
(3)<-1
(2)<-n
(1)<-m

n<-n-1

n=0?
No

Yes

n=0?

Yes

図 3.3: 掛け算プログラムの流れ図

分岐命令を実行するためには、(1)命令メモリから読み出した命令が分岐命令であ
り、(2)条件が成立する場合に限り、PCに命令の下位 8ビットの飛び先をセットでき
るようにすれば良い。
図 3.4にこの改造したアキュムレータマシンのハードウェア構成を示す。台形印は、

マルチプレクサあるいはデータセレクタというハードウェアモジュールで、この場合、
2つの入力の片方を、選択信号 pcselに応じて選んで出力する。ここでは、pcsel=1な
らば、pcに operandを送り、S=0ならば、pcには以前通り pc+1を送る。マルチプレ
クサは、今後データパスを複雑にしていく場合不可欠なものである。また、accumに
は、accset信号を Lにして、分岐命令実行時にはデータをセットしないようにしてや
る。この辺は ST命令と同様である。太線に BEZ, BNZ実行時のデータの流れを示す。
まとめると、命令コードと accumの値に応じて pcsel, accset, we, comの値を以下のよ
うに制御する必要がある。

• ALU演算命令 (LD,NOPを含む): accset=1, pcsel=0,we=0, com:opecodeの下位 3

7:0

Cont.pcsel
10

=0?

図 3.4: アキュムレータマシンの構成 (分岐命令付き)

ビット

• ST命令:accset=0, pcsel=0,we=1,com:ドントケア

• 分岐命令：accset=0, pcsel:条件が成立すれば 1、そうでなければ 0、we=0、com:
ドントケア

この信号はコントローラ (Cont)で生成される。コントローラはここでは非常に簡単な
組み合わせ回路で済むが、命令が増えるにつれて複雑になっていく。

演習 3.3

今、1番地にAという数が格納されているとする。1＋ 2＋…Aを計算して 0番地（あ
らかじめ 0に初期化されている）に格納するプログラムを書き、機械語に変換せよ。
ここでは例題同様、3番地は 1という数字で初期化されているとする。

アキュムレータマシンの命令コード

本章のアキュムレータマシンの命令コードをまとめる。

NOP 　　　　　 0000 --------: No Operation: 何もしない
LD X 　　　　　 0001 XXXXXXXX: Load X: acc-m<- X番地中の値
AND X 　 　　　 0010 XXXXXXXX: AND X: acc-m <- acc-m AND X番地中の値
OR X 　　　 0011 XXXXXXXX: OR X: acc-m <- acc-m OR X番地中の値
SL 　　　 0100 --------: Shift Left: acc-m <- acc-m<<1
SR 　　　 0101 --------: Shift Right: acc-m <- acc-m>>1
ADD X 　　　 0110 XXXXXXXX: Add X: acc-m <- acc-m + X番地中の値
SUB X 　　　 0111 XXXXXXXX: S-b X: acc-m <- acc-m - X番地中の値
ST X 　　　 1000 XXXXXXXX: Store X: accum -> X番地中
BEZ X　アキュムレータの内容が0ならば、PCをXXXXXXXXにする。 1001 XXXXXXXX

BNZ X　アキュムレータの内容が0でなければ、PCをXXXXXXXXにする。1010 XXXXXXXX

3.2 アキュムレータマシンのVerilog記述
3.2.1 Verilog記述の解説

図 3.1の verilog記述を説明しよう。分岐命令がないのでまだコンピュータとは言
えないことからモジュール名は datapath2となっている。ここでは、図の構造がほと
んどそのままの形で記述されている。命令メモリから読み出された来た命令 (idatain)
は、opcodeと operandに分離される。この opcodeが ST命令 (4’b1000)のときに書き
込みが行われるようにしている。逆に accumには、ST命令以外で aluで演算された

値が格納されるようになっている。pcはリセット時に 0になり、後は順に 1ずつ増え
て行く。すなわち、0番地から順に命令が実行される。

‘include "def.h"

module datapath2(

input clk, input rst_n,

input [‘OPCODE_W-1:0] opcode,

input [‘DATA_W-1:0] ddatain,

output we,

output [‘ADDR_W-1:0] pcout,

output [‘DATA_W-1:0] accout);

reg [‘DATA_W-1:0] accum;

reg [‘ADDR_W-1:0] pc;

wire [‘DATA_W-1:0] alu_y;

assign we = (opcode == ‘OP_ST);

assign accout = accum;

assign pcout = pc;

alu alu_1(.a(accum), .b(ddatain), .s(opcode[‘SEL_W-1:0]), .y(alu_y));

always @(posedge clk or negedge rst_n)

begin

if(!rst_n) pc <= 0;

else pc <= pc + 1;

end

always @(posedge clk or negedge rst_n)

begin

if(!rst_n) accum <= 0;

else if(opcode !=‘OP_ST) accum <= alu_y;

end

endmodule

ここでは、一つだけ新しい構文が出てくる。それは、

always @(posedge clk or negedge rst_n)

begin

if(!rst_n) pc <= 0;

else pc <= pc + 1;

end

の部分である。
今回のアキュムレータマシンにはリセット入力が付いている。ご存じと思うが、す

べての CPUにはリセットボタンが付いていて、これを押すとシステムは初期化され
てしまう。無闇に初期化するとデータを破壊する可能性があるので、このボタンは目
立たない所にあって長時間押していないとリセットが掛からないようになっている。
我々のアキュムレータマシンにもこのリセット入力が付くわけだが、ここでは、リセッ
ト信号線が Lの時、リセットが掛かるようになっている。このような信号線をアク
ティブ-Lと呼び、ことでは信号名の後ろに nを付けて区別する。ちなみに多くのディ
ジタル回路でリセット入力がアクティブ-L1なので、このアキュムレータマシンもこ
の習慣に従っている。
さて、プログラムカウンタ pcは、リセット時に 0にしておく必要がある。このため

の記述が always @(posedge clk or negedge rst n)である。ここでは、clkが 0から１に変
化する (立ち上がる:posege)か (or)、rst nが 1から 0に変化する (たちさがる:negedge)
時 (@:at)はいつでも (always)、リセットが掛かる記述になっている。これを非同期リ
セットと呼ぶ。リセット時の動作は、if文の中に記述され、今回は単に pcを 0にし
ている。ちなみにリセットには同期と非同期があるが、ここでは非同期リセットしか
使わない2。
もう一つ分かりにくいかと思う記述は、aluモジュールを利用する部分である。

alu alu_1(.a(accum), .b(ddatain), .s(opcode[‘SEL_W-1:0]), .y(alu_y));

今回、ALUの制御入力 sには、opcodeの下位 3bitを入れる必要がある。ここでは、
SEL Wは def.hの define文で 3にしてあり、コロン:で書いた範囲のビット、すなわち
2:0が切り出されて、sに接続される。
では次に、これをテストするテストベンチ test.vを解説しよう。

/* test bench */

‘timescale 1ns/1ps

‘include "def.h"

module test;

parameter STEP = 10;

reg clk, rst_n;

wire [‘DATA_W-1:0] ddataout ;

wire [‘ADDR_W-1:0] pcout;

wire [‘OPCODE_W-1:0] opcode;

1理由については web の解説を参照
2理由については web の解説を参照

wire [‘ADDR_W-1:0] operand;

wire we;

reg [‘DATA_W-1:0] dmem [‘DEPTH-1:0];

reg [‘INST_W-1:0] imem [‘DEPTH-1:0];

assign {opcode, operand} = imem[pcout];

always #(STEP/2) begin

clk <= ˜clk;

end

always @(posedge clk)

if (we) dmem[operand] = ddataout;

datapath2 dp2(.clk(clk), .rst_n(rst_n), .pcout(pcout), .accout(ddataout),

.opcode(opcode), .we(we), .ddatain(dmem[operand]) );

initial begin

$dumpfile("datapath2.vcd");

$dumpvars(0,test);

$readmemh("dmem.dat", dmem);

$readmemb("imem.dat", imem);

clk <= ‘DISABLE;

rst_n <= ‘ENABLE_N;

#(STEP*1/4)

#STEP

rst_n <= ‘DISABLE_N;

#(STEP*12)

$finish;

end

always @(negedge clk) begin

$display("pc:%h inst:%h acc:%h", pcout, {opcode,operand}, dp2.accum);

$display("dmem:%h %h %h %h", dmem[0], dmem[1], dmem[2], dmem[3]);

end

endmodule

以前の演習での記述と同じく、周期 10nsec(=100MHz)のクロックを発生している。
命令メモリとデータメモリの初期設定を行っている。データメモリと同様に imemを

定義して、その初期値を以下の文で与えている。

$readmemb("imem.dat", imem);

命令は 2進数の方が読みやすいと思い、2進数になっている。imem.datを以下のよ
うに初期化してみよう。データファイルの中でも とコメントは同様に用いることが
できる3。

0001_00000000 // LD 0

0110_00000001 // ADD 1

1000_00000010 // ST 2

このメモリから読み出した命令は opcodeと operandに分離され、operandはそのま
まデータメモリのアドレスに使われる。

assign {opcode, operand} = imem[pcout];

この書き方は格好いいのだが、サイズをまちがえると酷い目に合うので注意が必要
である。

演習 3.4:

X,Y,W,Zが 0,1,2,3番地に格納されているとする。(X-Z) AND (Y-W)を演算する命
令コードを書き、実行して答えを確認せよ。なお、答えは accum上に出た時点で終わ
りにして良い。

3.2.2 分岐命令の付加

次に分岐命令を取り付けてみよう。BEZ(1001)命令ではアキュムレータが 0の時、
BNZ(1010)命令では 0でない時に、分岐条件が成立する。この信号を pcselとして図
に表しているが、Verilogの記述ではこの信号名を使っていない。これは、always文
中の if文中に、この程度の条件ならば簡単に書けるからである。条件が成立すれば、
pcに operandを書き込みそれ以外は pcを 1増やす。もちろん、rst nが 0ならばそち
らが優先されて pcは 0になる。

‘include "def.h"

module accum(

input clk, input rst_n,

input [‘OPCODE_W-1:0] opcode,

3実はちょっと問題があり、web を参照のこと

input [‘ADDR_W-1:0] operand,

input [‘DATA_W-1:0] ddatain,

output we,

output [‘ADDR_W-1:0] pcout,

output [‘DATA_W-1:0] accout);

reg [‘DATA_W-1:0] accum;

reg [‘ADDR_W-1:0] pc;

wire [‘DATA_W-1:0] alu_y;

wire op_st, op_bez, op_bnz, op_addi ;

assign op_st = opcode == ‘OP_ST;

assign op_bez = opcode == ‘OP_BEZ;

assign op_bnz = opcode == ‘OP_BNZ;

assign we = op_st;

assign accout = accum;

assign pcout = pc;

alu alu_1(.a(accum), .b(ddatain), .s(opcode[‘SEL_W-1:0]), .y(alu_y));

always @(posedge clk or negedge rst_n)

begin

if(!rst_n) pc <= 0;

else if (op_bez & (accum == 0) | op_bnz & (accum != 0))

pc <= operand;

else pc <= pc + 1;

end

always @(posedge clk or negedge rst_n)

begin

if(!rst_n) accum <= 0;

else if(!op_st & !op_bez & !op_bnz ) accum <= alu_y;

end

endmodule

ここでは、ALUを使った命令以外が 3つに増えているため、これをそれぞれ識別し
て信号名を付けている。このように命令によって対応する信号を設けることをデコー
ド (decode:解読)と呼ぶ。

assign op_st = opcode == ‘OP_ST;

assign op_bez = opcode == ‘OP_BEZ;

assign op_bnz = opcode == ‘OP_BNZ;

命令をデコードすると、対応する信号の値によって所定の制御を行えば良いので記
述がすっきりする。
実装した accum.vで、掛け算のプログラムを実行してみよう。imem.datを下のよう

に設定する。

0001_00000000 // LD 0

0110_00000001 // ADD 1

1000_00000000 // ST 0

0001_00000010 // LD 2

0111_00000011 // SUB 3

1000_00000010 // ST 2

1010_00000000 // BNZ 0

1001_00000111 // BEZ 7

このプログラムでは 1 番地と 2 番地の掛け算を行って 0 番地に書き込むため、
dmem.datは、例えば以下のようにしておけば、2× 3が計算されるはずだ。

0000

0002

0003

0001

テストベンチは以前の例題とほとんど同じなので省略する。

演習 3.5

1番地にXが格納されている。X+(X-1)+...+1を計算するプログラムをシミュレーショ
ンで実行せよ。

アキュムレータマシンの中身

4.1 抽象化のレベル

今まで、CPUの構成要素はブラックボックス化して扱い、その詳細に触れることは
しなかった。ALUは計算をする所、レジスタはデータを格納する所など、その機能
を中身を気にしないで理解してもらえば良かった。Verilogの記述もある意味で、抽
象化された記述である。ブラックボックス化あるいはモデル化はシステムを理解する
上で重要で、中身を完全に知ることはできないし、その必要もない。コンピュータは
物理的には大規模な集積回路 (LSI)チップで実現されるが、命令のレベルからチップ
までには下に示す階層が存在する。

• 命令セットアーキテクチャ：命令がどのように動くかを決める。次の章でくわ
しく紹介する。

• マイクロアーキテクチャ：CPU、メモリ、I/Oが、どのタイミングでどのように
演算やデータのやりとりを行い、制御を行うかを、大雑把にモデル化したもの。

• RTL (Register Transfer Leve)：マイクロアーキテクチャをさらに詳細化し、演算、
データのやりとりを詳細にモデル化したもの。

• ゲートレベル：RTLを実現するディジタル回路の論理ゲートの接続を示すレベル。

• 回路レベル：論理ゲートのトランジスタ接続を示すレベル。現在使われるトラ
ンジスタとは、CMOS (Complimentary Metal Oxide Semicondoctor)トランジス
タであり、NMOSと PMOSという対称的な性質を持つ二種類のトランジスタを
使って回路を構成する。

• レイアウトレベル：トランジスタを半導体チップ上に実装する方法。

今までは、マイクロアーキテクチャと RTLを中心に、命令セットアーキテクチャを
理解しようと試みてきた。ここでは、きちんと命令セットアーキテクチャを紹介し、
RISCの構造を説明する前に、下の階層であるゲートレベルについて触れる。この章
は、コンピュータを下のレベルまで理解しなくても良い場合は、スキップしても良い。

4.2 それぞれのモジュール

4.2.1 ALU(Arithmetic Logical Unit)

ALUは、加算、減算、乗算、論理演算、シフト、比較など、データを処理するため
の様々な演算を行ってくれる装置である。

ALUは図 2.4に示すように A, B二つのデータ入力と制御入力 sがあって、出力と
して Yがある。comの内容に応じて、A,Bの入力に対して所定の演算を行って Yに
出力するのが ALUの役目である。

a b

com
 y

16bit

16bit

3bit

16bit

図 4.1: ALUのモデル

ここでは、

s 出力
000 A　　　　　　　　　　
001 B

010 A AND B AND

011 A OR B OR

100 Aを 1ビット左シフト SL

101 Aを 1ビット右シフト SR

110 A+B ADD

111 A-B SUB

を想定した。では、この ALUの中身がどうなっているかを知っておこう。

マルチプレクサ

I0

I1
Y

S

S

I0

I1

Y

図 4.2: 2入力マルチプレクサ

ALUは複数の演算を行う演算ユニットがあって、これを comの値によって切り替え
る機能を持つ。これは実はマルチプレクサ、あるいはデータセレクタと呼ばれるディ
ジタル回路で実現される。2入力のマルチプレクサを図 4.2に示す。これは図中に示
すゲートの組み合わせで簡単に実現できる。マルチプレクサは、ディジタル信号の切
り替えスイッチで、S=0で入力 I0が、S=1で入力 I1が Yに出力される。ゲートの図
は面倒なので書かないが、4入力のマルチプレクサ、8入力マルチプレクサも想定で
きる。図 4.3には 8入力のマルチプレクサを用いて ALUを構成する方法を示す。

com2

I0 I1 I2 I3 I4 I5 I6 I7

Y

& | << >> + -

A B

com1
com0

S2
S1
S0

図 4.3: ALUの実現方式

このマルチプレクサという回路は、すごく簡単だが、実は CPUのデータパスを構
成する上で鍵となる部品であり、後で繰り返し登場することになる。

4.2.2 加算器

図 2.3中のそれぞれの演算に相当するモジュールのうち、ANDや ORはそれぞれ
ANDゲート、ORゲートで構成できる。シフトは実は 1ビットシフトならば、線の繋
ぎ変えだけで良い。となるとやはりここで中心になるのは加算器ということになる。
まずは、もっとも簡単な方法を説明しよう。

二進数のたし算を行うためには、下の桁から順に一桁ずつ足して行き、結果を上に
繰り上げる。この繰り上がりのことをキャリ (Carry:桁上げ)と呼ぶ。

11

0110

+ 0011

------

1001

この例では一番上の段にキャリを示している。加算を実行するためには、それぞれ
の桁の入力を A, B,下の桁からの入力を Ciとすると、以下の真理値表を実現すれば
良い。

A B Ci S Co

0 0 0 0 0

0 0 1 1 0

0 1 0 1 0

0 1 1 0 1

1 0 0 1 0

1 0 1 0 1

1 1 0 0 1

1 1 1 1 1

S

Co

A

B

Ci

S

A B

Ci Co

図 4.4: 全加算器

この論理回路も簡単なゲートの組み合わせで図 4.4に示すように実現できる。これ
が 1桁の全加算器 (full adder)である。n桁の全加算器を作る場合、必要な桁分だけ並

べて、下の桁の Coを自分の Ciに、自分の Coを上の桁の Ciに接続して数珠繋ぎに
する。これで、ちょうど手で計算する場合と同様に下のけたから桁上げ信号が伝搬し
て行って、加算ができる。一番下の桁 Ci 入力は使わないので Lowレベルにして、0
を入れる。この様子を図 4.5に示す。キャリ信号がさざ波 (リプル)のように伝わって
行くことからこの加算器をリプルキャリ加算器 (ripple carry adder)と呼ぶ。

S

A B

Ci Co

S

A B

Ci Co

S

A B

Ci Co

S

A B

Ci Co
’0’

A0 B0 A1 B1 A2 B2 A3 B3

S0 S1 S2 S3

A3-A0 B3-B0

CoCi
S3-S0’0’
+

図 4.5: リプルキャリ加算器

加算器は、ALUの中でもっとも複雑な構成要素なので、動作速度の足を引っ張る
ことが多い。リプルキャリアダーは、単純だが桁数分だけ遅延が積み重なるため、動
作が遅く、よほど簡単な場合以外は使われない。キャリが発生するかだけを調べる先
見回路を使うキャリ先見加算器 (Carry Look-ahead Adder)、アレイ状に部分加算器を
並べるプリフィックス加算器 (Prefix Adder)などが実際の CPUでは用いられる。

4.2.3 減算器

第 2章で触れたが、減算を行うためには 2の補数を加算すれば良く、2の補数を作
るには以下の手順を踏めばよかった。

• 1と 0を反転する。

• +1する。

1と 0の反転は NOTゲートあるいはインバータと呼ばれる簡単な論理ゲートで実
現できる。+1は加算器で可能だが、この目的だけで加算器を使うのはもったいないの
で、ちょっとトリックを使って、図 4.6に示す回路でA-Bの減算器が構成可能である。
ここでは、Bを反転して Aと加算するが、この際、加算器の一番下の桁のキャリ入

力を Hレベルにして 1を入れておく。このことで、A+(NOT B)+1が実現できる。加
算器と減算器は多くの部分が共有可能なので、図 2.3のように別モジュールにするの
はバカバカしい。そこで、多くの場合、図 4.7に示す加減算器を使う。
図の回路では、comが 0の時は、普通に Bが入力されて加算器になる。1の時は、

NOT Bが入力されると共に、もっとも下の桁の入力が 1になることで+1が行われ、
結果として減算器となる。

B0..B3 A0..A3

CoCi
S0..S3’1’
+

図 4.6: 減算器

B0..B3 A0..A3

CoCi
S0..S3’1’
+

I0 I1

Y
S

B0..B3

A0..A3

Y0..Y3

com

図 4.7: 加減算器

4.2.4 レジスタ

基本的にレジスタは Flip Flop(FF)から作る。FFは 1ビットの記憶回路で、最も良
く使われる D-FFは、図 4.8に示す記号で表され、D入力の値をクロックの立ち上が
りに同期して記憶して Qに出力する。FFの内部構造はここでは深く触れないが、ト
ランジスタを 12-14個程度必要とするやや複雑な回路である。

3章で触れたように CPUを含めて多くのディジタル回路はクロックに同期して全回
路が動作する同期式を用いている。この方式では、クロックの立ち上がりで、いつで
も FFにデータがセットされることになり、使い難い。そこで、通常用いるのは、D-FF
に Enable(イネーブル)入力を付けた Enable付き D-FFである。この FFは EN(Enable)
入力が 1の時にのみ D入力からデータを取り込み、0の場合は、現在記憶している
データをとっておく。この Enable付き D-FF を必要なビット数分並べてクロックと
EN入力を共通化したものがレジスタである。レジスタは、EN入力が 1の時にクロッ
クの立ち上がりに同期してデータを記憶する。
アキュムレータは、この Enable付き Flip Flopで実現するが、2章に示したアキュ

ムレータではプログラムカウンタ PCは毎回 1を足すので、Enableはなくても良い。

D
Q

CLK

D
Q

CLK

EN

0

1
D

D
Q

CLK

EN

図 4.8: D-FF

レジスタよりも大量にデータを記憶させる素子がメモリであるが、これについては 10
章に紹介する。ここで使うメモリは、同期式の SRAM(Static Random Access Memory)
である。半導体チップ内のメモリの多くはこの形を取る。

4.3 論理合成と論理圧縮

4.3.1 Verilog記述とゲートレベル

Verilog記述の一部は、対象とするハードウェアモジュールとそのまま対応する。例
えば、図 4.2に示すマルチプレクサは、以下の記述がそのまま対応する。

assign Y = S ? I1: I0;

AND, ORなどの論理演算はそれぞれ対応する論理ゲートに変換される。

assign Y = A & B;

assign Y = A | B;

図 4.8に示す D-FFは、Verilog記述の以下に対応する。

always @(posedge CLK)

if(EN) Q <= D;

しかし、Verilogの記述が常に一対一でゲートレベルの回路図に対応するのではない。

assign Y = A +B ;

と書いた場合、これがどのような加算器に変換されるかは、論理合成用ツールとそ
れに与えるパラメータに依存する。論理合成用ツールは遅延時間に余裕があれば、コ
ストの小さいリプルキャリ方式の加算器を生成し、高速な演算が要求されれば、プリ
フィックス方式の加算器を生成する。また、実は先に示したマルチプレクや論理演算
の記述も場合によっては他の回路と共用されたりする関係で、違った形で論理合成さ
れる可能性がある。次にこの論理合成ツールについて解説する。

4.3.2 合成ツールと使い方

今までは、Verilogはハードウェア構造を記述し、シミュレーションするだけだっ
た。しかし、記述されたハードウェアは何らかのデバイス上で実現しなければ意味が
ない。このためには Verilog記述をゲート部ベルの接続図 (ネットリスト)に変換し、
様々の最適化を行う作業が必要である。これを論理合成、圧縮と呼ぶ。今日はこの合
成ツールの使い方を習得し、これを使って今まで設計してきた POCOを合成、圧縮す
ることが目的である。
論理合成圧縮ツールは、プログラミング言語におけるコンパイラに相当するが、対

象がハードウェアであるために、コンパイラよりもはるかに処理が複雑で、時間がか
かり、また、ツール自体も高価である。現在、学生レベルで無料あるいは安価に使え
る論理合成圧縮ツールには以下のものがある。

• FPGA(Field Programmable Gate Array)、すなわち書き換え可能な大規模 LSIの
ベンダの提供する合成ツール。Xilinx社の ise、Altera社の Quartus II は、web
パックと呼ばれる簡易版を無料で提供している。webパックは一部の機能が制
限されているが、本書で記述するの回路ならば十分合成可能である。

• VDEC(VLSI Design and Education Center: http://www.vdec.u-tokyo.ac.jp/welcome.html)
で、研究教育目的で安価で本格的な CADを利用可能である。

• PARTHENON、このテキストで、以前、用いていた合成ツール。元々NTTが開
発したが現在は NPO法人が管理し、独自の記述言語である SFLのみを受け付
け、Verilogは受け付けない。

ここで、論理合成、圧縮用のツールとその利用法については、本書中ではではこれ
以上記述せず、web上で以下の環境を提供することにする。論理合成圧縮用のツール
はバージョンアップが激しく、対象となる LSIの進歩も急速である。本書中に詳細を
記述してもあっという間に記述が古くなってしまうだろう。web上に置けばその都度
更新することで最新の利用法をサポートすることができる。現在利用可能なのは以下
の環境である。

• Xlinix社のWebPackを利用した環境。Windows/Linuxの走る PCがあれば無料
で個人レベルでインストールして使うことができる。また、同社が販売してい
る Spartan 3E用のテストボード上で POCOを実装し、もぐらたたきやピンポン
ゲームなどを動かしてみることができる。

• VDECにより提供される Synposys社の Design Compilerを用いた論理合成およ
び圧縮。これは、ライセンスの利用を許可された大学や高等専門学校のみで実
験ができる。計算センターや、研究室の指導教員に相談されたい。CADツール
のみの演習は、オクラホマ大の開発したフリーのライブラリを用いて可能であ
る。さらに進んで、富士通 e-shuttle社のライブラリを用いて Synopsys社の IC
Compilerを利用して実際のチップを設計実装することができる。これも web上
で掲示しているので、利用のためには研究室の指導教員に相談されたい。

汎用レジスタマシンへの拡張

5.1 汎用レジスタマシン

今までアキュムレータマシンの命令について検討してきた。3章で分岐命令を導入
したことで、簡単なプログラムが書けるようになった。プログラムを書くことのでき
る命令の一揃いを命令セットと呼び、命令セットの作り方、あるいはその作り方に基
づいて出来上がった命令セットのことを命令セットアーキテクチャと呼ぶ。命令セッ
トアーキテクチャとは 4章で紹介した抽象化の上のレベルに当たり、ハードウェアと
ソフトウェアのインタフェースである。ハードウェア設計者は、この命令セットアー
キテクチャを実現するために様々な要求に応じて様々な構成を作ることができる。お
金は掛かっても高速な CPUが欲しい場合と、遅くても良いので安くて電力消費が少
ないものを作りたい場合では、構成は全く違ったものになる。しかし、同じ命令セッ
トアーキテクチャを使っていれば、同じプログラムが動く。一方、プログラマは、ど
のようなハードウェアになっているのかを気にしないで、命令セットだけを知ってい
れば、プログラムを書くことができる。
この章では、より高度な命令セットアーキテクチャをもっとも簡単な実現モデル (マ

イクロアーキテクチャ)を示しながら紹介する。
さて、アキュムレータマシンは、ループを含むアルゴリズムの実行が可能であった

が、以下の問題があった。

• アキュムレータだけしかレジスタがないので、演算の度に結果をメモリに格納
する必要がある。

• アキュムレータだけしかレジスタがないため、メモリのアドレスを動的に指し
示す機能がない。つまりポインタの実現ができない。

さらに、これはアキュムレータマシンの一般的な問題ではないのだが、アドレスの
指定が長くなりすぎると面倒なので、アドレス空間を 4ビット、すなわち全体で 16と

した点にも問題がある。コンピュータの歴史を振り返ると、開発するプログラムサイ
ズは年々増大し、必要とされるアドレス空間は直線的に増加している。現在の高性能
のマイクロプロセッサが 64ビットアーキテクチャを採用している主な原因は 32ビッ
トアーキテクチャではアドレス空間が不足するためである。このことを考えると、ア
ドレス空間 4ビットはいくら演習用でもちょっとひどすぎる。
そこで、以下の改造を行おう。

• レジスタ数を 8個に増強する

• アドレスを 16bitに拡張して、アドレス空間を 64Kとする

まず、8個のレジスタは、基本的に全てが同じ機能を持たせるようにする。初期の
マシン（現在でも 8ビットマシンなどは）ではレジスタの機能が専門化され、レジス
タによって可能な演算や操作（メモリのアドレスのみ格納する等）が決っている場合
が多かった。これを専用レジスタマシンと呼ぶ。しかし、専用レジスタマシンは、レ
ジスタの機能を考えてプログラムをするのが大変で、コンパイラも作りにくい。この
ため、半導体の面積の余裕ができるにつれ、一部の例外を除いてすべてのレジスタが
同様にすべての機能を実現できるマシン、すなわち汎用レジスタマシンに移行した。
我々も汎用レジスタマシンを考えよう。
汎用レジスタマシンでは、レジスタが複数あるので、代表的な演算命令のオペラン

ドが 2または 3となる。すなわち、2オペランド命令あるいは 3オペランド命令と
なる。

ADD r0, r1 r0<-r0+r1

ADD r0, r1, r2 r0<-r1+r2

ここで、r0は結果が格納されるレジスタであり、ディスティネーションレジスタ（オ
ペランド）と呼ぶ。一方、r1(r2)は、元となるデータが格納され、演算によって変化
しないレジスタであり、ソースレジスタ（オペランド）と呼ぶ。マシンによっては、
オペランドに、アキュムレータマシン同様、メモリのアドレスを書くことができるも
のもある。

ADD r0, 1000 r0<-r0+(1000)

ADD 1000, 2000, 2002 (1000)<-(2000)+(2002)

最初の命令は r0と 1000番地の中身を足して、答えを r0に格納し、次の命令は 2000
番地と 2002番地の中身を足して答えを 1000番地に格納する。
ここでは、ディスティネーションオペランドを一番左に書く表記法を採用している。

この表記法は IBMや Intelのマシンで使われている方法で、現在一般的である（ディ
スティネーションオペランドを一番右に書くやり方もあり、DECやモトローラのマシ
ン、日立の SHなどに使われている）。

5.2 汎用レジスタマシンの分類

汎用レジスタマシンの命令セットは、演算命令のオペランドにメモリを指定できる
かどうかによって以下のように分類される。

• まったく指定できない: register-registerアーキテクチャ、load-storeアーキテク
チャまたは RISC(Reduced Instruction Set Computer) と呼ばれる。オペランド
にメモリを指定できないため、演算を行う場合、必ずレジスタに値を Loadし
てからレジスタ間で演算し、答えをメモリに Storeする。一定の処理を行うた
めの命令数は多くなるが、個々の命令は単純化され固定長で実装可能である。
SPARC,MIPS,ARM,SH-4 などのマイクロプロセッサはこの命令形式を用いて
いる。

• どこでも指定できる: memory-memoryアーキテクチャ、オペランドのどの部分
でもメモリが指定できる。もちろんレジスタ間でも演算が可能だが、メモリ中の
データ同士を直接計算する場合が多くなる。一定の処理を行うための命令数は少
ないが、個々の命令は可変長にする必要があり、複雑になるため CISC(Comptex
Instruction Set Computer)と呼ばれる。DECのVAX-11で用いられ、80年代はメ
ジャーであったが、性能向上が難しいことから現在はほとんど使われていない。

• 1つだけ指定できる: register-memoryアーキテクチャ、オペランドに一ヶ所だけ
メモリを指定できる。多くの場合 2オペランド命令でソースオペランドに限ら
れる。register-register型とmemory-memory型の中間的な性質を持つ。一定の処
理を行うための命令数はさほど多くならず、命令は可変長にする必要があるが、
memory-memory型ほど悲惨なことにはならない。現在、デスクトップやラップ
トップで最も良く用いられている Intel/AMDの IA32,IA64はこの型に属する。

例えば 0x1000番地のデータと 0x1010番地の内容を加算して 0x2000に格納する場
合、register-register型では以下のようになる。

0x1000の中身をレジスタに持ってくる (LD)
0x1010の中身を別のレジスタに持ってくる (LD)
加算
0x2000に答えを格納 (ST)

このように register-register型は、命令数が多いことがわかる。しかし、メモリのアド
レスは LD,STのみで済み、命令長を固定にすることが可能である。
一方、memory-memory型では一命令ですむ。

0x1000の中身と 0x1010の中身を足して 0x2000に格納

図 5.1: 汎用レジスタマシンのデータパス

しかし、オペランドにはレジスタも指定できるので、命令長は長くなったり短くした
りしなければならないことがわかる。

register-memory型は、ディスティネーションは、レジスタのみ許される場合が多い
ので以下のようになる。

0x1000の中身をレジスタに持ってくる (LD)
0x1010の中身を加算
0x2000に格納 (ST)

これも、ADD命令にメモリアドレスを書く場合と書かない場合で、命令長を変える
必要があるが、その可変の程度はmemory-memory型よりもひどくはない。一方、一
命令では済まず、3命令必要になっている。

3つの方式のハードウェア構成を考えてみよう。複数の汎用レジスタは、レジスタ
ファイルというハードウェアモジュールで実装される。レジスタファイルは小規模な
マルチポートメモリであり、ここでは図 5.1に示すように、読み出しに 2ポート (a,b)、
書き込みに 1ポート (c)を持っている。今、レジスタファイルの中に 8個のレジスタ
が入っているとすると、好きな番号のレジスタを aポート、bポートから独立に読み
出すことができる。同時に、cポートからデータを書き込むことができる。
これを単純にアキュムレータの代わりに入れ替えた構成が、図の真ん中に示す register-

memory型である。この点で、register-memory型はアキュムレータマシンのもっとも
自然な発展であると言える。次に左に示す register-register型は、メモリから ALUへ
の直接入力が存在せず、全てのデータが一度レジスタを介して演算される。一方で、
memory-memory型はレジスタとメモリが同様に ALUに接続されている構成である。
さて、register-memory型は、アキュムレータマシンの自然の発展であり、また、最

も良く用いられている Intel/AMDの CPUの命令セットは、この形である。しかし、こ
の型が最もうまく行っていると勘違いしてはならない。実は Intel/AMDのマイクロプ
ロセッサは、Pentium II以降は、register-memory型の命令を内部的に register-register
型に変換して実行している。register-memory型の見掛けを維持しているのは、アキュ
ムレータマシンの自然な発展として register-memory型を採用してしまったが故に蓄
積したプログラムとの互換性 (Compativility)を保持するために過ぎない。高速に実行
するという点でいうと、register-register型つまり RISCはとことん優れており、わざ
わざ実行時に変換して実行せざるを得ないのだ。つまり、これらのプロセッサは外面
が register-memory型でも、一皮剥けば RISCであり、組み込み用ではほとんどの CPU
が RISCであることを考えると、プロセッサの命令セットという点では RISCの圧勝
である。
そこで、ここでも、今までのアキュムレータマシンを register-register型、すなわち

RISCに拡張していくことにする。ここで、この RISCに名前を付け、POCOと呼ぶ
ことにする。

5.3 16bit RISC: POCO
POCOの基本構成を以下のように決める。

• 命令メモリ:16bit×アドレス 16bit=アドレス空間 64K

• データメモリ:16bit×アドレス 16bit=アドレス空間 64K

• レジスタ:16本 (全部 16bit):r0-r7

• 機械語命令：16bit固定長

これは、小規模な組み込み CPUとしてはそこそこリアルな構成である。

5.3.1 メモリの読み書きをどうするか？

まずもっとも問題となるのはメモリの読み書きである。メモリのアドレスを 16bit
に拡張したため、LD命令と ST命令で直接アドレス指定すると、命令の全体長を越
えてしまう。また、ポインタや配列の実現ができないなどのアキュムレータマシンの
問題点を解決できない。
そこで、POCOでは他の RISCと同じくレジスタの中身で、メモリのアドレスを表

す方法を採用する。これをレジスタ間接メモリアクセスと呼ぶ。

LD r0, (r1) 　　 r0 <- (r1)

上記の命令を実行すると、r1に中身がアドレスとなり、読み出されたデータが r0に
格納される。r1が 0ならば 0番地、100ならば 100番地が読み出される。r0-r7には
16ビットの数が格納されるので、全アドレス空間をアクセスすることができる。こ
の場合、r1はポインタそのものであり、この値に演算を行うことで配列、スタックな
どのデータ構造を実現することができる。

STも同様に表される。

ST r0, (r1) r0 -> (r1)

この場合、r0の中身が、r1で指定するアドレスに書き込まれる。すなわち、r1が 0
ならば 0番地、100ならば 100番地に r0の中身が書き込まれることになる。ここで、
データの移動の方向が ST命令のみは左（レジスタ）から右 (メモリ）になる点に注
意されたい。

5.3.2 基本演算命令

次に、基本的な演算命令はレジスタ同士の演算である。

ADD r0, r1 r0 <- r0+r1

SUB r1, r2 r1 <- r1-r2

今回、ALUについてはビット幅が 16ビットになっただけで、アキュムレータマシ
ンと同様の機能を想定する。この場合以下の命令が定義できる。

AND r0, r1 r0 <- r0 and r1

OR r1, r2 r1 <- r1 or r2

SL r0 r0 <- r0 << 1

SR r2 r2 <- r2 >> 1

さて、POCOは RISCすなわち register-register型なので、演算はレジスタ同士でし
か許していない。ところが、ここで問題があることに気づく。メモリからのデータが
レジスタ間接指定で、演算がレジスタ同士でしか許されないのならば、レジスタにア
クセスする番地を入れることができないではないか？最初にレジスタに値を設定する
手段を設けておかないと、メモリからデータを取ってくることができなくなってしま
う。これでは困る。そこで、命令中に直接数字を指定する命令を用意する。これがイ
ミーディエイト命令である。

LDI r1,#3　 r1 <- 3

これは、r1に 3という数値を直接入れてしまう命令である。メモリの番地を指定し
ているのではなく、あくまで 3という数自体を入れていることに注意されたい。この
指定方式を直値（即値）指定あるいはイミーディエイト (Immediate)指定と呼び、こ
の指定を使った命令を直値命令あるいはイミーディエイト命令と呼ぶ。同様に、演算
についてもこの考え方を用いて、

ADDI r2,#5　 r2 <- r2 + 5

とできると便利である。命令のオペランドが直接レジスタに入ったり、演算に使わ
れたりするのは、やや特殊なケースと言って良い。このため、アセンブラ表記ではこ
のことを強調するために、イミーディエイトデータの前に＃を付けて表すのが普通で
ある。
ここでさらに一つ問題がある。このイミーディエイトデータを指定する領域を命令

内に確保しなければならないが、命令は全体で 16ビットに納めなければならない。こ
の中にはオプコードが最低 5ビット（32命令分）は必要であるし、レジスタを指定
する 3ビット (r0-r7で 8つ分)も必要である。したがって、イミーディエイトデータ

に使える分は 8ビットしかない。そこで、8ビットのデータを 16ビットに幅を広げる
手段が必要である。ここで 2章で紹介した符号拡張が登場する。POCO内で使う符号
付き数は、通常のコンピュータと同じく 2の補数表現であらわす。そこで、符号を考
えて拡張してやるのが原則である。しかし一方、符号を考えない数も扱う必要がある
ため、イミーディエイト命令は符号付きと符号無しの両方を用意する必要がある。

POCOでは以下の 4つを用意する。

LDI rd,#X Load Immediate rd <- X (符号拡張)
LDIU rd,#X Load Immediate Unsigned rd <- X (符号拡張なし)
ADDI rd,#X ADD Immediate rd <- rd + X (符号拡張)
ADDIU rd,#X ADD Immediate Unsigned rd <- rd + X (符号拡張なし)

SUBIは用意する必要はない。これは、

ADDI r1, #-1

が可能であるからだ。他の演算も重要度はさほど高くないので、イミーディエイト
命令としてはこれだけで良いだろう。ちなみに、イミーディエイトフィールドを 8bit
にする、という選択は、「命令の長さが足りないから」という理由だけでなく、「16bit
全部持たせると無駄だから」という点からも合理的である。プログラムを書いてみる
と多くの場合、イミーディエイト命令では、1(あるいは 2とか 3とか小さな数)を足
すとか引くとかが多く、意味のある桁数はさほど長くない。このような場合、全部持
たせるというのは命令セットの設計上無駄が大きくなってしまう。
以上で基本的な演算が可能になる。例えば 0番地の内容と 1番地の内容を加算して

2番地に格納するプログラムは以下のようになる。

LDIU r0,#0

LD r1,(r0)

ADDIU r0,#1

LD r2,(r0)

ADD r1,r2

ADDIU r0,#1

ST r1,(r0)

この場合アドレスが小さいので、LDIU, ADDIUは LDI, ADDIと同じである。こ
こでは r0をポインタとして 1ずつ進める方式を取っているが、その都度値をじかに
LDIU命令で設定しても良い。実はここまでのイミーディエイト指定では、下位 8bit
しか指定されず、上位 8bitは符号ビットか 0になってしまう。このため、16bitレジス
タの上位 8bitに値を設定することができず不便だが、これに対処する命令 LHI(Load
High Immediate)は、演習で実装しよう。

5.3.3 命令フィールドの設計

他の RISC同様、POCOでは 16bit固定長に全命令を実現する必要がある。まず、オ
プコードを 5bitとする。これだと 32命令実現可能である。32はいくら教育用といえ、
少なすぎるような気がするが、後で RISC共通の方法を使って拡張する。
イミーディエイト命令は、レジスタを 1つ指定する必要があるので、オプコードの

後は、そのレジスタの番号を入れて残りをイミーディエイトデータに割り当てて、以
下のようにする。

LDI rd,#X 01000 ddd XXXXXXXX rd <- X (符号拡張)
LDIU rd,#X 01001 ddd XXXXXXXX rd <- X (符号拡張なし)
ADDI rd,#X 01100 ddd XXXXXXXX rd <- rd + X (符号拡張)
ADDIU rd,#X 01101 ddd XXXXXXXX rd <- rd + X (符号拡張なし)

例えば

ADDI r5,#100 01100 101 01100100

となる。
このようにオプコード (opcode: 01100),レジスタ (ddd)、イミーディエイト (XXXXXXXX)

など、命令中のそれぞれ意味を持った部分のことを命令フィールドと呼ぶ。
次に ADD などレジスタを 2 つ指定するタイプの命令について考えよう。同様に

opcode は 5 ビットとし、レジスタ二つを指定するフィールドを設けると、残りは、
16-5-3*2=5bit余る。ここで、この余りのフィールドを用いて命令数を増やすことを
考える。すまわち、全ての ALUを用いた命令は、opcodeを 00000とし、この余りの
5ビットのうち下位 3ビットを ALUのコマンドに入れて演算を指定する。この下位 5
ビットを function(funct)フィールドと呼ぶ。3bit目、4bit目の 2ビットが完全に余る
が、これは将来のために取っておくことにし、ALU命令ではここを 00とする。すな
わち、opcodeが 00000で、functionが 00XXXであれば、ALU命令であり、XXXが
ALUのコマンドとなる。
今までと同じ ALUを利用すれば以下の命令が定義される。

NOP 00000 --- --- 00000

MV rd,rs rd <- rs 00000 ddd sss 00001

AND rd,rs rd <- rd AND rs 00000 ddd sss 00010

OR rd,rs rd <- rd OR rs 00000 ddd sss 00011

SL rd rd <- rd<<1 00000 ddd --- 00100

SR rd rd <- rd>>1 00000 ddd --- 00101

ADD rd,rs rd <- rd + rs 00000 ddd sss 00110

SUB rd,rs rd <- rd - rs 00000 ddd sss 00111



72 第 5章 汎用レジスタマシンへの拡張

ddd同様、sssにはソースレジスタ rsを示す番号 (r0-r7:000-111)が入る。RISCでは
先の図 5.1に示すように、ALUの B入力につながっているのはレジスタファイルの B
ポートなので、ALUのコマンド 001: THBは、LDではなくレジスタ間の転送命令で
あるMV(move)となる点に注意されたい。THAは、今まで同様に NOP(No-operation)
命令となり何も行われない。この方法はなんだかセコいようだが、命令数を増やし、
ハードウェアを簡単にするのに有効なので、ほとんどの RISCで同様の方法を用いて
いる。
それでは LD, ST 命令を ALU 命令と同じ形で表現することにしよう。ここでは

functionの 5,4ビット目が 01で LD, STを表すこととした。STを先に持ってきたの
はアキュムレータマシンと命令コードを同じにした方が慣れていていいかと思ったた
めで、深い意味はない。aaaはメモリアドレスを示すレジスタという意味で、やはり
r0-r7までのレジスタ番号が入り 000-111で表す。

ST rs, (ra) rs -> (ra) 00000 sss aaa 01000

LD rd, (ra) 　　 rd <- (ra) 00000 ddd aaa 01001

演習 5.1

0番地,1番地にそれぞれ A,Bが格納されている。(A-B) OR Bの演算を行い、2番地に
答えを格納するプログラムをアセンブラ表記、機械語の両方で示せ。

5.4 POCOの内部構造
アキュムレータマシンと同じく、POCOのデータパスは、大きく分けて、命令を命

令メモリから取ってくる部分（フロントエンドと呼ぶ）取ってきた命令に基づいて演
算やメモリアクセスを行う（バックエンドと呼ぶ）に分けることができる。図 5.2に
概略図を示す。
フロントエンドでは PC(Program Counter)の指し示すアドレスに格納されている命

令を読み出す。読み出された命令は idatainからバックエンドに送られる。PCは次の
命令を読み出すためにカウントアップされる。

5.4.1 レジスタ間演算命令の実行

次にバックエンドに目を向けよう。バックエンドの主な構成要素は、r0-r7までのレ
ジスタがまとめて格納されているレジスタファイルと演算を行う ALU、データメモ
リであり、データの流れを切り替えるためにマルチプレクサが配置されている。レジ
スタファイルは図 5.3に示す 3ポート構造を想定する（現時点では、実は読み出しの
aポートアドレスと書き込みの cポートアドレスは共有可能である）。レジスタファイ

図 5.2: データパスの概略図

ルは一種のメモリであり、aadrから与えられた番号（3ビット）のレジスタが内容が
rf aに読み出され、badrに与えられた番号（3ビット）のレジスタの内容が rf bに読
み出される。一方、書き込みについては、cadrに与えた番号のレジスタに対して、rwe
が 1の時のクロックの立ち上がりで、rf cからのデータが書き込まれる。aadr, badr,
cadrは後で解説するように命令中のレジスタを表すフィールドにより指定する。

reg_file
....

aadr

cadr

rf_c

rf_a rf_b
badr

rwe

図 5.3: レジスタファイル

POCOは、register-registerマシンなので、メモリから ALUへの直接パスは存在し
ない。メモリから読み出したデータは、マルチプレクサを介してレジスタファイルの
rf cから書き込まれる。
まずは、レジスタ同士の演算命令を実行する場合を考える。
レジスタ同士の演算の機械語フォーマットは以下の通りである。まず opcodeは 00000

で、これに続く 3ビット (10から 8ビット)でディスティネーションレジスタを、次
の 3ビット (7から 5ビット)でソースレジスタを、最後の function5ビット (4から 0
ビット)で命令の種類を表す。
例えば、ADD r0,r1を例に取ると、この命令の機械語は、以下のフォーマットになっ

ている。

ADD r0,r1 00000 000 001 00110

10から 8ビット目のディスティネーションレジスタ番号をレジスタファイルの A
ポートアドレス (addra)に繋いで、ディスティネーションレジスタを Aポートから読
み出す。7から 5ビット目のソースレジスタ番号の方は Bポートアドレス (addrb)に
繋いで、ソースレジスタを Bポートから読み出す。ALUの入力用マルチプレクサの
制御入力 alu bselは 00として、レジスタファイルからの値を入力してやる。

ALUで行う演算の種類は comで指定する。POCOの命令コードは、最後の funct
コードの下 3ビットが、comと一致するようになっている。つまり、110ならば加算

が行われる。先に述べたように、この方法はセコいようだが、単に繋げば良いため楽
なことがわかる。ディスティネーションレジスタとソースレジスタ間の演算結果は、
レジスタファイルに書き込まれる。このため、Cポートのアドレスには、ディスティ
ネーションレジスタ番号を入れ、レジスタファイル書き込み用マルチプレクサの制御
線 rf cselを 0にする。書き込み制御信号 rwe=1とすればレジスタファイルへの書き
込みが行われる。上記のデータの流れについて図 5.4に示す。

a b

com
y

ddataout

ddatain

Data
Memory

cadr

daddr

ext ext0

r0-r7

Back-end

Instruction

aadr10:8 badr

7:5

2:0

A-port

B-port

rwe=1

図 5.4: レジスタ間演算命令の動き

5.4.2 LD/ST命令の実行

LD命令、ST命令の場合、メモリのアドレスを格納しているレジスタは、ソースレ
ジスタとして 8-6ビット目に指定されている。Aポートの出力は、直接データメモリ
のアドレスに接続しているので、レジスタから値を読み出せば自動的にアドレスが与
えられる。LD命令の場合、読み出されたデータは、ALUからの演算結果に代わって
Cポートで与えられたディスティネーションレジスタに格納される。このため、レジ
スタファイルの書き込み用マルチプレクサの制御線 rf cselを 1とする。一方、Bポー
トの出力は直接データメモリの入力データに接続されているので、ST命令が実行さ
れる場合、ソースレジスタの中身が書き込まれる。この場合はデータメモリの書き込
み制御線 (we)を 1にし、レジスタファイルの書き込み制御線 (rwe)を 0にする。この
様子を図 5.5に示す。

a b

ddataout

ddatain

cadr

daddr

ext ext0

r0-r7

Back-end

Instruction

aadr10:8 badr

7:5

2:0

A-port

B-port

rwe=1  (LD)

we=1
(ST)

図 5.5: LD/ST命令の動き

5.4.3 イミーディエイト命令の実行

イミーディエイト命令は、ソースレジスタの指定がなく、下位 8ビットがコード中
に指定されている。例を以下に示す。命令は opcodeにより判定する。

LDI r0,#-32 01000 000 11100000

LDIU r0,#224 01001 000 11100000

ADDI r0,#-32 01100 000 11100000

ADDIU r0,#224 01101 000 11100000

ここで、機械語中の 8ビットを 16ビットのレジスタ中のデータと演算する際にビッ
トの長さを揃えて 16ビットにする必要がある。この際に、符号を考えて 16ビット化
する、すなわち符号拡張する方式と 0を埋めて 16ビット化する 0拡張する方式があ
る。例えば ADDIは前者なので、0xffe0すなわち-32が r0に加算され、ADDIUは後
者なので 0x00e0すなわち 224が加算される。
この命令を実行するには、まず機械語の下 8ビットを ALUの B入力まで引っ張っ

て来る必要がある。引っ張ってきた結果は、符号拡張 (ext)あるいはゼロ拡張 (ext0)
されて入力される。ちなみに ext, ext0は、図中では箱で表しているが、実際は最上位
ビットや 0を並べて出力するだけで、非常に簡単なハードウェアである。ALUの B
入力選択用のマルチプレクサの制御入力は符号拡張の場合は 01、ゼロ拡張の場合は
10にしてやる。
さて、イミーディエイト命令は直接 B入力のデータを書き込むこと (LDI, LDIU)と

加算する (ADDI, ADDIU)だけなので、これを直接 ALUの comに入れてやる。ALU
コマンド選択用のマルチプレクサ制御入力 comselを命令に応じて切り替える。レジ
スタファイルへの書き込みは、レジスタ同士の演算命令と同じである。この様子を図
5.6に示す。

5.4.4 制御回路

今までの解説で、各部のデータの流れはご理解いただけたと思う。この流れを実現
するように、命令の種類に応じて制御信号を設定してやればそれぞれの命令を動作さ
せることができる。制御する対象はALUのコマンド入力を選択する comsel、ALUの
B入力の選択用 alu bsel、レジスタファイルの書き込みを選択する rf cselで、これら
はマルチプレクサの制御信号である。マルチプレクサの制御信号は、何が選択されて
も構わない場合があり、このような時はドントケアとなる。一方、レジスタファイル
の書き込み信号 rweとメモリの書き込み信号 weは、書き込む際は 1にするが、その
他の場合は 0にしておかなければならない。レジスタやメモリはデータを取っておく

図 5.6: イミーディエイト命令の動き

のが役目なので、書き込み信号がドントケアということはあり得ず、書く場合以外に
は 0にしておかなければならない。表 5.1にそれぞれの命令に対する制御信号を示す。

表 5.1: 各命令の制御信号
comsel alu bsel rf csel we rwe

ADDI 10 01 0 0 1
LDUI 01 10 0 0 1
SUB 00 00 0 0 1
LD 00 - 1 0 1
ST – - - 1 0

最後に今までの POCOのデータパスの全体図を示す。線がごちゃごちゃしている
印象を受けるかもしれないが、これでも CPUとしては単純な部類に入る。実際、ご
ちゃごちゃしても線が全部書けるということは、普通の CPUではあり得ないのだ。

図 5.7: POCOのデータパス

演習 5.2

ADDUI, LDIを表に付け加えよ。

5.4.5 POCOのVerilog記述

レジスタファイルの Verilog記述

最初に図 5.3のレジスタファイルの Verilog記述は以下のようになる。

‘include "def.h"

module rfile (

input clk,

input [‘REG_W-1:0] aadr, badr, cadr,

output [‘DATA_W-1:0] a, b,

input [‘DATA_W-1:0] c,

input we);

reg [‘DATA_W-1:0] r0, r1, r2, r3, r4, r5, r6, r7;

assign a = aadr == 0 ? r0:

aadr == 1 ? r1:

aadr == 2 ? r2:

aadr == 3 ? r3:

aadr == 4 ? r4:

aadr == 5 ? r5:

aadr == 6 ? r6: r7;

assign b = badr == 0 ? r0:

badr == 1 ? r1:

badr == 2 ? r2:

badr == 3 ? r3:

badr == 4 ? r4:

badr == 5 ? r5:

badr == 6 ? r6: r7;

always @(posedge clk) begin

if(we)

case(cadr)

0: r0 <= c;

1: r1 <= c;

2: r2 <= c;

3: r3 <= c;

4: r4 <= c;

5: r5 <= c;

6: r6 <= c;

default: r7 <= c;

endcase

end

endmodule

この書き方はレジスタを個別に宣言していて、あまり賢いやり方ではない。

reg [‘DATA_W-1:0] r[0:7];

とメモリの形で宣言すれば、もっとずっと楽である。なぜわざわざ個別に宣言したか
というと、メモリの形で宣言すると後で gtkwaveでシミュレーションをする際に中身
を見ることができなくなってしまうためである。通常、メモリは膨大なので、波形シ
ミュレータでその中身をいちいち表示することはしない。そうは言っても演習時にレ
ジスタが見えないと困るので、今回はこのようにした。あくまで演習用と考えて欲
しい。
ここで always文の中で if文と case文を使っている。case文は、括弧内に示した信

号について、そのデータに応じた処理をコロン：以降に書くことができる。どれも当
てはまらない場合は、default：以降が実行される。大変見やすくて便利な文だが、if
文と同じく、always文でレジスタを記述する場合およびその他の特殊な構文中でしか
使えない。

weが 1である時に、書き込みを行う cadrの値に応じて入力 cがそれぞれのレジス
タに書き込まれる。もちろん書き込みが行われるのはクロックの立ち上がりに同期し
ている。case文は便利だが、特殊な構文中だけしか使えないため、読み出しの部分で
は、? : のマルチプレクサ構文を使っている。

POCO本体の Verilog記述

さて、このレジスタファイルを使って、図 5.7を比較的そのまま Verilogで記述した
ものを示す。まず、以下のように定義をしておく。

‘define DATA_W 16

‘define SEL_W 3

‘define REG 8

‘define REG_W 3

‘define OPCODE_W 5

‘define IMM_W 8

‘define DEPTH 65536

‘define ENABLE 1’b1

‘define DISABLE 1’b0

‘define ENABLE_N 1’b0

‘define DISABLE_N 1’b1

‘define OP_REG ‘OPCODE_W’b00000

‘define OP_BEZ ‘OPCODE_W’b10000

‘define OP_BNZ ‘OPCODE_W’b10001

‘define OP_LDI ‘OPCODE_W’b01000

‘define OP_LDIU ‘OPCODE_W’b01001

‘define OP_ADDI ‘OPCODE_W’b01100

‘define OP_ADDIU ‘OPCODE_W’b01101

‘define F_ST ‘OPCODE_W’b01000

‘define F_LD ‘OPCODE_W’b01001

‘include "def.h"

module poco(

input clk, rst_n,

input [‘DATA_W-1:0] idatain,

input [‘DATA_W-1:0] ddatain,

output [‘DATA_W-1:0] iaddr, daddr,

output [‘DATA_W-1:0] ddataout,

output we);

reg [‘DATA_W-1:0] pc;

wire [‘DATA_W-1:0] rf_a, rf_b, rf_c;

wire [‘DATA_W-1:0] alu_b, alu_y;

wire [‘OPCODE_W-1:0] opcode;

wire [‘OPCODE_W-1:0] func;

wire [‘REG_W-1:0] rs, rd;

wire [‘SEL_W-1:0] com;

wire [‘IMM_W-1:0] imm;

wire rwe;

wire st_op, addi_op, ld_op, alu_op;

wire ldi_op, ldiu_op, addiu_op;

assign ddataout = rf_a;

assign iaddr = pc;

assign daddr = rf_b;

assign {opcode, rd, rs, func} = idatain;

assign imm = idatain[‘IMM_W-1:0];

// Decorder

assign st_op = (opcode == ‘OP_REG) & (func == ‘F_ST);

assign ld_op = (opcode == ‘OP_REG) & (func == ‘F_LD);

assign alu_op = (opcode == ‘OP_REG) & (func[4:3] == 2’b00);

assign ldi_op = (opcode == ‘OP_LDI);

assign ldiu_op = (opcode == ‘OP_LDIU);

assign addi_op = (opcode == ‘OP_ADDI);

assign addiu_op = (opcode == ‘OP_ADDIU);

assign we = st_op;

assign alu_b = (addi_op | ldi_op) ? {{8{imm[7]}},imm} :

(addiu_op | ldiu_op) ? {8’b0,imm} : rf_b;

assign com = (addi_op | addiu_op ) ? ‘ALU_ADD:

(ldi_op | ldiu_op ) ? ‘ALU_THB: func[‘SEL_W-1:0];

assign rf_c = ld_op ? ddatain : alu_y;

assign rwe = ld_op | alu_op | ldi_op | ldiu_op | addi_op | addiu_op ;

alu alu_1(.a(rf_a), .b(alu_b), .s(com), .y(alu_y));

rfile rfile_1(.clk(clk), .a(rf_a), .aadr(rd), .b(rf_b), .badr(rs),

.c(rf_c), .cadr(rd), .we(rwe));

always @(posedge clk or negedge rst_n)

begin

if(!rst_n) pc <= 0;

else

pc <= pc+1;

end

endmodule

アキュムレータマシンと同じく、POCOは，クロック、リセットおよび命令メモリ
に対するアドレス iaddr,データ入力 idatain、データメモリに対するアドレス daddr,
データ入力 ddatain,データ出力 ddataout、書き込み制御信号 weを入出力として持つ。

ddataoutはレジスタファイルの a出力が、daddrには b出力が接続され、iaddrには pc
が接続される。これらはもう決っているので最初に宣言する。

assign ddataout = rf_a;

assign iaddr = pc;

assign daddr = rf_b;

さて、命令メモリから読んで来た idatainを命令コード (opcode)、ディスティネー
ションレジスタ番号 (rd)、ソースレジスタ番号 (rs)、ファンクションフィールド (func)
に分離する。

assign {opcode, rd, rs, func} = idatain;

この書き方は、複数の信号線をくっつけて束にする連結演算子 ,を左辺に使ってい
て格好良い書き方であり、下の書き方をまとめたものとなる。

assign opcode = idatain[15:11];

assign rd = idatain[10:8];

assign rs = idatain[8:5];

assign func = idatain[4:0];

このまとめた書き方は、行数が少なく、読んで分かりやすい。しかし、ビット幅が
うまく合っていないと、誤った信号に分離されてしまっても、バグに気づき難い欠点
もあるので利用時は注意が必要である。1

イミーディエイト命令時は、下位 8ビットをイミーディエイト値として使うため、
これも分離しておく。

assign imm = idatain[‘IMM_W-1:0];

もちろん、immは fs,funcと物理的には同じなのだが、意味が違うため、違った信
号名にしておく。
次に opcodeで命令を判別する。これが以下の部分で、これをデコーダ (解読器)と

呼ぶ。デコードの結果、命令の種類毎に信号線を用意した。これで、表 5.1の信号を
簡単に発生できる。

// Decorder

assign st_op = (opcode == ‘OP_REG) & (func == ‘F_ST);

assign ld_op = (opcode == ‘OP_REG) & (func == ‘F_LD);

assign alu_op = (opcode == ‘OP_REG) & (func[4:3] == 2’b00);

1Verilog はシミュレーション時に左右のビット幅が合っていなくても実行されてしまい場合によっては
気づかない。しかし論理合成時には Warning が出るので、この時にちゃんと見ていれば気づく

assign ldi_op = (opcode == ‘OP_LDI);

assign ldiu_op = (opcode == ‘OP_LDIU);

assign addi_op = (opcode == ‘OP_ADDI);

assign addiu_op = (opcode == ‘OP_ADDIU);

デコードしておけば、例えば ST命令実行時にはデータメモリの書き込み信号を 1
にするという記述を以下のように簡単に記述できる。

assign we = st_op;

さて、次は ALU周辺のマルチプレクサを記述する。alu bは ALUの B入力、com
はコマンド入力である。A入力は直接レジスタファイルの出力が繋がるのでマルチプ
レクサを記述する必要はない。

assign alu_b = (addi_op | ldi_op) ? {{8{imm[7]}},imm} :

(addiu_op | ldiu_op) ? {8’b0,imm} : rf_b;

以前から使っているように、Verilogではマルチプレクサを ? : で表す。表 5.1の制
御信号との対応に注目されたい。3入力以上のマルチプレクサは?を 2回使って条件
を順番にチェックしている。
ここで、まず分かり難い記述は、ALUの B入力であろう。ここでは、ADDIと LDI

の実行時には符号拡張した値を選択する。
Verilogでは 8bitの immを 16bitに符号拡張する場合、

{{8{imm[7]}},imm}

と記述する。まず immの最上位ビットつまり 7bit目を切り出す (imm[7])。次にこ
れを 8個分並べる。これを Verilogでは

{8{imm[7]}

と書くことができる。これは

{imm[7], imm[7], imm[7], imm[7], imm[7], imm[7], imm[7], imm[7]}

と同じだが、よりすっきりした書き方である。この 8bit分を元の immとくっつけて
16bitにするので、これはすなわち符号拡張である。一方、ADDIU, LDIUは符号拡張
せず、8bitの 0をくっつければ良いので、以下のように書けば良い。

{8’b0,imm}

これ以外は、レジスタ同士の演算ということになるので、レジスタファイルから読
み出した rf bを ALUの B入力に入れてやる。
次にALUの制御入力には comには、まずイミーディエイトロード命令のLDI、LDIU

の場合には B 入力をそのまま素通しするため、THB(Through B) を入れる。ADDI,
ADDIUの場合には加算をするので、もちろん ADDを入れる。それ以外は、レジス
タ同士の演算と考えられる。この場合、アキュムレータマシン同様、命令コードの一
部を入れてやる。ここでは、functコードの下 3bitがこれに当たる。

assign com = (addi_op | addiu_op ) ? ‘ALU_ADD:

(ldi_op | ldiu_op) ? ‘ALU_THB: func[‘SEL_W-1:0];

この alu b、comを ALUに接続する。A入力はレジスタファイルの Aポート出力
をそのまま繋ぎ、出力には alu yという信号名を付けておく。

alu alu_1(.a(rf_a), .b(alu_b), .s(com), .y(alu_y));

次はレジスタファイル周辺の記述である。Aポートはディスティネーションレジス
タを取り出すので、アドレスに rdを繋ぎ、Bポートはソースレジスタを取り出すの
で、アドレスに rsを繋ぐ。それぞれの出力 rf aは直接 ALUに rf bはマルチプレクサ
に接続済みである。

rfile rfile_1(.clk(clk), .a(rf_a), .aadr(rd), .b(rf_b), .badr(rs),

.c(rf_c), .cadr(rd), .we(rwe));

それではレジスタファイルに書き込む側を考えよう。書き込むレジスタ番号は rdな
ので、これを Cポートのアドレスに繋ぐ。次に書き込むデータは、LD命令ではデー
タメモリ (ddatain)から、それ以外では ALUから来るので、これをマルチプレクサで
切り替える。

assign rf_c = ld_op ? ddatain : alu_y;

書き込み用の信号線 rweは、結果をレジスタに書き込む全ての命令で 1にする必要
があるので、これらの論理和として表現する。

assign rwe = ld_op | alu_op | ldi_op | ldiu_op | addi_op | addiu_op ;

最後の部分は pcの記述である。

always @(posedge clk or negedge rst_n)

begin

if(!rst_n) pc <= 0;

else pc <= pc+1;

end

ここではまだ分岐命令を導入していないので、単に毎クロックでカウントアップさ
れる。

テストベンチ

POCOのテストベンチは、アキュムレータマシンとさほど変わっていない。

/* test bench */

‘timescale 1ns/1ps

‘include "def.h"

module test;

parameter STEP = 10;

reg clk, rst_n;

wire [‘DATA_W-1:0] ddataout ;

wire [‘DATA_W-1:0] daddr, iaddr;

wire we;

reg [‘DATA_W-1:0] dmem [0:‘DEPTH-1];

reg [‘DATA_W-1:0] imem [0:‘DEPTH-1];

always #(STEP/2) begin

clk <= ˜clk;

end

poco poco_1(.clk(clk), .rst_n(rst_n), .idatain(imem[iaddr]),

.ddatain(dmem[daddr]), .iaddr(iaddr), .daddr(daddr),

.ddataout(ddataout), .we(we));

always @(posedge clk)

begin

if(we) dmem[daddr] <= ddataout;

end

initial begin

$dumpfile("poco.vcd");

$dumpvars(0,test);

$readmemh("dmem.dat", dmem);

$readmemb("imem.dat", imem);

clk <= ‘DISABLE;

rst_n <= ‘ENABLE_N;

#(STEP*1/4)

#STEP

rst_n <= ‘DISABLE_N;

#(STEP*100)

$finish;

end

always @(negedge clk) begin

$display("pc:%h idatain:%b ", poco_1.pc, poco_1.idatain);

$display("reg:%h %h %h %h %h %h %h %h",

poco_1.rfile_1.r0, poco_1.rfile_1.r1, poco_1.rfile_1.r2,

poco_1.rfile_1.r3, poco_1.rfile_1.r4, poco_1.rfile_1.r5,

poco_1.rfile_1.r6, poco_1.rfile_1.r7);

$display("dmem:%h %h %h %h", dmem[0], dmem[1], dmem[2], dmem[3]);

end

endmodule

アキュムレータマシンと同様に、

iverilog test_poco.v poco1.v alu.v rfile.v

vvp a.out

でシミュレーションの実行が可能である。
ここで、若干違っているのは、initial文の直後に

$dumpfile("poco.vcd");

$dumpvars(0,test);

が入っていることである。この vcd形式というのは、シミュレーション結果などを
取っておくための標準形式の一つで、この二つの文により、波形ビュア－が利用可能
になる。ここではフリーの波形ビュア－ gtkwaveを用いる。

gtkwave poco.vcd

で起動する。この利用法と結果表示例はwebに示すが、非常に使いやすいビュア－
で一通りのことならばマニュアルなしで操作できる。pcの様子、各レジスタの変化等
を観察してみよ。

演習 5.3

imem.dat, dmem.datを書き替え、A OR (A-B)の演算を行うプログラムをシミュレー
ションして結果を確認せよ。

演習 5.4

イミーディエイト論理演算命令

ORI rd,#X

を実装せよ。ここで、Xは 0拡張せよ。

演習 5.5

LDI, LDIUはレジスタの下位 8ビットに数を入れるのには便利だが、これだと上位 8
ビットは、全て 1か全て 0が常に入ることになる。これでは困るので上位 8ビット
に、指定した値 X を入れて、下位 8 ビットは 0にクリアするのが LDHI(Load High
Immediate)である。この命令の働きを図 5.8に示す。

LDHI r0, #5

00000101 00000000r0

0

図 5.8: LDHIの操作

LDHIは以下のフォーマットとする。

LDHI rd,#X rd <- X|0 01010 ddd XXXXXXXX

LDHIとADDIUの組み合わせで、16ビットの数をレジスタ中に作ることができる。
例えば、0x8090番地をアクセスしたければ、以下のように行えば良い。

LDHI r0, #0x80 01010 000 10000000

ADDIU r0, #0x90 01101 000 10010000

LD r1, (r0) 00000 001 000 01001

さて、この LDHI(Load High Immediate)命令を Verilog記述に付け足して実現せよ。

演習 5.6

LDHIでは下位 8bitを 0にするが、これは時に不便な場合がある。そこで、LDHIと
全く同じ操作を行うが、下位 8bitはレジスタの元の値を保持する命令 LDHI2を実装
せよ。

6.1 分岐命令

前の章で定義した命令はごく基本的なものだけで、これでは普通のプログラムを書
くには不足する。ここではより高度な命令に拡張していく。
アキュムレータマシンと違って POCOではレジスタ数が増えたため、アキュムレー

タの分岐命令はレジスタのどの中身を調べるか、を指定する必要がある。このため、
レジスタ番号のフィールドを付け加えて、以下のように分岐命令を拡張する。

BEZ rd, X if (rd==0) pc <- pc + X 10000 ddd XXXXXXXX

BNZ rd, X if (rd!=0) pc <- pc + X 10001 ddd XXXXXXXX

ここで、もう一つ拡張の必要がある。今まではメモリのアドレス空間が狭かったた
め、4bitで直接指定することができた。すなわち、今までの方法は、4bitで表した X
を直接 pcに入れてしまっていた。これを絶対指定:absolute(直接指定:direct)と呼ぶ。
絶対指定は分かりやすいが、アドレス空間が広い場合は指定のビット数が長くなりす
ぎて現実的ではない。
そこで、多くのマシンで使われている方法がプログラムカウンタ (PC)相対 (relative)

指定である。この方法では飛び先 Xを直接 pcに入れるのではなく、現在の pcの値に
加えた値を飛び先とする。もちろん Xは符号拡張されるので、マイナス方向にも飛
ぶことができる。ここで注意したいのは、多くの CPUでは、分岐の起点は分岐命令
ではなく、その次の命令になっている点である。跳び越す命令数は分岐命令の次の命
令を起点にして数える。すなわち、Xを 0とすると、分岐しないで次の命令がそのま
ま実行され、-1とすると自分自身に分岐する。
この方法では、分岐命令の飛べるアドレスの範囲は、-方向には 128、+方向には 127

に限定される。これでは困るようだが、一般的にはさほど不便ではない。これは、プ
ログラムの動き方が極めて局所性 (Locality)が強いことから、飛び先は概ね近場に限

られているからである。とはいえ、やはり、これだけでは困るので、後でもっと別の
タイプの分岐命令を導入する。

2番地の中身と 3番地の中身を掛け算して 0番地に格納するプログラムは以下のよ
うになる。

LDIU r0, #2

LD r1,(r0)

LDIU r0, #3

LD r2,(r0)

LDIU r3, #0

ADD r3,r1

ADDI r2, #-1

BNZ r2,-3

LDIU r0, #0

ST r3,(r0)

BEZ r2,-1

6.1.1 アセンブラ shapa

POCO でプログラムを書く場合、人手で機械語に直していては大変である。そこ
で、アセンブラというプログラムを使って、アセンブリ表記を機械語に変換してやる。
POCO用のアセンブラ shapaを web上に置く。shapaは rubyで書かれており、web上
のプログラムは通常の Linuxではそのまま動作する。POCO.rb中に命令の定義が書か
れており、これを書き換えれば、様々な命令に対応可能である。掛け算のプログラム
は以下のようにラベルを付けて書くことができる。

　　　 LDIU r0, #2
　　　 LD r1,(r0)
　　　 LDIU r0, #3
　　　 LD r2,(r0)
　　　 LDIU r3, #0
loop: ADD r3,r1

ADDI r2, #-1

BNZ r2,loop

LDIU r0, #0

ST r3,(r0)

end: BEZ r2,end

これを、例えば mult.prgというファイルに入れておく場合、同じディレクトリ中に
shapaと poco.rbを置いておく。

そして、

./shapa kadai.asm -o imem.dat

とすると、imem.dat中に機械語が生成される。poco1.v, test poco.v, alu.v, rfile.v, def.h,
dmem.datをこれと同じディレクトリに置いておき、

iverilog test_poco.v poco1.v alu.v rfile.v

vvp a.out

とすればシミュレーションの実行が可能である。

6.1.2 比較と分岐

BNZ, BEZだけでは数の大小の比較ができない。そこで、POCOでは、以下の命令
を設ける。

BPL rd, X if (rd>=0) pc <- pc + X 10010 ddd XXXXXXXX

BMI rd, X if (rd<0) pc <- pc + X 10011 ddd XXXXXXXX

これらの命令はレジスタの内容の正負で分岐するかどうかが決る。これらの命令
は、レジスタの符号ビットだけを見れば良いので、実装が楽だという利点がある。一
方で、大小比較を行うには、引き算を行った後に、これらの命令を使う必要が出るた
め、レジスタの内容を破壊してしまう欠点がある。
以下に、0番地から並んでいる 8つのデータの中から最大のものを選ぶプログラム

を示す。r3に最大値が格納されるが、一度引いて比較した値を再び足し戻す処理が必
要となる。

LDIU r0, #0x00

LDIU r3, #0x00

LDIU r4, #8

loop: LD r2,(r0)

SUB r2, r3

BMI r2, skip

ADD r3, r2

skip: ADDI r0, #1

ADDI r4, #-1

BNZ r4, loop

end: BEZ r4, end

6.1.3 フラグを使う方法

POCOでは採用しないが、レジスタを破壊する欠点を嫌ってフラグ (Flag)を使う方
法を採用するプロセッサもある。この場合、フラグとは演算の結果の性質を格納する
小規模の専用レジスタを指す。良く使われるフラグは、以下の例がある。

• Zero Flag: 演算の結果が 0になるとセットされる

• Carry Flag: 演算の結果、桁あふれがおきるとセットされる

• Sign Flag: 演算の結果、マイナスになるとセットされる

分岐命令は、このフラグがセットされていれば分岐するという形式を取る。例えば、
Branch Zeroならば、Zero Flagがセットされていれば分岐する、という命令に相当す
る。このような分岐命令で、POCOの命令セットと同様のプログラムが書けるのは容
易に理解できるだろう。
フラグは演算の結果セットやリセットされるが、フラグだけを変化させる専用の命

令を設ける場合がある。これを比較命令 (Compare)と呼ぶ。例えば、Compare r0,r1と
は、r0と r1を比較し、等しければ Zero flagが、r0r1が成立すれば Sign flagがセット
される。
比較命令を使えば、レジスタを破壊せずに、フラグのみをセットして条件分岐が可

能である。このようにフラグを使う方法は、便利で効率が高いが、一面、命令コード
の入れ替えが難しい、割り込み発生時にフラグを保存する必要がある等の問題がある。
このため、POCOではフラグを持っていない。フラグについての詳しい解説は webを
参照されたい。

6.1.4 ジャンプ命令

Branchつまり分岐命令は、判断を伴うものを呼び、単純にプログラムカウンタの値
を変更する命令のことをジャンプ命令と呼ぶ。ジャンプ命令は分岐命令と同様、相対
指定を行うが、レジスタ指定の必要がない分遠くに飛ぶことができる。

JMP X pc <- pc + X 10100 XXXXXXXXXXX

POCOの場合、飛び先の番地を指定するビットを 11ビット確保できる。しかし、こ
の方法でも 11ビットの範囲に入らなければ、飛べないことになり、ときに不便であ
る。そこで、以下の最終手段を用意しておく。

JR rd pc <- rd 00000 ddd --- 01010

この命令は、レジスタ間接ジャンプと呼び、レジスタの値をそのままプログラムカ
ウンタにセットする。レジスタは 16bitなので、アドレス空間のどこにでも飛ぶこと

ができる。一種の絶対ジャンプだが、レジスタの値はプログラムにより制御できるた
め、結果によってジャンプ先を変えたり、後に述べるようにサブルーチンコールから
のリターン命令として利用可能である。

演習 6.1

アキュムレータマシンでやった演習と同じく、1番地にXが入っている時、1+2+...+(X-
1)+Xを計算するプログラムを POCOのアセンブラで書け。

演習 6.2

メモリの 0番地から並んだ 8つの数のうち、0の個数を数えるプログラムを書け。

サブルーチンコール

JALと JALR

ジャンプ命令は一度飛んでいったら、戻って来ない。しかし、プログラムは一定の処
理をサブルーチンとして定義し、これを様々な場所から呼び出して利用する命令が必
要である。これをサブルーチンコールと呼ぶ。サブルーチンコールは、図 6.1に示す
ように、呼び出して、戻ってくる機能が必要になり、このため、プログラムカウンタ
をどこかに保存しておく必要がある。保存場所としては特定のレジスタを利用するの
が一般的である。多くの場合、最大番号のレジスタを用いるので、POCOでは r7を
保存場所として定める。

JAL subr

JR r7

JAL subr1

JR r7

JAL subr2

JR r7

a) Subroutine Call b) Nest of Subroutine Call

図 6.1: サブルーチンコール

JAL X r7 <- pc, pc<- pc + X 10101 XXXXXXXXXXX

JALR rd r7 <- pc, pc <- rd 00000 ddd --- 11000

飛び先は JMP命令と同様に 11ビットを指定することができる。11ビットではアド
レス空間全体をカバーできないため、JRと同様にレジスタ間接指定を使えるように
している。これが JALRである。この命令では飛び先をレジスタの内容で指定するこ
とができる。PCの格納場所としては、同様に r7である。
戻り番地は r7に格納されるため、リターン命令は JR r7となる。下のプログラムは

0番地のデータの 2乗を計算するものである。掛け算のサブルーチンMultをコール
している。このサブルーチンは r1 X r2の結果が r3に格納される。r1は破壊されない
が、r2は破壊されてしまう点に注意。なお、ここではプログラムを見やすくするため
にラベルを使っている。通常、ラベルはアセンブラにより、自動的に相対アドレスに
変換される。

LDIU r0, #0x00

LD r1, (r0)

MV r2, r1

JAL mult

ST r3, (r0)

end: JMP end

//Subroutine Mult

mult: LDIU r3, #0x00

loop: ADD r3, r1

ADDI r2, #-1

BNZ r2, loop

JR r7

スタック

サブルーチンは、プログラムの各所から呼び出せるという利点がある。しかし、ここ
で問題がある。呼び出すプログラム（メインルーチン）とサブルーチンで利用するレ
ジスタが重なってしまうと、サブルーチンを呼び出してリターンしてきた時に、サブ
ルーチン内でレジスタが破壊されて、プログラム実行上の情報が失われる可能性があ
る。そこで、サブルーチンは、呼び出された時に、利用するレジスタをどこかにしまっ
て置く必要がある。これを実現するのがスタック (Stack)と呼ぶ。スタックは棚であ
り、先に積んだものを後で取り出し、後で積んだものを先で取り出すことから Last In
First Out (LIFO)または First In Last Out (FILO)とも呼ぶ。スタックにデータを積む操
作を push、取り出す操作を popと呼ぶ。RISCでは通常スタックは、ソフトウェアで
実現する。このためには、スタックとして利用するメモリ領域を指すレジスタが必要

になる。これをスタックポインタと呼ぶ。図 6.2の例では r6がスタックポインタであ
り、プログラムをスタートする場合これをスタックとして使う領域の最後に設定して
おく必要がある。

....

r6

....

r6

r3

....

r6
r4

r3
-2

-2

a) Stack push

....

r6
r4

r3+2

....

r6

r3

+2

....

r6

b) Stack pop

図 6.2: スタックプッシュとポップ

r3,r4を順にスタックに pushする操作は以下のようになる。

ADDI r6,#-1

ST r3,(r6)

ADDI r6,#-1

ST r4,(r6)

すなわち、あらかじめスタックポインタを減らして領域を確保し、そこにデータを
格納する。逆に格納したデータを元に戻す pop操作は以下のようになる。

LD r4,(r6)

ADDI r6,#1

LD r3,(r6)

ADDI r6,#1

スタック操作の優れた点は、サブルーチンの入れ子（ネスト）に対応可能な点であ
る。サブルーチンの中でサブルーチンを呼んでも、呼ばれたサブルーチンで利用する

レジスタが順にスタックに積まれ、逆順で戻るため、スタック領域を使いつくすまで
いくらでもサブルーチンを呼ぶことが可能である。実際、再帰呼び出し (recursive call)
を行う場合、多くのデータがスタックに積まれることになる。ここで、戻り番地は r7
に格納されていることから、忘れずに r7をスタックに pushし、リターン操作である
JR r7を実行する前に popして元に戻しておくことが必要である。

6.2 BNZ/BEZ命令の実行
分岐命令を実装するために、フロントエンド部分に改造を加え、命令の下位 8ビッ

トと PC+1を加算して飛び先番地を計算する回路を付け加える。さらに、マルチプレ
クサを取り付け、分岐が成立した場合に、飛び先番地を PCに入れてやる。

BEZ命令は、aポートから読み出された rdが 0かどうかチェックする。条件が成立、
つまりこの場合 0だった時に pcsel=1として、命令の下位 8ビットを符号拡張して加
算器に入れてやる。条件が成立するかどうかは制御回路で簡単に判断できる。図 6.3
にこの様子を示す。

com

reg_file

rwe

00 01

rf_csel

aadr

0 1

1
1
0pcsel

=1

alu_bsel

=0?
zero

we

0

図 6.3: BEZ命令実行のデータパス

JMP命令は、符号拡張が 11ビット範囲であることを除けば分岐命令とほぼ同じ回
路で実現できる。JR命令は、飛び先番地がレジスタファイルの a出力から表われる
ので、これを直接 pcに入れてやる。この辺は簡単だが、JAL命令は面倒である。こ

の命令は、ジャンプする部分は JMP命令と全く同じ回路で良いが、その他に pc+1を
r7に書き込まなければならない。今まで、レジスタファイルの aポートアドレスと c
ポートアドレスはいつも同じで、rdを示す命令の 10-8bitを入れていたが、JALに限っ
ては cポートアドレスに 7を入れてやる。さらに、レジスタファイルへの書き込み用
のマルチプレクサを拡張して、pc+1を入れてやる必要がある。
これらの拡張を全て行った構成を図 6.4に示す。

図 6.4: JALが実行可能なデータパス

さまざまな線が増えたため、かなりごちゃごちゃした印象を与えるが、これで拡張
は終わりである。あまり細部にこだわらず、今度は Verilogの記述を見てみよう。

6.2.1 Verilogの変更

図 6.4に示す分岐命令用のハードウェアの変更はほとんどそのまま Verilog記述に
反映することができる。ここでは変更箇所のみ示す。まず wire文で、各命令に対応
する信号を宣言してから、デコーダに以下の部分を付け加える。これで、それぞれの
命令に対応した信号が利用可能となる。

assign jr_op = (opcode == ‘OP_REG) & (func == ‘F_JR);

assign bez_op = (opcode == ‘OP_BEZ);

assign bnz_op = (opcode == ‘OP_BNZ);

assign bpl_op = (opcode == ‘OP_BPL);

assign bmi_op = (opcode == ‘OP_BMI);

assign jmp_op = (opcode == ‘OP_JMP);

assign jal_op = (opcode == ‘OP_JAL);

まず、JAL命令用にレジスタファイルの cポートの入力とアドレスの拡張が必要と
なる。そこで、以下のようにする。

assign rf_c = ld_op ? ddatain : jal_op ? pc+1 : alu_y;

assign rwe = ld_op | alu_op | ldi_op | ldiu_op | addi_op |

addiu_op | ldhi_op | jal_op ;

assign cadr = jal_op ? 3’b111 : rd;

rfile rfile_1(.clk(clk), .a(rf_a), .aadr(rd), .b(rf_b), .badr(rs),

.c(rf_c), .cadr(cadr), .we(rwe));

すなわち、JAL命令の際は、cポートのアドレス cadrが 7となり、入力には pc+1
が設定される。次に、pcについては以下のように修正する。分岐命令はそれぞれの
条件をチェックする。BMIや BPLは読み出してきたレジスタの符号ビット (MSB)を
チェックすれば良い。

always @(posedge clk or negedge rst_n)

begin

if(!rst_n) pc <= 0;

else if ((bez_op & rf_a == 16’b0 ) | (bnz_op & rf_a != 16’b0) |

(bpl_op & ˜rf_a[15]) | (bmi_op & rf_a[15]))

pc <= pc +{{8{imm[7]}},imm}+1 ;

else if (jmp_op | jal_op)

pc <= pc + {{5{idatain[10]}},idatain[10:0]}+1;

else if(jr_op)

pc <= rf_a;

else

pc <= pc+1;

end

ここで、JMPと JALでは irの下位 11ビットを符号拡張している点に注意された
い。また、JRでは直接レジスタファイルの aポートから読み出されたレジスタが設
定されている。always文中であるため、if文が使えるため、条件は見やすいと思う。



6.2. BNZ/BEZ命令の実行 101

POCOの基本命令

今回で、POCOの命令上の拡張は打ち止めである。最後にまとめておく。

NOP 00000 --- --- 00000

MV rd,rs rd <- rs 00000 ddd sss 00001

AND rd,rs rd <- rd AND rs 00000 ddd sss 00010

OR rd,rs rd <- rd OR rs 00000 ddd sss 00011

SL rd rd <- rd<<1 00000 ddd --- 00100

SR rd rd <- rd>>1 00000 ddd --- 00101

ADD rd,rs rd <- rd + rs 00000 ddd sss 00110

SUB rd,rs rd <- rd - rs 00000 ddd sss 00111

ST rs, (ra) rs -> (ra) 00000 sss aaa 01000

LD rd, (ra) 　　 rd <- (ra) 00000 ddd aaa 01001

LDI rd,#X rd <- X (符号拡張) 01000 ddd XXXXXXXX

LDIU rd,#X rd <- X (符号拡張なし) 01001 ddd XXXXXXXX

ADDI rd,#X rd <- rd + X (符号拡張) 01100 ddd XXXXXXXX

ADDIU rd,#X rd <- rd + X (符号拡張なし) 01101 ddd XXXXXXXX

LDHI rd,#X rd <- X|0 01010 ddd XXXXXXXX

BEZ rd, X if (rd==0) pc <- pc + X 10000 ddd XXXXXXXX

BNZ rd, X if (rd!=0) pc <- pc + X 10001 ddd XXXXXXXX

BPL rd, X if (rd>=0) pc <- pc + X 10010 ddd XXXXXXXX

BMI rd, X if (rd<0) pc <- pc + X 10011 ddd XXXXXXXX

JMP X pc <- PC + X 10100 XXXXXXXXXXX

JAL X r7 <- pc, pc<- pc + X 10101 XXXXXXXXXXX

JR rd pc <- rd 00000 ddd --- 01010

JALR rd r7 <- pc, pc <- rd 00000 ddd --- 11000

演習 6.3

例題の乗算ルーティンを使って、0番地に格納されているデータの階乗を計算するプ
ログラムを書け。(まずは値を小さくしてテストすることをお勧めする。アセンブラ
shapaを利用せよ。)

演習 6.4

JALRを poco1.vに付け加え、その機能をテストせよ。

第 7

CPUの性能とコストの評価

7.1 CPUの性能評価式
7.1.1 CPIとクロック周期

CPUの性能は当然のことながら、プログラムの実行時間により評価する。OS等の
ロスを除いた純粋にユーザプログラムの実行時間は以下の式により表される。

CPUtime =プログラムの CPU クロックサイクル数×クロック周期

ここで、実行された命令数をカウントし、一命令あたりの平均クロック数をCPI: Clocks
Per Instructionとすると、CPU実行時間は、以下の式で表わすことができる。

CPUTime =命令数× CPI×クロック周期

CPIは、CPUに固有のものでなく、実行するプログラムによって変化することに注
意されたい。つまり、実行時間の長い命令をたくさん用いるプログラムでは CPIは長
くなる。CPIがどのようになるかは、一つのプログラム中での命令の利用頻度に依存
する。

POCOの場合、全ての命令は 1クロックで終了するので、CPIは単純に 1となるが、
通常の CPUは命令毎に実行クロック数が異なるので、どのようなプログラムで動か
すかによって CPIは変わってくる。
クロック周期は、マイクロアーキテクチャの動作速度で決る。これは、実際にゲー

トレベルで設計して、遅延時間を見積ることで評価することができる。

例題 7.1： CPU の性能の尺度に MIPS(Million Instructions Per Second) というのが
ある。これは、１秒間に何百万回命令を実行可能か、という数値である。POCO が
100MHzで動くとすると何MIPSになるか？

答： MIPS値は実行周波数のメガ (Mega:10の 6乗) Herz表示を平均 CPIで割れば
良いので、100/1 = 100MIPSとなる。ただし、このMIPSという尺度は、一定時間で
実行できる命令数のみを考え、個々の命令の能力が考慮されていないため、命令セッ
トが異なるマシン間の比較には目安程度にしか使えない。
では、POCOはいったいどの程度の周波数で動くのだろうか？周波数は回路の遅延

時間の逆数になる。6章の POCOのマイクロアーキテクチャでこれを検討してみよう。

図 7.1: POCOのクリティカルパス

図 7.1の太線が、POCOの中で、もっとも遅延時間の長い信号伝搬経路 (パス)であ
る。このようなパスをクリティカルパスと呼ぶ。
まず、命令を命令メモリから読み出すために時間が掛かる。これを命令メモリのア

クセスタイム tinst と呼ぶことにする。次は読んで来た命令のレジスタ番号からレジス
タファイルを読み出すのに必要な時間が加わる。これがレジスタファイルのアクセス
タイム tr f ile である。ここから経路が二つに分かれる。
一つは、データメモリを読み出す経路である。これは、データメモリのアクセス

タイム tdata+2入力マルチプレクサの遅延 t2mux +レジスタファイルへの書き込み時間
tsetup となる。
もう一つは、ALUで計算を行う経路である。これは、3入力マルチプレクサの遅延

t3mux+ALUでの遅延 talu+ 2入力マルチプレクサの遅延 t2mux となる。これの大きい方
を取れば良い。

例題 7.2： tinst = tdata= 2nsec、tr f ile = 1nsec、t2mux=0.3nsec、t3mux=0.5nsec、tsetup =

0.2nsec、talu = 3nsecと仮定すると、POCOの動作周波数はどのようになるか？ちなみ

に、nsec(ナノ秒)は 10のマイナス 9乗である。

答： データを読み出す経路は、2 + 1 + 2 + 0.3 + 0.2 =4.5nsec、ALUで計算を行う
経路は、2 + 1 + 0.5 + 3 + 0.3 + 0.2 = 7.0nsecである。このため、ALUで計算を行う
経路がクリティカルパスになる。動作周波数は、1/7 GHz = 0.143 GHz = 143MHzで
ある。ちなみに、G(ギカ)は 10の 9乗である。
上記の演習は、適当に各部の遅延を見積った概略のものである。本当に遅延を評価

するには、実装するチップを想定して論理合成および圧縮を行う必要がある。4章で
紹介した論理合成、圧縮ツールを使って POCOを合成、圧縮してみよう。合成、圧縮
ツールには STA(Static Timing Analysis)機能が付いているので、クリティカルパスを
レポートしてくれる。

POCOのクリティカルパスが本質的に長いのは、全命令を 1サイクルで終わらせて
しまうためである。そこで、マルチサイクル化する方法とその Verilog記述について
web上に示す。実際はマルチサイクル化しても性能の向上には中々結び付かないが、
メモリその他の資源を共有することができるため、コストを削減することは可能であ
る。最適化には論理合成圧縮ツールが必須であるので、この辺に興味をお持ちの読者
は是非 webを参照されたい。

7.1.2 MIPS、MFLOPS、MOPS

POCOの CPIは 1であり、動作周波数をMHzで表したものが、一秒間で何百万回
命令を実行可能か、すなわち、MIPS値と同じになる。先の例では 143MIPSというこ
とになる。MIPS値は、あるプログラムを実行するのに何命令必要か、という命令数
についてのファクタが入らないため、評価基準として、あまり正しくない。しかし、
同じシリーズの命令同士の比較には使えるため、Intelの 86系の命令セットが主流の
デスクトップやラップトップコンピュータの評価には利用される場合がある。
また、浮動小数点演算能力を持った科学技術用のマシンの性能尺度としてMFLOPS(Million

Floating Point OPerations Per Second)がある。これは１秒間に何百万回浮動小数点演算
を行うことができるかを示す。決まった数値計算アルゴリズムを実行するのに要する
演算数（命令実行数ではない）は、マシンの命令セットアーキテクチャに依らず一定で
あるので、この数値はMIPSよりは信頼できるのではないか、というのがMFLOPSの
根拠である。しかし、平方根や指数関数など関数演算が入るとこの命令を持つマシン
と持たないマシンが出てくるため、話が面倒になる。この問題を解決するため、関数
演算を複雑さによって、浮動小数演算の数命令分としてカウントする正規化MFLOPS
という方法も用いられる。このようにMFLOPSという尺度は、本来はMIPS値より
も根拠があるのだが、スーパコンピュータなどで良く用いられるピークMFLOPSは、
そのコンピュータ出すことのできる瞬間最大性能 (逆に考えるとどのような瞬間もこ
れ以上の性能は出すことができない数値)で実効性能ではないので、性能尺度として

は全く信用してはならない。
さらに、同様の考え方として、信号処理用の専用プロセッサであるDSP(Digital Signal

Processor)などでは、1秒間に何百万回、所定の演算を実行できるかという性能尺度
として、MOPS(Million Operations Per Second)を用いる場合がある。

7.1.3 性能評価手法

POCOの場合、まだキャッシュも装備しておらず、主記憶もシミュレーションモデ
ルなので、評価するのはある意味で簡単だが、一般的なコンピュータシステムの性能
は、CPUだけでなく、搭載しているメモリ、キャッシュ、あるいは OS、コンパイラ
などのソフトウェアの影響も受ける。ここで、CPU性能をきちんと評価する方法を解
説しておく。

評価用のプログラム

CPUの性能は基本的にプログラムを実行して、その時間を測ることで評価する。性
能の測定に使われるプログラムは以下のようなものがある。

• 実プログラムに基づくベンチマーク集: 実際のプログラムと、その入力をセッ
トにしたベンチマーク集。一般的なWS/PCの評価には SPECベンチマークが
用いられる。これには非数値系の SPECint と数値系の SPECfpに分けられる。
SPECintは、主としてC言語で記述されており、GCC, Latex,表計算プログラム、
CADの最適化問題等から成り、SPECfpは FORTRANが中心で、SPICEなどの
科学技術計算から成っている。一定の金額を払えば誰でも利用可能である。他
にスーパコンピュータ評価用の Perfect Club、マルチプロセッサ用の SPLASH-II
などがある。

• 簡単なトイプログラム: ソートや、エラトステネスのふるい、8-queenなど。ア
センブラプログラムで記述できるし、アルゴリズムは誰でも知っているので、
ある程度再現性もある。POCOを今評価するならこの方法しか手がないが、一
般的にはこの程度の簡単なプログラムでは、特にメモリシステムの挙動がから
む正確な評価を取ることはできない。

• プログラムカーネル: 実プログラムに基づく評価はリアルだが、内部処理が何
をやっているのか見にくい。このため、数値計算用のマシンの評価では、様々
なプログラムから繰り返し部分のみを抜き出したカーネルが使われる場合も多
い。LLL(Lourence Livermore Loops)、Linpackが有名。数値計算のプロはこの
ループの番号を聞いただけで、その内部の処理の性質を理解することができる。
スーパーコンピュータの Top500を決めるのは Linpackを用いている。また組み



7.1. CPUの性能評価式 107

込み CPUの評価用の EEMBCやMiBenchなどのプログラム集もカーネルに分
類される。

• 評価専用プログラム: ベンチマークやカーネルは、いくつものプログラムを動か
さねばならず、しかもプログラムによって結果の順位が逆転したりして、どち
らが高速が判断することが難しい。このため、様々な振る舞いを一つのプログ
ラムに押し込んだ評価専用プログラムを作り、これを一つ動かせば評価を取れ
るようにしたもの。数値処理用のWhetstone、非数値用の Dhrystoneが有名で、
かつては評価に頻繁に利用された。しかし、これらのプログラムはコンパイラ
の最適化を効かなくするために、実際のプログラムではほとんど考えられない
ような振る舞いをする上、これらの評価用プログラムに特化した最適化を行う
コンパイラまで現れた。このため、Hennessy&Pattersonのテキスト [2]で厳し
くこの欠点が指摘された結果、現在はほとんど使われなくなった。

7.1.4 評価結果の整理と報告

評価専用プログラムが使われなくなった現在、評価を行うためには、複数のプログ
ラムを稼働させるのが普通である。この場合は、結果が複数得られることになるが、
比較を行う場合は、どちらかのマシンに正規化して、実行したプログラムの相乗平均
を取って比較するのが一般的である。これは、ただの相加平均だと基準に取るマシン
を変えると比率が代わってしまうことを避けるためである。しかし、相乗平均は非線
形性を持つので、出てきた結果がそのまま性能比となるわけではなく、この点には注
意しなければならない。
また、評価を取ってこれを報告する場合は、ちょうど自然科学実験のレポートで、

実験器具や方法を細かく記述するのと同様に、以下の点を細かく報告しなければなら
ない。この辺は計算機アーキテクチャの世界といえどもきちんとしなければならない。

• ハードウェア:

– 動作周波数

– キャッシュ容量

– メモリ容量

– Disk容量

• ソフトウェア:

– OSの種類、バージョン

– コンパイラの種類、オプション

演習 7.1 例題と同じ条件だが、メモリが遅く 4nsecであった場合、POCOの動作周
波数はどうなるか？また、MIPS値はどうなるか？

7.2 コストの評価

CPUチップとしてのコストは基本的にはゲート数（面積）により決る。次に価格
は、ゲート数あるいは面積によって支配される。一般論では、半導体のコストは、面
積の 4乗に比例する。これは、面積が大きくなればなるほど、一つの wafer（半導体
を作成するための板）から取れる die(チップに格納される半導体 1個分の切片)の数
が減ることと、埃等による欠損から良品が取れる割合（歩留りという）が悪化するた
めである。
とはいえ、新しい半導体チップは開発費およびマスク代が占める割合が大きいため、

チップのコストを直接評価するのは、きわめて難しい。論理合成ツールには合成後、
ゲート数 (FPGAの場合は Slice数などの内部基本論理素子数)をレポートしてくれる
ので、これを基準にコストを見積る。

7.3 消費電力

次に、最近特に重要になってきた可搬型のシステム用には、消費電力も重要なファ
クタである。一般に性能を上げるためには多くのゲートを高速に動作させる必要があ
るため、消費電力が大きくなってしまう。CMOSデバイスにおいて消費電力は、

動的な消費電力＋漏れ電流による消費電力

に分けられる。
前者は、電圧の 2乗に比例し、CMOSのゲートが ON/OFFする周波数に比例する。

すなわち、ゲート数が多く、クロック周波数が大きければその分電力が大きくなる。
しかし、全てのゲートが動作しているわけではないので、論理シミュレーションを行っ
てどの程度のゲートが動作するかを測定する必要がある。最近は、CPUの負荷が低
い場合は電圧とクロック周波数を落とし、電力を節約し、高い性能が必要な時だけ電
圧とクロック周波数を上げる DVFS(Dynamic Voltage and Frequency Scaling)という手
法が普及している。
後者の漏れ電流は、90nm以降のプロセスの微細化に伴ってその割合が大きくなっ

てきた。動作をさせなくても電力を消費してしまう点でバッテリー駆動の製品では深
刻な問題である。これを防ぐためには必要に応じて電力供給を切断する Power Gating
や、遅いが漏れ電流の小さいゲートと速いが漏れ電流の大きいゲートを組みあわせる
設計技術が利用される。

消費電力の削減は最近の CPUにとって大きな課題であり、低電力用のアーキテク
チャの研究開発が盛んである。

第 8

入出力：I/O

8.1 I/Oとは
今までこのテキストはCPUのみのシステムを扱っていた。しかし、実際のコンピュー

タシステムは、I/Oすなわち、入出力なしには意味のある作業を行うことができない。
簡単なコンピュータシステムでは、I/Oは図 8.1a)に示すように、CPU-メモリを結

ぶバス上に設ける。高性能のシステムでは図 8.1b)のように、プロセッサ-メモリ間の
バスの性能を阻害しないためと、標準化されたインタフェースを用いるために、標準
I/Oバスをブリッジを介して接続する方法もある。パーソナルコンピュータにおける
代表的な I/Oバスは PCIバスで、様々な機能の I/Oボードを接続することができる。
代表的な I/Oは、キーボード、ビットマップディスプレイ、ディスク、モデムインタ
フェース（RS232C）、Ethernet、A/D変換器等を接続するパラレルインタフェースな
どがある。ここでは、PICOにごく簡単な I/Oモデルを接続することで、I/O接続の基
本を紹介する。実際は、I/Oは接続する機器によって性質が異なり、構造、動作も奥
が深い。

I/O

I/O Ethernet

I/O
A/D Converter

Memory

CPU

Bus

I/O

I/O Ethernet

I/O
A/D Converter

Memory

CPU

I/O Bus

Bus
Bridge

System
Bus

a) b)

図 8.1: I/Oの構成例

8.1.1 I/Oモジュールの接続法

I/Oの接続法は、大きく以下の２つに分けることができる。

• メモリマップト I/O:メモリと同一アドレス空間で I/Oをアクセスする。このた
め、I/Oアクセスのための命令を専用に設ける必要はない。

• プログラムド I/O(I/Oマップド I/O): I/O独自のバスあるいは I/Oアドレス空間
を持ち、専用の I/O命令 (IN命令、OUT命令)の実行によって I/Oアクセスを
行う。I/Oとメモリの空間が分かれていることから、セパレート I/Oと呼ばれる
場合もある。

前者の方法は簡単だが、広大なメモリの空間に比較的小規模な I/Oをマップするため、
メモリ空間に虫食いができる問題がある。これに対し後者の方法は、専用の命令、バ
ス等を必要とする。今回は付加命令の必要のない前者の方法を用いる。

8.1.2 I/Oモジュールのモデル

簡単な I/Oモジュールのモデルを示す。I/Oモジュールは、多くは以下のレジスタ
を持つ。

• コントロール（コマンド）レジスタ: 多くの I/Oは、切り替え機能やパラメータ
を持っており、プログラム可能になっている。このプログラム用に、設定デー
タを書き込むためのレジスタ。通常は、書き込み専用である。

• ステータスレジスタ：その I/Oの状態をCPUが知るためのレジスタ。例えばデー
タが到着したかどうか、などは、このレジスタをアクセスすることにより知る
ことができる。データが到着したか、データの転送が終了するかなどを示す小
規模なレジスタをフラグと呼ぶ。これは分岐命令の所で軽く紹介したフラグと
同じ意味である。読み出し専用。

• データレジスタ: 実際にその I/Oとの間で交換するデータを受け渡すレジスタ。
I/Oの種類によってビット幅も異なる。通常は双方向。

これらのレジスタは、異なったアドレス空間に割り付けられ、CPUからはアドレス
によって識別される。I/Oモジュールによっては、コマンドレジスタとステータスレ
ジスタは同一番地を共有し、書くとコマンド、読むとステータスになることもある。
図 8.2に、古典的なシリアルインタフェース（モデム用:RS232C)のコントローラのコ
マンドとステータスの例 (Intel社 8251の一部)を示す。モデムは音声信号にデータを
変換し、電話線を用いてデータを送る装置であり、このためには、データを 1bitずつ
転送するシリアルインタフェースが必要である。この規格が RS232Cであり、図中に
示すタイミング図のように、1bit分のスタートビット (Lレベル)の後、データ（この

場合はキャラクタ：最大 8bit)を順に転送し、最後にストップビット (Hレベル)を一
定 bit分転送する。
このコントローラでは、コマンドによって、ストップビットの長さ (ST1,ST0)、転

送データ (キャラクタ)の長さ（L1,L0)、パリティの有無、種類 (P1,P0)、転送周波数
(ボーレート)(B1,B0)の設定が可能である。一方、同じ番地を読み出すと、ステータス
レジスタを読むことができ、これにより、データ到着 (RxRDY)、送信終了 (TxRDY)、
エラー (Err2-0)を知ることができる。

ST0ST1 P1 P0 L1 L0 B1 B0

Command

Status

ST1,ST0: Stop bit
0      0       Disable
0      1       1bit
1      0       1.5 bit
1      1        2bit

P1,P0: Parity
X   0    No parity
0    1   Odd parity
1    1   Even parity

L1, L0:  Character length
0     0     5bit
0     1     6bit
1     0     7bit
1     1     8bit

B1, B0 Bout late
0    1     x1 clock
1    0     x16 clock
1    1     x64 clock

HS1 Err1 Err0Err2 RxRDYTxRDY

HS1 (DSR): Handshake line
Err2−Err0: Transmission Error
RxRDY: Receive Ready
TxRDY: Send Ready

......

Start 
bit

Character (Data: 5−8bit) Stop
bit

Serial data transmission

D0 D1 D2

MSB LSB

MSB LSB

図 8.2: シリアル I/Oの例

この I/Oモジュールのプログラムは、以下のようにして行う。

• データの受信

– ステータスレジスタを読み込むことにより、データの到着を知る。

– データレジスタからデータを読み込む。これと共に次のデータが到着可能
となる。これを繰り返す。

• データの送信

– ステータスレジスタを読み込むことにより、バッファが空で次のデータが
送信可能かどうかを確認する。

– データレジスタにデータを書き込む。これによりデータの送信が起動さ
れる。

8.1.3 演習用のディスプレイ

さて、次は今回演習に用いる I/Oモデルを紹介する。これは極めて単純なものと
した。

• 0x8000番地に ASCIIコードのデータを書き込み、一定の時間が立つとディスプ
レイに表示される。

• 表示されるまでの間、次のデータの書き込みはできない。これを示すステータ
スレジスタ (フラグ) がやはり 0x8000番地に割り当てられている。すなわち、
この I/Oは、データとステータスが同じ番地に割り当てられている。ここでは、
0x8000番地の一番下のビットがフラグであり、これが 0になっていれば出力中
で、1ならば書き込み可能である。

さて、ASCIIコードとは、8(7)ビットで表される文字コードで、英文字と数字を表
すのに利用される。man asciiと打ち込んでみよう。対応が表示される。
以下のプログラムを実行すれば、Aと Bが出力されるはずである。

LDHI r0,#0x80

LOOP: LD r2,(r0)

BEZ r2,LOOP

LDIU r3,#0x41

ST r3,(r0)

LOOP2: LD r2,(r0)

BEZ r2,LOOP2

LDIU r3,#0x42

ST r3,(r0)

END: JMP END

このプログラムでは最初の LOOPでフラグをチェックし、1ならば A(0x41)を、次
の LOOP2でもう一度フラグをチェックして、1ならば B(0x42)を出力している。ディ
スプレイを摸擬するために、disp.vを使っている。このため、iverilogでシミュレー
ションを実行するには、webから test poco.vと disp.vを取って来て、

./shapa ex1.prg -o imem.dat

iverilog poco_test.v poco1.v alu.v rfile.v disp.v

vvp a.out

やれば良い。

演習 8.1

メモリの 0番地から順に格納された文字コードを 5つ、ディスプレイに出力するプロ
グラムを書け。プログラムは大したことはないが、shapaを使うことをおすすめする。
表示する文字は何でも良いが、面倒ならば web上の dmem.datを使うと良い。

8.2 バイトアドレッシング

8.2.1 LBと SB

前回のプログラムは、8bit=1byteの I/Oをアクセスするのに、ワード単位の LD、ST
命令を用いている。先に挙げた ASCIIコードの文字列や I/Oデータなどは 8bitが多
く、これをいちいち 16bit(あるいは 32bit)のワードで扱うとメモリの無駄が多い。そ
こで、一般的な計算機ではバイト単位のデータの読みだし、書き込みを行なう命令を
持つ。POCOでも以下の命令を用意する。

00000dddaaa01100LB rd,(ra) Load Byte 8bitデータの読みだし
00000sssaaa01011SB rs,(ra) Store Byte 8bitデータの書き込み

表 8.1: バイトアクセス命令

8.2.2 メモリのアドレッシング

今まで POCO は、データも命令も 16bitを 1ワードとして、ワード単位にアドレ
スを割り当ててきた。しかし、実はこれは一般的な方法ではない。世の中には ASCII
コードをはじめとして 8ビットのデータを扱う機会も多く、ワード単位にアドレスを
付けると効率が悪いためである。

そこで、今回 POCOのデータメモリのアドレスをバイトアドレッシング方式に変
更しよう。この方式は、8bit単位にアドレスが振る方法で、汎用プロセッサでは一般
的に使われる。さて、8ビット単位にアドレスを降っても、データメモリの幅自体は
16ビットなのでこれを二つに分離して順序付けを行う必要がある。メモリ中のデー
タの順序付けの方法には、図 8.3に示すように以下の二つがある。

• Big Endian: 2進数として見た場合の最上位 bit(MSB:Most Significant Bit)を 0番
地と考え、最下位 bit: (LSB:Least Significant Bit)を大きい方の番地にする。大
きい方の番地が LSB側であることから Big Endianと呼ぶ。

• Little Endian: 2進数として見た場合の最下位 bit(LSB)を 0番地と考え、大きい
方の番地をMSB側に持ってくる方法。小さい方の番地がMSB側であることか
ら Little Endianと呼ぶ。

図は 16ビットなので右と左を入れ替えただけだが、32ビットメモリ、64ビット
メモリの場合を考えると順番の問題が理解できると思う。この二つの方法は、市販の
CPUで統一されていないため、しばしば混乱を引き起こす。違った Endian同士でデー
タを交換すると、桁がひっくり返ってしまう可能性がある。混乱を嫌って、立ち上げ
時に指定することで、両方用いられるようにしている CPUもある。POCOでは Big
Endianを用いることにする。この方法では、16bitデータを扱う場合は、{偶数番地
8bit、奇数番地 8bit}の組を一つの 16bitデータとして偶数番地で代表させる。この時、
奇数番地から 16bitをアクセスしようとすると、メモリを 2回読み出す必要が生じて
しまう。このように、メモリの境界に合わないデータ配置のことをミスアラインメン
トと呼ぶ。ミスアラインメントを許すかどうかは、CPU毎に異なるが、POCOでは
実装が面倒になるため、許さないこととする。このため、以後、16ビットデータをア
クセスする場合は、偶数番地しか許さなくなる。これは注意が必要である。

LSBMSB

0 1

3

5

7

9

....

2

4

6

8

....

LSBMSB

01

3

5

7

9

....

2

4

6

8

....

Big Endian Little Endian

図 8.3: メモリの番地付け

もう一つ、問題がある。SBの場合は格納されないレジスタの上位 8ビットは単純
に切り落としてしまえば良い。しかし、LBの場合、意味のあるデータは 8ビットな
のにレジスタは 16ビットである。上位 8ビットはどうすれば良いだろうか？通常は、
イミーディエイト命令同様、符号拡張とゼロ拡張を用意しておく。すなわち LB命令
は符号拡張し、他に符号拡張しない LBU(Load Byte Unsigned)を設けておくことにし
よう。ここでは以下のように定義しよう。

00000dddaaa01101 // LBU rd,(ra): Load Byte Unsigned

ちなみに、この改造で POCOのデータメモリ空間は 216=64K× 8bitとなり、メモ
リの実体は 64K× 16bitの半分となる。さらに、やや変だが、命令メモリはワードア
ドレッシングのままとしておく。これは、命令は全て 16ビットなのでバイトアドレッ
シングを導入しても意味がないためである。

LB、SB命令を用いて前回用いたの出力装置への出力を行ってみよう。メモリマッ
プト I/Oでは、I/Oもバイトアドレッシングで番地が割り当てられる。ここでは LSB
側の 0x8001番地に割り当てられているとしよう。
メモリ中の 5個のデータ（バイトデータ）を出力するプログラムは以下のようにな

る。試してみよう。

LDHI r0,#0x80

ADDI r0,#1

LDIU r1,#0

LDIU r4,#5

LOOP: LB r2,(r0)

BEZ r2,LOOP

LB r3,(r1)

SB r3,(r0)

ADDIU r1,#1

ADDI r4,#-1

BNZ r4,LOOP

END: JMP END

8.2.3 LB/SBの実装

LB/SBの実装は案外面倒である。これは、以下の問題があるからだ。

• SBで偶数番地に書き込む場合は奇数番地に書き込んではならず、奇数番地に書
き込む場合は偶数番地に書き込んではならない。

• SBで偶数番地に書き込む場合、メモリの 15bit-8bitにレジスタの下位 8ビット
の値を格納しなければならない。これは今までの ST命令や奇数番地への SBと
違った場所である。

• LBで偶数番地を読む場合、メモリの 15bit-8bitをレジスタの下位 8ビットに格
納しなければならない。これは、今までの LD命令や奇数番地からの LBと違っ
た場所である。

このため、まず、データメモリを 8ビット単位に分割し、それぞれへの書き込み制
御信号を別々に持たせる。すなわち、we0(偶数番地：上位 8ビット用)と we1(奇数番
地:下位 8ビット用)が Hレベルの場合にそれぞれのメモリにデータが書き込まれると
する。また、偶数番地の読み書き用にはビットをシフトする必要がある。この辺は、
16ビットなので楽だが、32ビット以上ではさらに複雑（というか面倒）になる。
以下、関連の記述を示す。

..

wire lb_op, lbu_op, sb_op;

assign sb_op = (opcode == ‘OP_REG) & (func == ‘F_SB);

assign lb_op = (opcode == ‘OP_REG) & (func == ‘F_LB);

assign ddataout = (sb_op & ˜rf_b[0]) ? {rf_a[7:0],8’b0}: rf_a;

assign we0 = st_op | sb_op & ˜rf_b[0] ;

assign we1 = st_op | sb_op & rf_b[0] ;

..

assign rf_c = ld_op ? ddatain :

lb_op & ˜rf_b[0] ? {{8{ddatain[15]}},ddatain[15:8]}:

lb_op & rf_b[0] ? {{8{ddatain[7]}},ddatain[7:0]}:

jal_op ? pc+1 : alu_y;

assign rwe = ld_op | alu_op | ldi_op | ldiu_op | addi_op | addiu_op | ldhi_op

| lb_op | jal_op ;

さらに、データメモリも8ビットのメモリ×2に変更する必要があるため、test poco.v
も変更する必要があるのだが、これは各自チェックされたい。

演習 8.2

POCO1.vを改造し、LBUを実装せよ。テストは imem enshu test.datを imem.datに、
test enchu dmem0.dat を dmem0.dat に名前を変えて行え。r3 が 00ffになれば正解で
ある。

8.3 割り込み

8.3.1 割り込みとは？

今までのプログラムでは、I/Oモジュールのデータ送信が終了したかどうかを、モ
ジュールのコントロールレジスタを常にチェックすることで検出した。このような操
作をポーリング (Polling)と呼ぶ。しかし、これでは、CPUが入出力を行うためには、
常に I/Oモジュールのコントロールレジスタを定期的にチェックしなければならない
ことになる。適切な時期にチェックに行くためのプログラムを書くのは大変であるし、
しょっちゅうチェックに行くと性能が低下してしまう。そこで、このことを防ぐため
には、I/Oの方から、入出力を行う必要があることを CPUに対して積極的に教える機
構が必要になる。これが割り込み (Interrupt)である。普通、CPUは Operating System
か、少なくとも簡単なモニタプログラムの下でユーザプログラムを動かしており、割
り込みが発生すると、通常、OSまたはモニタプログラムが呼び出されて、必要な I/O
処理を行う。割り込みは、計算機の歴史の中でもかなり初期の段階で導入された概念
だが、後に必要性によって、以下の類似の機能が装備されるようになった。これらは、
ひっくるめて例外処理 (Exception)として扱われる。

• I/O割り込み: I/Oモジュールからの要求によって CPUの動作を変える。

• トラップ／割り出し: プログラムから割り込みをかけて、Operating Systemを呼
び出す。

• Page Fault: Memory Management Unitを用いて仮想記憶に基づく記憶管理を行っ
ている際に、実メモリ上に存在しない物理アドレスをアクセスした。この場合、
計算機の管理をする基本ソフトウェアであるOS(Operating System)が、ディスク
などの補助記憶から必要なページを持ってくる必要がある。この機構は、キャッ
シュの記憶階層を紹介する部分で触れるが、本格的な理解のためには、OSを学
ぶ必要がある。

• Segmentation Fault: OSで管理されている、そのユーザが使ってよいメモリ領域
以外をアクセスした場合に生じる。UNIX上で C言語で誤ったプログラムを書
くと往々にしてこのメッセージが表れる。

• Break Point: 特定の番地にしかけておき、そこをアクセスすると割り込みが生ず
る。デバッグ時に利用する。

• Trace: 命令を一つ実行する度に割り込みがかかる。

これらの例外処理は基本的に同様な機構で扱うことができる（厳密に言うと、Page
Faultなどの実行中に生じる例外処理は、I/O割り込みとは性質が違う）ので、ここで
は I/O割り込みについてのみ実例を示す。

8.3.2 割り込みの方式

割り込みをかけるためには、I/Oモジュールから CPUへの割り込み要求線が必要で
ある。I/Oモジュールは、この要求線に信号を送る（例えば Lレベルにする）ことに
より、CPUに割り込みの存在を知らせる。図 8.4は、最も単純な実装例である。この
場合、複数の I/Oモジュールは一本のオープンドレイン（コレクタ）線を共有してい
る。オープンドレイン（コレクタ）はディジタル回路の出力の構成法の一つで、複数
の出力を一つにまとめることができ、どれかが一つの出力でも Lにすると全体が Lに
なる。

I/O I/O I/O

L

Request
Module

CPU
Interrupt Request
line (Open Drain)

図 8.4: 割り込み要求線

CPUは、割り込み信号線が Lになっており、内部状態として割り込みが許可され
ていれば、戻り番地と分岐判定用のフラグをどこかに保存してから、強制的に割り込
み処理ルーチンにジャンプする。割り込みは、サブルーチンコールと違ってどこでか
かるかわからない。分岐判定直前にかかるかもしれないので、分岐判定にフラグを用
いる場合は、この情報も保存する必要がある点に注意しなければならない。幸いにし
て今回の PICOの実装は、比較結果はレジスタに反映されるので、そのレジスタを保
存すれば済む。
メインルーチン実行中に割り込み要求が発生した場合の処理の様子を図 8.5に示す。
割り込み処理ルーチンにジャンプすると、普通は、割り込みは強制的に禁止 (Disable)

される。これは禁止しないと、割り込み処理ルーチンに飛んだと思ったら次の命令で
再び割り込みがかかってしまうためである。複数の I/Oモジュールが存在する場合は、
割り込み処理ルーチン内でどのモジュールが割り込みをかけたかを知るために、CPU
は I/Oモジュールのコントロールレジスタを順に探して行く。この辺は割り込みを持
たない演習プログラム同様であるが、割り込みがかかったのだから、必ずどこかのモ
ジュールが要求しているはずなので、そこで当たるはずである。この点が、要求が存
在しないかもしれないのにチェックを繰り返す単純なポーリングとは本質的に異なっ
ている。
要求を出したモジュールを見つけたら、CPUは、その要求を満足するアクセスを

I/Oモジュールに対して行う。このことにより I/Oモジュールは割り込み要求を引っ込
めるはずなので、CPUは、割り込みから復帰後に、元々行っていたプログラムを再実
行することができる（もちろん割り込み処理中に別のモジュールがまた要求を出すこ
ともある。この場合は復帰したらすぐまた割り込み処理ルーチンに飛ぶことになる。）。

Interrupt 
handler

Interrupt
Disable

Interrupt
Enable

Interrupt
Request

Main routine

RTI

I/O Access

図 8.5: 割り込み処理の様子

戻り番地とフラグの保存場所は以下の２つの方法が一般的である。

• 専用のレジスタ IAR(Interrupt Address Register)、IFR(Interrupt Flag Resiter)など
に格納する。POCOは分岐にフラグを用いないため、IFRは不要である。

• システムスタックに格納する。

サブルーチンコールの戻り番地をレジスタに格納する方法を取っている CPUは前者、
サブルーチンコールにシステムスタックを用いる方法の CPUは後者を用いる場合が
多い。ただし、汎用のレジスタを、戻り番地の保存に使うことはしない。これは、サ
ブルーチンコールと違って、割り込みはいつかかってくるかプログラマの予想がつか
ないため、戻り番地の保存に使うことを決めたら、そのレジスタはそれ以外の用途に
は全く使えなくなってしまうためである。
次に割り込み処理ルーチンへのジャンプの方法も様々である。

• 固定番地にジャンプする。

• 特定のレジスタ又はメモリの番地に書かれている番地にジャンプする。

後者の方法は、割り込み処理ルーチンの配置番地の柔軟性が高くなるが、割り込み処
理の実装がめんどうである。

割り込み処理からの復帰は、専用の命令（例えば RTI:ReTurn from Interrupt)を用い
る場合が多い。これは、割り込み処理からの復帰は、PCを (分岐にフラグを用いる場
合フラグも)元に戻し、次の割り込みを許可する複合操作が必要とされるためである。
さて、今回 POCOでは、最も単純な割り込みの実装法を採用する。すなわち、(1)

戻り番地は専用レジスタの IARに格納し、(2)割り込みがかかった際には固定番地に
飛ぶ。
先に述べたように、POCOでは分岐にフラグを用いていないため、フラグを保存す

る必要がなく、この点は非常に楽である。割り込みがかかったときの飛び先番地は、
ここではなんとなく 0xff00番地とした。したがって、割り込み処理プログラムの先頭
は必ず 0xff00番地となる。また、一般的に CPUは、リセット時には割り込み禁止状
態にしておく。これは、スタート時にスタックや I/Oモジュールの初期化を行う必要
があるが、この際には割り込みがかかっては困るためである。そこで、ある時点で割
り込みを許可するための命令 EINT(Enable Interrupt)が必要である。ここでは以下の
命令を設ける。

00000xxxxxx01110EINT Enable Interrupt 割り込みを許可
00000xxxxxx01111RTI ReTurn from Interrupt 割り込みプログラムから復帰

表 8.2: 割り込み命令

これらは、レジスタ指定の必要はないが、R形式を用いることにする。さて、割り
込み処理の Verilogでの記述の解説は後に回して、まず、割り込み機構の働きを観て
みよう。
割り込みはもちろん出力装置についても可能だが、ここではより入力時刻が予想し

にくいキー入力装置を考える。
この装置は、0x9001番地に 1バイト分のデータレジスタが割り当てられており、人

がキーを押すと、割り込み信号 intreqが Hレベルになると共に、データレジスタに 8
ビットの文字列がセットされる。0x9001番地からデータを読み出すと、割り込み信号
はクリアされ、次のキー入力が可能になる。上記は典型的な入力装置の動きである。
まず、分かりやすさのため、メインルーティンは、単純に無限ループとする。

00000_000_000_01110 // EINT

10100_11111111111 // END: JMP END

今回は何も行わないが、もちろん本来は無限ループの代わりに何か意味のある処理
をやっているはずである。
次に割り込み処理ルーチンは下のようになる。これを 0xff00番地に置く。

01010_000_10000000 // : LDHI r0,#0x80

01101_000_00000001 // : ADDIU r0,#1

00000_010_000_01100 // : LOOP:LB r2,(r0)

10000_010_11111110 // : BEZ r2,LOOP

01010_100_10010000 // : LDHI r4,#0x90

01101_100_00000001 // : ADDIU r4,#1

00000_101_100_01100 // : LB r5,(r4)

00000_101_000_01011 // : SB r5,(r0)

00000_000_000_01111 // RTI

さて、割り込みが掛かったということは、キー入力が行われたことは間違いない。
しかし、ディスプレイについては出力が可能かどうかわからないので、LOOPで確認
し、可能となった後にキー入力装置から読み出してきたものを書き込んで、表示して
メインルーティンに戻る。

演習 8.3

例題でやった処理は、割り込み処理ルーチン中でディスプレイの表示が終わるのを待
つため、表示がキー入力よりも遅いと（そんなことは普通ないと思うが）、入力を表
示し損ねる可能性がある。そこで、以下の方法を考える。

• レジスタ r3を入力ポインタとし、r4を出力ポインタとする。

• メインルーチンでは r3,r4の間は常に r4の指すメモリの文字コードを出力し、
一つ出力したら r4をカウントアップする。

• 割り込み処理ルーチンでは、入力した文字コードを r3の示すメモリの番地に格
納し、終わったら r3をカウントアップする。

この方法は通信用バッファで用いられる典型的な方法である。
メインルーチンは例えば以下のようになる。

01010_000_10000000 // 0 : LDHI r0,#0x80

01101_000_00000001 // 1 : ADDIU r0,#1

01010_001_10010000 // 2 : LDHI r1,#0x90

01101_001_00000001 // 3 : ADDIU r1,#1

01000_011_00000000 // 4 : LDI r3,#0

01000_100_00000000 // 5 : LDI r4,#0

00000_000_000_01110 // EINT

00000_101_011_00001 // 6 : LOOP: MV r5,r3

00000_101_100_00111 // 7 : SUB r5,r4

10000_101_11111101 // 8 : BEZ r5,LOOP

00000_010_000_01100 // 9 : LOOP2: LB r2,(r0)

10000_010_11111110 // 10 : BEZ r2,LOOP2



124 第 8章 入出力：I/O

00000_010_100_01100 // 11 : LB r2,(r4)

00000_010_000_01011 // 12 : SB r2,(r0)

01101_100_00000001 // 13 : ADDIU r4,#1

10100_11111110111 // 14 : JMP LOOP

ここでは、r0にディスプレイの番地 0x8001を、r1にキー入力の番地 0x9001を設
定して初期化しておく。
割り込みルーチンを書いて、このバッファシステムをシミュレーションして見よ。

割り込みルーチンでは r1はすでに設定されているものとして良い。また、割り込み
ルーチンで使えるレジスタはメインルーチンと重なってはならない。

8.3.3 割り込み時のレジスタ保存

前回の課題のメインルーチンは以下のようになっていた。

01010_000_10000000 // 0 : LDHI r0,#0x80

01101_000_00000001 // 1 : ADDIU r0,#1

01010_001_10010000 // 2 : LDHI r1,#0x90

01101_001_00000001 // 3 : ADDIU r1,#1

01000_011_00000000 // 4 : LDI r3,#0

01000_100_00000000 // 5 : LDI r4,#0

00000_000_000_01110 // EINT

00000_101_011_00001 // 6 : LOOP: MV r5,r3

00000_101_100_00111 // 7 : SUB r5,r4

10000_101_11111101 // 8 : BEZ r5,LOOP

00000_010_000_01100 // 9 : LOOP2: LB r2,(r0)

10000_010_11111110 // 10 : BEZ r2,LOOP2

00000_010_100_01100 // 11 : LB r2,(r4)

00000_010_000_01011 // 12 : SB r2,(r0)

01101_100_00000001 // 13 : ADDIU r4,#1

10100_11111110111 // 14 : JMP LOOP

このプログラムはメインルーチンと割り込みルーチンが協調して処理を行っている。
r0はディスプレイの番地 0x8001、r1はキー入力の番地 0x9001に初期化されている。
また、r3は割り込み処理ルーチンが、r4はメインルーチンがポインタとして利用する
ことがあらかじめ決っている。しかし、他のレジスタ、例えば r2,r5についてはどう
だろう？サブルーチンコールと違って割り込みはいつ生じるか予測が付かない。した
がって r2,r5の利用中に割り込みが発生し、割り込みルーチンでこのレジスタを利用
すると内容が破壊されて、正常にプログラムが動かなくなる。前回の演習では割り込

みルーチンとメインルーチンではレジスタの番号が重ならないようにした。しかし、
一般的にはこれではレジスタの数が不足する可能性がある。
ここでサブルーチンコールの際に学んだスタックを思い出して欲しい。割り込み処

理ルーチンでも全く同じようにスタックを使ってレジスタを退避し、復帰する。この
ためにはスタックポインタをメインルーチンの開始時に設定しておく必要がある。こ
こではスタックポインタを r6として、スタック領域を 7ffe番地から下の方に向かっ
て伸びるように設定しよう。

LDHI r6,#0x80

...

EINT

...

割り込みルーチンは、最初にルーチン内で利用するレジスタ (例えば r2と r5を使
うとする)をスタックに退避し、RTI前にこれを元に復帰する。

ADDI r6,#-2

ST r2,(r6)

ADDI r6,#-2

ST r5,(r6)

...

...

LD r5,(r6)

ADDI r6,#2

LD r2,(r6)

ADDI r6,#2

RTI

退避する際には 2を（バイトアドレッシングにしたので、、）引いてから STし、復帰
の際は LDしから 2を足す点に注意。スタックなので、積んだ逆順で戻す必要がある。
繰り返すが、割り込みはメインルーチンのどこで発生するか分からないので、割り

込み処理ルーチン中では必ずこの操作を行う必要がある点に注意されたい。

8.3.4 割り込みの実装

I/Oからの割り込みだけならば、実装はさほど難しくはない。命令読み出し状態 (IF)
で、割り込み要求線 (intreq)がアクティブになっているかを検出し、割り込み可能状
態 (int enableが H)であれば、pcを割り込みアドレスレジスタ (iar)に保存すると共
に、割り込み番地を設定する。RTI命令を検出すると、iarに保存してあった番地を pc

に設定することでメインルーチンに復帰する。この部分について、まず pcについて
は以下のよう拡張する。

wire rti_op;

assign rti_op = stat[‘EX] & (opcode == ‘OP_REG) & (func == ‘F_RTI);

...

always @(posedge clk or negedge rst_n)

begin

if(!rst_n) pc <= 0;

else if(jmp_op | jal_op)

pc <= pc +{{5{ir[10]}},ir[10:0]} ;

else if(pcsel)

pc <= pc +{{8{imm[7]}},imm} ;

else if(jr_op)

pc <= rf_a;

else if(rti_op)

pc <= iar;

else if(stat[‘IF]) begin

if(int_enable & intreq)

pc <= ‘INTAD;

else pc <= pc+1; end

end

割り込みがかかったら割り込み番地から命令をフェッチしなおすため、IFを 2回続
けて実行することになる。この制御は以下のようにする。

always @(posedge clk or negedge rst_n)

begin

if(!rst_n) stat <= ‘STAT_IF;

else

case (stat)

‘STAT_IF: if(!(int_enable & intreq)) stat <= ‘STAT_EX;

‘STAT_EX: stat <= ‘STAT_IF;

endcase

end

iarについては割り込みがかかったら pcを設定すれば良いので簡単である。

reg [‘DATA_W-1:0] iar;

...

always @(posedge clk)

if(stat[‘IF] & int_enable & intreq) iar <= pc;

割り込み許可フラグ int enableは、EINT命令で Hとなり、割り込みを受け付ける
と Lとなって割り込み禁止となる。RTI命令が実行されると再び Hとなる。したがっ
て以下の記述で制御する。

reg int_enable;

wire eint_op;

assign eint_op = stat[‘EX] & (opcode == ‘OP_REG) & (func == ‘F_EINT);

...

always @(posedge clk or negedge rst_n)

begin

if(!rst_n) int_enable <= ‘DISABLE;

else if(eint_op | rti_op) int_enable <= ‘ENABLE;

else if(stat[‘IF] & int_enable & intreq)

int_enable <= ‘DISABLE;

end

例外処理で面倒なのは、命令実行中に生じるページフォルトである。しかし、この
例外処理については、メモリ階層の所で触れることにする。

8.3.5 多重割り込み

前回のプログラムでは、割り込み処理ルーチンはきわめて簡単なものであったので、
ずっと割り込み禁止の状態で走った。しかし、場合によっては割り込み処理ルーティ
ンが長くなり、処理を実行中にも別の割り込みを受け付ける必要がある場合もある。
これを多重割り込みと呼ぶ。多重割り込みを実現するためには、CPUに割り込みレ
ベルを設けるのが最も都合が良い。
この場合、CPUに対する割り込み線を拡張して複数持たせ、信号線によって、割

り込みレベルを区別する。さらに、CPUに割り込みレベルを持たせる。メインルー
チンの割り込みレベルを 0とし、現在実行中の割り込みレベルに比べて、高いレベル
の割り込み要求を検出したらその割り込みを受け付け、割り込み処理ルーティンに飛
ぶ。この際に CPUの割り込みレベルは、受け付けた要求と同じレベルまで自動的に
高くなる。現在実行中の割り込みレベルより低い割り込み要求は受け付けられず待た
される。
しかし、このような割り込みレベルを持っていない場合、ソフトウェアで、割り込

み処理中に割り込みを許可し、疑似的に多重割り込みを許してやる方法もある。この
場合、戻り番地やフラグをスタックに待避する必要があるので、IARと（分岐にフラ

グを用いる場合はフラグ保存用レジスタも)汎用レジスタ間の転送命令が必要になる。
また、割り込み許可／禁止をプログラムで扱う必要があるため、以下の命令が必要に
なる。

00000xxxxxx10000 DINT Disable Interrupt 割り込み禁止
00000dddxxx10001 MFIAR Move From IAR d <- IAR

00000xxxsss10010 MTIAR Move To IAR s -> IAR

とはいえ、このような命令を用いてソフトウェアで多重割り込みを実現する方法は、
かなりのプログラミング能力を要する。

演習 8.4

DINT, MFIAR, MTIAR命令を poco.vに実装せよ。

8.3.6 DMA(Direct Memory Access)

割り込みを用いて入出力プログラムを起動することにより、他の作業を行ないなが
ら、I/Oデバイスとデータのやりとりをすることができる。先に紹介したシリアルイ
ンタフェースなどの遅い I/Oに対しては、プログラムでデータの転送を行なっても十
分である。データ出力を例にとると、図 8.6a)に示すように、メモリの内容を LB命
令で読み出し、I/Oの番地に SB命令で書き込むことで転送を行う。ところが、ネッ
トワークやディスクなどの高速な I/Oに対しては、このように CPUを介して LB/SB
命令を用いてデータの授受を行なうのでは、デバイスの速度について行くことができ
ない。

CPU

Memory

I/O

LD/LB

ST/SB

a) I/O Access by CPU

CPU

Memory

I/O

b) Direct Memory Access

DMA

図 8.6: DMA転送による出力

このような場合、I/Oデバイスが CPU からバスの利用権を一時的に借り受け、図
8.6b)に示すように、直接メモリとの間でデータの連続転送を行なう。このような操
作を DMA(Direct Memory Access)と呼ぶ。DMAは以下の手順で行なう。

1. 転送を行ないたいデバイスは、CPUに対してバスの利用権を請求する。

2. CPUは命令の区切りでバスを解放してデバイスに明け渡す。転送を行ないたい
デバイスが複数存在する場合は、どれか一つを選ぶ。この操作をバスアービト
レーションと呼び、選択回路をアービタと呼ぶ。

3. 利用権を獲得したデバイスは、直接メモリとの間でデータの転送を行なう。こ
の間、CPUはバスを利用できないため、メモリ中の命令をフェッチできず、休
止状態となる。

4. データ転送を終了したデバイスは、バス利用権を CPUに戻す。CPUはプログ
ラムの実行を再開する。

5. DMAは CPUがバスを明け渡した際に行なうため、CPUはデータの転送がどう
なったか全く検知することができない。したがって、一連のデータ転送処理が
完全に終了した場合、データ転送を行なったデバイスは、CPUに対して割り込
みをかけて、転送処理の終了を知らせる。

DMAは、転送を開始するまでの手続きに時間を要するため、ある程度まとまった大
きさのデータをまとめて転送するのに用いる場合が多い。ただし、DMAを行なうた
めにバスを長期間占有すると、その間 CPUはまったくプログラムを実行することが
できないので注意が必要である。
最近のシステム LSIでは、CPUと複数の I/Oデバイスが同一チップ上に実装されて

おり、これらの間で DMAを効率良く行なうために、専用の DMAコントローラを持
つ。DMAコントローラの設計は、個々の I/Oデバイスの性質に依存する点が大きい。

第 9

パイプライン処理

9.1 パイプライン処理の基本

POCOは 1サイクルCPUで全ての命令を 1サイクルで実行する (CPI=1)代わりにク
リティカルパスが長くなり、動作周波数が落ちる。web上で検討しているが、マルチ
サイクル化することにより、動作周波数を改善することは可能であるが、その分 CPI
が増えてしまって全体の性能向上には結びつかない。

CPIを増やさずに、動作周波数を大きく改善するためには、パイプライン処理を取
り入れる必要がある。最近の CPUのほとんど全ては、パイプライン処理を基本とし
た構造を持っている。特に RISCの命令セットはもともとパイプライン処理の効率化
を念頭に置いて作られており、80年代中頃から 90年代にかけてマイクロプロセッサ
の性能向上に最も寄与した技術の一つである。
パイプライン処理は、作業の効率を上げるために工場などで用いる流れ作業と同じ

考え方である。まず作業全体をいくつかの段階（ステージ）に分割し、それぞれを担
当が受持ち、前段階の処理を終えた途中結果（情報）をもらって、自分の作業を行い、
次のステージの担当に送る。流れ作業の流れの速度は、最も作業時間の長いステージ
によって決まる。最も作業時間の長いステージの直後のステージは、そのステージか
ら途中結果を貰えないため、作業を開始できない。一方、最も作業時間の長いステー
ジの直前のステージでは、そのステージに結果を送ることができず、待ち状態になる。
これが全てのステージに伝搬することから、結局最も作業時間の長いステージに合わ
せて全体が動くことになる。
すなわち、流れ作業による性能向上は、

（全体の作業時間）／（最も長いステージの作業時間）

ということになる。したがって、効率を上げるためには、各ステージの作業時間をな
るべく等しくすることが重要である。全体の作業が完全に等しいステージに分割でき

れば、性能はステージ数倍になる。CPUの場合、理想的には、クロック周期をステー
ジ数分の 1にすることが可能である。
しかし、パイプライン処理は以下のオーバヘッドを考える必要がある。

• ステージ間のデータの受渡しにはレジスタが必要になる。このため全体のハー
ドウェアが増え、レジスタにデータを取り込む時間が必要になる。

• 各ステージはそのステージで専用に使える資源が必要なのでハードウェア量が
増大する。しかし、POCOはもともと 1サイクル CPUで、資源の共有を行って
いないため、このためのハードウェア量の増加は考えなくて良い。

しかし、通常、上記のオーバヘッドが問題ならない程の性能向上を実現することがで
きる。

9.2 POCOのパイプライン処理
9.2.1 パイプライン構成

POCOの全体構成は、大きく考えて、命令フェッチを行うフロントエンドと実行を
行うバックエンドの二つの部分に分けられる。そこで、これを素直にパイプライン化
すると、図 9.1a)のように、命令フェッチ部 (IF)と実行部 (EX)をそれぞれステージ
と見なして、パイプライン処理する方法が考えられる。POCOでは、元々pc, pc用の
加算器、命令メモリと、ALU、データメモリは独立しているため、ほとんど苦労す
ることなしにパイプライン構造が実現できる。一点重要なのは、今までと違って取っ
てきた命令を蓄えて置くレジスタが必要になる。これを、命令レジスタ (instruction
resgister:ir)と呼ぶ。
しかし、良く考えてみると、これは両者のバランスが悪い。IFステージは、命令メ

モリを読み出し、pcに 1あるいは飛び先を加える操作を行うだけである。これに対
して EXステージは、レジスタ間演算命令を例に取ってみると以下の操作が必要にな
る。(1)レジスタファイルから rs, rdで示したレジスタを読み出し、(2)ごちゃごちゃ
している入力のマルチプレクサを経由し、(3)ALUで演算し、(4)答えをメモリからの
データと切り替えるためのマルチプレクサを経由してレジスタファイルの入力にデー
タを与え (5)書き込みを行う。
命令メモリとデータメモリの読み書き時間が十分高速とすると (この仮定はきちん

としたキャッシュを使わないと成り立たず、これは後で紹介する。) IFよりも EXは 2
倍から 3倍の時間が掛かることが予想される。このため、パイプライン処理は図 9.1b)
に示すようにうまく働かない。
そこで、EXステージを三分割して、図 9.1c)のように動作させることを考える。ど

のように分割すれば良いだろう? 通常、EXステージで実行する処理で最も時間を食
うと考えられるのは ALUにおける演算である。そこで、図 9.2に示すように ALUの

図 9.1: POCOのパイプライン処理

演算の直前 (areg, breg)と、直後にレジスタ (creg)を挿入する。そして、演算のパス
を以下のように分割する。

• ID(Instruction Decode)ステージ: レジスタを読み出して ALUの入力マルチプレ
クサを経由してレジスタ areg, bregに値をセットする。同時に ir中の命令コー
ドをデコードして、ALUの comを作り、これを com idにセットする。

• EX(Execution)ステージ: ALUは com idに従って areg, bregの値を演算し、答
えを cregにしまう。LD命令の場合は、データレジスタからデータを読み出し
てこれを dregにしまう。

• WB(Write Back)ステージ: LD命令ならば dregの値、それ以外ならば cregの値
をレジスタファイルに書き込む。

この方法のミソは、最も時間の掛かる ALUの前後の処理を EXステージから分離
し IDステージとWBステージとして独立させた点にあり、CPU全体が純粋に ALU
内の遅延時間で動作可能となる。

9.2.2 各ステージの処理

図 9.2は、今まで解説した POCOにパイプライン用のレジスタを挿入した図である
が、パイプラインに命令が流れていく様子が掴み難い。そこで、データパスを横にし
てみたのが図 9.16である。この図は、LD,ST,LDI,LDIU,ADDI,ADDIUのみから成る
単純な構造のパイプラインで、IF, ID, EX, WBの 4つのステージが左から右に並んで
いる。すなわち、IFでフェッチされた命令は左から右方向に流れていく。

図 9.2: パイプライン化されたデータパス

IFステージ

pcに従って命令メモリから命令をフェッチし、irに格納する。今までとの違いは、こ
れを毎クロック行う点にある。このため、pcは分岐命令以外は毎クロックカウント
アップされる。つまり常に pcset=1, irset=1であるので、図中では省略する。
では、分岐命令はどうだろう？実は分岐命令は次に実行する命令が分岐する場合と

しない場合で違ってくるので、パイプラインの動きが乱れてしまう。このため、ちょっ
と棚上げしておいて、後で検討することにしよう。分岐命令を考えなければ IFは非
常にシンプルである。

IDステージ

実行する命令は irに保存されている。ID(Instruction Decode)ステージの役目は、この
命令に従って、レジスタファイルからデータを読み出すと共に、ALUの comやレジ
スタファイルへの書き込み用マルチプレクサの制御に至るまで、すべての制御信号を
生成してやることである。この処理のことを解読：デコードと呼ぶ。
基本的には今までの POCOと同様に、rd, rsに従ってレジスタファイルからレジス

タを読み出すと共に、イミーディエイトフィールドの拡張を行う。レジスタの rsに
従って読み出された値は、マルチプレクサを経由して bregにセットする。イミーディ
エイトデータが利用される場合は、符号拡張してから同様に bregにセットする。a入
力についてはレジスタファイルから読み出した値を直接 aregにセットする。ALUの
com入力についても、EXステージですぐに演算ができるように、このステージで値
を決めておき、これを次の EXステージで使えるように保存する。
このように、デコードした結果は、今このステージで使わず、EXステージやWB

ステージで利用する場合もある。このため、IDステージでは様々な信号をレジスタ
（1ビットのものは Flip Flop)に格納しておく必要がある。このようにパイプライン間
のデータの受け渡しに使うレジスタのことをパイプラインレジスタと呼ぶ。

IDステージで生成した信号を後のステージに送るためのパイプラインレジスタに、
ここでは idを付けて表す。例えばALUのコマンドは com idという名前のレジスタに
格納して EXステージに送る。同じように命令の解読信号 (例えば rwe、st op、ld op
など)も、全てレジスタに格納して EXステージに渡してやる。1

EXステージ

計算のためのお膳立ては全て IDステージで整っているため、ここでは純粋に演算だ
けを行って結果を cregに格納する。メモリのアクセス命令の場合を考えて、bregの
値をデータメモリのアドレスへ、aregの値をデータへ送ってやる。ST命令では、メ
モリの書き込みを行わなければならないため、IDステージで作られた st op id信号
が 1ならば、we nを Lにしてやる。LD命令では、読み出したデータを専用レジス
タ dregに格納する。レジスタファイルに書き込む命令では書き込むレジスタ番号の
入った rd id、書き込み信号 rwe idをそのままここでもレジスタに格納して、id ed、
rwe exとして次のWBステージに送ってやる。また LD命令かどうかを示す ld op id
もレジスタに格納して、ld op exとしてWBステージに渡してやる。

WBステージ

レジスタファイルに書き込むべき答えは、LD命令ならば dregに、ALU命令ならば
cregに入っている。これを ld op ex信号で選んで、レジスタファイルに送ってやる。
レジスタファイルは便宜上 IDステージに書いてあるが、書き込み部分については、
WB ステージから送った信号で制御する。すなわち、選んだ書き込みデータ rf c を
rwe exが 1の時だけ番号が rd exで示されるレジスタに書き込んでやる。

9.2.3 パイプラインの動作

以下の命令がパイプライン中を流れて実行される様子を示す。太字で示す部分に有
効なデータが流れている。

01000_000_00000000 // 0 : LDI r0,#0

01000_001_00000001 // 1 : LDI r1,#1

01000_010_00000010 // 2 : LDI r2,#2

00000_011_000_01001 // 3 : LD r3,(r0)

1この考え方で行くと areg, breg も alu a id や alu b id と呼ぶべきかもしれないが、これらは重要なので
もっと簡単に呼べるように特別な名前とした．

00000_000_001_00110 // 4 : ADD r0,r1

00000_010_010_00110 // 5 : ADD r2,r2

01100_011_00000011 // 6 : ADDI r3,#3

最初の図 9.3は命令がスタートしてから 3クロック目のパイプラインの様子を示す。
ここでは、LDI r0,#0が EX, LDI r1,#1が ID、LDI r2,#2が IFで実行されている。IF
では、LDI r2,#2が命令メモリから読み出されている。次のクロックの立ち上がりで、
この命令は irに格納され、IDステージに渡される。

IDでは LDI r1,#1の制御信号が Decorderで生成されると共に、レジスタファイル
が読み出される。ただし、この場合、読み出された r1の値は使われず、irの下 8ビッ
トを符号拡張された値が bregに格納される。同時に THBを示す ALUコマンドもレ
ジスタに格納されている点に注意されたい。

EXでは LDI r0,#0が実行されている。ここでは bregの値が、前のクロックで設定
された THB操作に従って cregに格納される。

次の図 9.4は 4クロック目である。ここでは、LDI r0,#0がWB, LDI r1,#1が EX、
LDI r2,#2が IDで実行され、LD r3,(r0)が IFで命令メモリから読み出されている。

WBでは LDI r0,#0が実行されている。ここではすでに前のクロックで結果が出た
cregの値が選ばれて、レジスタファイルの r0に格納される。EX, ID, IFの操作は基本
的に前のクロックと同じである。

図 9.4: 4クロック目の様子

次の図 9.5は 5クロック目である。ここでは、LDI r1,#1がWB, LDI r2,#2が EX、
LD r3,(r0)が IDで実行され、ADD r0,r1が IFで命令メモリから読み出されている。

WBでは LDI r1,#1が実行されている。ここでは先のクロック同様に、cregの値が
選ばれて、レジスタファイルの r1に格納される。EXでは、前のクロック同様に LDI
r1,#2が実行され、2が cregに書き込まれている。IDでは、LD r3,(r0)が実行される。
この様子に注目されたい。今までと異なり、rd, rsによって示されたレジスタの値で
ある r3,r0 が両方共 areg, bregに格納されている。(しかし今度は bregの方は使われ
ない。)
次の図 9.6は 6クロック目である。ここでは、LDI r2,#2がWB, LD r3, (r0)が EX、

ADD r0,r1が IDで実行され、ADD r2,r2が IFで命令メモリから読み出されている。
WBの動作は今までと同じだが、EXでは LD命令が実行されている。このために、

前のクロックで格納されたアドレスがメモリに送られている (データも送られている

が、LD 命令なので使われない)。読み出されたデータは dregに格納される。また、
ADD r0,r1が実行される IDにも注目されたい。両方のレジスタが読み出されて areg,
bregに設定されると共に、命令コードの functの下のビットが ALUのコマンドレジ
スタに格納されている。

図 9.6: 6クロック目の様子

次の図 9.7は 7クロック目である。ここでは、LD r3,(r0)がWB, ADD r0,r1が EX、
ADD r2,r2が IDで実行され、ADD r2,r2が IFで命令メモリから読み出されている。
WBでは dregの値が選ばれて、レジスタファイルに格納される。はるばるパイプライ
ンレジスタを経由して運ばれてきた ld op ex信号がここで利用される。EXでは ADD
命令が実行されて答えが cregに格納される。IDではもう一つの ADD命令が実行さ
れ、同様にレジスタとコマンドの値がレジスタが設定される。
これ以降は省略するが、毎クロック各ステージで命令が実行されている様子が理解

していただけるだろう。

演習 9.1

演習用の imem enshu.datをシミュレーションして、r0がどのように変化するか 1ク
ロックずつ調べよ。

9.2.4 パイプライン処理とタスクレベルの並列処理

パイプライン処理と並んで性能を上げるために使われる方法としてタスクレベルの
並列処理がある。これは、たくさんの人間や処理装置がそれぞれ独立に同じ作業を行
うことで高速化する方法である。要するに手伝ってくれる人が居て、皆で同じことを
同時にやれば、仕事が早く終わるわけで、これも日常的に使われている。
パイプライン処理は並列処理に比べて以下の利点を持っている。

• 並列処理は、それぞれの人間が、処理全体を通して行うことが必要とされるが、
パイプライン処理は、どれか一つの作業のみを行えば良い。（人間が一つの工程
のみ繰り返し行うと、単調すぎるためにかえって効率が低下する。しかし、論理
回路の場合一つの作業のみで済むということは、単純な回路で済むことになっ
て有利である。）

• 並列処理は、全ての工程を行うために必要な工具、施設、作業場所が必要であ
るが、パイプライン処理は、ある工程のみを行う工具、施設、作業場所があれ
ば良い。

• 並列処理は、各人受け持つ作業が完全に独立でなければ同時に実行するのが難
しいが、パイプライン処理は、一連の流れの作業を効率化することができる。

これらの利点により、工場などでは並列処理よりもパイプライン処理あるいは流れ作
業を行う場合がはるかに多かった2。一方で並列処理の利点は以下の通りである。パイ
プラインは一連の作業を均等にステージに分割すれば、性能をステージ数倍上げるこ
とができるが、ステージ間の受け渡しのロスがあるため、普通はさほど多くのステー
ジに切ることはできない。計算機の場合、多くてもパイプラインの段数は 15程度で
ある。これに対して並列処理は、同時に処理が可能であれば、100人同時に仕事をす
れば 100倍に生産量を向上させることができる。このため、パイプライン処理での性
能が限界に達した後に、プロセッサはまず命令レベルの並列処理を用いて性能を向上
させ、最近はスレッドレベルというプログラムの大きな塊間での並列処理を利用する
ようになった。マルチコアプロセッサ、メニーコアプロセッサなど、最近のプロセッ
サは並列処理を用いている。

2最近は、多品種少量生産あるいは新製品の出現サイクルが短くなったこと、また、人間の達成感が得ら
れることの効果が従来考えられていたより大きいことがわかったため、工場の流れ作業も見直されている。

9.3 パイプラインハザード

今までは、パイプラインがスムーズに流れる場合のみを考えて来た。しかし、パイ
プラインは常にスムーズに流れるわけではなく、様々な原因によってその動作を停止
する危険が生じる。この危険のことをパイプラインハザードと呼ぶ。一般的に、パイ
プラインハザードには以下の三種類がある。

• 構造ハザード:パイプラインの違ったステージ間で、同じ資源を使う場合に生じ
るハザード。POCOではこれは起きないが、例えば命令メモリとデータメモリに
同じものを使うと起きる。つまり、LD命令を実行している時に、次の次の命令
がフェッチできなくなってしまう。構造ハザードは資源を増やせば解決できる。
この場合、pocoでやったようにメモリを分ければ解決できるが、もちろんその
分コストが増える。構造ハザードを認めるのは、パイプラインを停止 (ストール
と呼ぶ)させて若干性能が落ちても、コストが小さい方が良い場合である。

• データハザード:データの依存性によって生じるハザード

• 制御ハザード:分岐命令に関連するハザード

命令メモリとデータメモリを単一のメモリで共有した場合生じる、構造ハザードを
例にハザードによる性能低下を考えよう。図 9.8に示すように、LD命令などメモリ
を利用する命令は、3クロック目の EXステージでメモリを利用する。このため、2番
目の命令 (Inst2)がフェッチする時にメモリを使えなくなり、一クロック待たなければ
ならない。これは、パイプライン中の泡 (バブル)となり、このため、命令の終了は 1
クロック遅れる。バブルによりパイプラインの性能が低下することをパイプラインス
トールと呼ぶ。

IF EXID WBLD

Inst1

Inst2

Inst3

IF EXID WB

IF EXID WBstall

IF EXID WB

図 9.8: 構造ハザード

パイプラインストールにより、CPIがどれだけ増えるか調べよう。泡が入った後の
CPIは以下の式で表される。

理想の場合の CPI +泡の入る確率×泡の入る個数　

例 題 9.1: メモリをアクセスする LD命令、ST命令の確率が合わせて 25%である場
合、パイプラインの CPIはどのようになるか。

答 : 理想の CPIは 1である。泡は 1個分なので、1+0.25*1=1.25となる。
構造ハザードは、リソースの複製により対処可能である。メモリについては命令メ

モリとデータメモリを分離すれば構造ハザードは生じない。実際にはメモリを分離す
るには相当のコストが掛かるため、キャッシュを分離して対処する。これは次の章で
触れる。
さて、POCOでは構造ハザードは生じないが、データハザードと制御ハザードは深

刻な問題である。順にこれを検討していく。

9.3.1 データハザード

実は先に例を示したプログラムは、不自然であったことに気づくかもしれない。以
下のプログラムは 0番地の内容と 1番地の内容を加算してその結果を 1番地に格納す
るプログラムである。

01000_000_00000000 // 0 : LDI r0,#0

00000_001_000_01001 // 1 : LD r1,(r0)

01000_000_00000001 // 2 : LDI r0,#1

00000_010_000_01001 // 3 : LD r2,(r0)

00000_001_010_00110 // 4 : ADD r1,r2

00000_001_000_01000 // 5 : ST r1,(r0)

00000_000_000_00000 // 6 : NOP

ところがこれを上記のパイプラインで動かすと、そもそも LD命令がうまく動かず、
全く答えが出ないことがわかる。なぜだろう？
これは、最初の命令でセットする r0の値（ここでは 0）をすぐ次のステージで利用

するためである。r0に 0が格納されるのは、WBステージである。しかし、LD命令
は、LDI命令がまだ EXステージに居る時に、レジスタファイルから値を読み出して
しまう。すなわち、r0が 0になる前の古い値を使ってしまうことになる。この様子を
図 9.9に示す。
このようにパイプラインの動作を妨げる障害のことをパイプラインハザードと呼

ぶ。ここで問題のハザードは、先行した命令の計算結果をすぐ次の命令が使うことか
ら起きる。すなわちデータの依存性が問題になることからデータハザードと呼ばれる。
データハザードを解決するためには、命令間に距離を置けば良い。図 9.10に示すよう
に、依存関係がある二つの命令間に NOP(あるいは関係のない命令)を置いてやれば、
正しく動く。前回のプログラムがうまく動いたのは実はこの関係が満足されていたた
めで、何だかプログラムが不自然だったのはこのためである。この場合のNOPは、パ

図 9.9: データハザード

イプラインの泡（バブル）と呼ばれる。バブルにより、パイプラインは停止（ストー
ル）し、性能が低下する。

例 題 9.2: 前の命令の計算結果を次の命令が利用する可能性が 80%とする。この依
存関係を NOP命令を二つ入れて対処する場合、性能低下はどの程度になるか。

答 : 理想の CPIを 1とすると、NOPによる泡は 2クロックになるため、1+0.8*2=2.6
となる。
これでは CPIは 2を上回ってしまい、いくら周波数が高くてもパイプライン処理の

利点が台無しである。

図 9.10: パイプラインのバブル

データハザードを解決すための手段はデータのフォワーディング (横流し)である。
LD命令で利用したい値 (0)は、一つ先の EXステージで先行命令によって計算済み
である。ただ、これがレジスタファイルに書き込まれるのが 2クロック先になってし
まうのが問題である。そうかといってレジスタファイルはすでに 3ポート構成であ
り、これ以上様々なステージからデータを書き込むために入力を付けることはできな
い。そこで、計算したての値をそのまま使えるように横流ししてしまおう、というの
がフォワーディングのアイディアである。まずは、レジスタファイルにフォワーディ
ングを設ける。すなわち、書き込みイネーブル信号が出されていて、書き込みレジス

タ番号と読み出しレジスタ番号が一致したら、レジスタ書き込む値をそのまま出力し
てしまうようにする。
これで、図 9.11に示すように書き込みを行う先行命令がWBステージに存在すれ

ば、その書き込みデータは IDステージで読めるようになり、NOPは一つに減る。

図 9.11: レジスタファイルのフォワーディング

次に同じような方法を EXステージの命令についても適用しよう。図 9.12に示す 2
つの方法が考えられる。最初の方法 (図 a))では、計算済み、あるいはデータを読ん
できた後にレジスタに入れた値をWB ステージから EXステージに対してそのまま
ALUの入力や LD命令のアドレスとして使えるように横流しする。
これに対して、二番目の方法 (図 b))では、ALUの出力、データメモリの出力を EX

ステージから IDステージに対して横流しし、レジスタファイルの値と入れ替えて areg,
bregに格納する。
二番目の方法は、遅延パスが長くなるが、後で分岐命令を付け加える際に楽なので、

今回は、こちらの方法を採用する。この方法では、IDステージを以下のように改造
する。すなわち、

• EXステージの命令の rdフィールドと IDステージの rdフィールドが同じで、

図 9.12: フォワーディング

EXステージの命令がレジスタに結果を書き込む命令（つまり ST命令や分岐命
令でない）である場合、EXステージの結果を横流しして、aregに格納する。

• EXステージの命令の rdフィールドと IDステージの rsフィールドが同じで、
EXステージの命令がレジスタに結果を書き込む命令（つまり ST命令や分岐命
令でない）である場合、EXステージの結果を横流しして、bregに格納する。

EXステージの結果は LD命令かどうかによって持ってくる場所が違うのでこれを判
別して切り替える必要がある。パイプラインの全体構成を図 9.13に示す。aregと breq
の前のマルチプレクサは、図中には書けなかったが、それぞれ以下の条件で切り替
える。

• areg: rd id=rd, rwe id=1

• breg: rd id=rs, rwe id=1

図 9.13: パイプラインの全体構成

各種データハザード

実は、データハザードにはここで扱ったものを含めて以下の三種類がある。

• RAW(Read After Write)ハザード: データを書く前に、値が読み出されてしまう
問題。ここで扱った問題。

• WAR(Write After Read)ハザード: 値を読み出す前にそれが書き潰されてしまう
問題

• WAW(Write After Write)ハザード：値を書き込む前に別の値を書き込んでしま
うタイプの書き潰し。一般的には、メモリやレジスタに対しては生じない（書
いた値を読む前に次の値を書くことはないはずなので、先にWAR問題が起き
るはずである。このような問題が起きるとすればプログラムのバグだろう）が、
I/Oに対するデータ転送などで問題になる。

現在の POCOでは、レジスタファイルへの書き込みは最終ステージである wbステー
ジで行われるため、WAR ハザードと WAWハザードは生じない。問題となるのは、
RAWハザードのみであるもちろん、RAR(Read After Read)はハザードを生じない。

9.3.2 コントロールハザード

今までの POCOのパイプラインには分岐命令がなかった。これを付け加えてみよ
う。以前同様、分岐先のアドレスを計算させるには PCに 1を足す時に使った加算器
を使う。また、レジスタの値が 0かどうかを比較する比較器を IDステージに取り付
ける。ここで、レジスタの値は、EXステージからフォワーディングされて来たもの
を使う点に注意しよう。こうしないと直前の命令でレジスタの値が変化した場合、比
較に使えないことになる。分岐命令を取り付けたパイプラインを図 9.14に示す。
ここでは、左に書いた命令メモリのアドレスに以下の命令が格納されているとする。

3 ADD r4,r2

4 ADDI r3,#-1

5 BNZ r3, -3

6 ADDI r0,#1

ここで、ADDI命令は EXステージに居り、BNZ命令が IDステージに居るケース
を考える。ここでは、図中に示すように直前の命令で計算された結果 (r3-1)も、フォ
ワーディング機構を使って分岐命令での判定に用いることができるようになっている。
ここで、判定の結果 r3が 0ではなく、分岐が成立する場合を考えよう。pcsel=1とな
り、6-3が計算されて飛び先アドレスの 3が PCから現れる。しかし、図 9.15に示す
ように、これが PCに格納されるのは、次のクロックである。すなわち、6番地の命
令は分岐が成立した場合でも、パイプラインに入って来てしまうのだ。この命令は本
来分岐が成立した場合は実行してはならないので、これは問題である。このような問

題は前回紹介したハザードの一種であり、分岐すなわちプログラムの制御によって生
じることから制御ハザードと呼ぶ。
制御ハザードを解決するもっとも簡単な方法は、IDステージで分岐命令と分かった

ら、強制的に IFステージで命令をフェッチするのを止めてしまうことである。ハザー
ドを防ぐためにパイプラインを止めることをパイプラインストールと呼ぶ。IFステー
ジの場合、止めるのが面倒な場合は、NOP命令と入れ替えてしまう方法も考えられ、
POCOでは NOPは機械語の全 bitを 0にすれば良いので楽である。しかし、いずれ
の場合でも、パイプラインには 1クロック分泡が入って、パイプラインの性能が低下
する。

例 題 9.3分岐命令の後の命令を全て NOPに置き換えるとすると、これにより CPI
はどの程度増加するか？ただし、分岐命令の存在確率は 25%となる。

1+0.25*1=1.25となる。
ここで、少し頭を使おう。分岐命令が成立しない場合は、フェッチした命令はその

まま使えるので、NOP命令に変更する必要はない。したがって、分岐命令が成立す
る場合だけフェッチした命令を NOPに変更する。このためのハードウェアは成立す
るかどうか示す信号は作ってあるので大変簡単である。
この方法は、分岐が成立しないと予測して命令をフェッチしてしまうとも考えられ

るので、もっとも簡単な分岐予測とされており predict-not-taken(飛ばない方に賭ける)
と呼ばれる。残念なことに、多くの分岐命令では、飛ぶ確率は飛ばない確率よりも高
い。これはループを形成するには前方で飛ぶ必要があるからだ。ここで、飛ぶ確率を
70%と仮定すると、predict-not-takenを使った場合の CPIは

1 + 0.25× 0.7)× 1 =　 1.175

と若干改善できる。

9.3.3 遅延分岐

もう一つ、もっと簡単な方法がある。この方法はハードウェアに全く手を加えず、
考え方のみ変える。「この CPUの分岐命令は効き目が遅く、飛ぶのが一命令分遅れる」
と考えるのである。そして、遅れた間の一命令分はパイプラインに入って普通に実行
されると考える。この方法を遅延分岐 (delayed branch)と呼び、効き目がでるまでの
間のことを遅延スロットと呼ぶ。ここで遅延スロットには分岐が成立しようがしまい
が実行される命令を埋めておけば性能は低下しない。例えば、以前紹介した乗算のプ
ログラムを考える。

LDI r0,#0

LD r3,(r0)

ADDI r0,#1

LD r2,(r0)

LDI r4,#0

loop: ADD r4,r2

ADDI r3,#-1

BNZ r3,loop

遅延スロット

ここで使われている BNZが遅延分岐である場合、遅延スロット内の命令はループ
内に入るので、ループを一回分回るのに 4命令を要することになる。ここで ADD命
令を遅延スロットに移動してみよう。

LDI r0,#0

LD r3,(r0)

ADDI r0,#1

LD r2,(r0)

LDI r4,#0

loop: ADDI r3,#-1

BNZ r3,loop

ADD r4,r2

このプログラムは見るからに変だが、ADD命令はちゃんとループ内で実行し、正
しい結果が得られることがわかる。しかもループを 1回まわるのは 3命令で済む。実
際に Verilogシミュレーションでこのことを確かめよう。
遅延スロットは、上記のように、分岐が成立しても成立しなくても実行することが

必要な命令をもってくるのが良い。しかし、分岐が成立した時だけ有効で、成立しな
い場合に実行しても害がないもの、逆に分岐が成立しない時だけ有効で、成立した場
合に実行しても害がないもので埋めても良い。このように性能を上げるために、命令
を移動することを命令スケジューリングあるいはこの場合はパイプラインスケジュー
リングと呼ぶ。どうしても埋まらない場合は NOP命令を入れておく。
命令スケジューリングは多くの場合、コンパイラで行う (Cのコンパイラのオプショ

ンを変えるとスケジュールを実行してくれる)。ここで、遅延スロットがどうしても
埋まらない可能性を 20%とすると、CPIは以下のようになる。

1 + 0.25× 0.2)× 1 =　 1.05

この場合、理想の CPIである 1にかなり近づいたといえる。

演習 9.2

1から nまで加算するプログラムに関して遅延分岐を想定してパイプラインスケジュー
リングを行え。

9.4 パイプライン処理のVerilog記述
前回シミュレーションしたパイプライン化 POCOの Verilog記述を検討しよう。本

来、パイプライン構造は、それぞれのステージをモジュールとして宣言して設計する
のがモジュラリティが高くでお勧めである。しかし、今回は、全体として記述量が少
ないので単一モジュール内にベタで記述した。Verilogはどこで信号名を宣言しても
良いので、各モジュールでローカルに用いる信号はそのモジュール内で極力定義して
いる。ただし、前のステージに戻る信号線は、最初に出てきたモジュールで宣言して
おり、必ずしもそのモジュールで主として利用する信号でない場合もある。前回の図
と信号線名を揃えておいたので、図と記述を見比べて信号を把握して欲しい。

図 9.16: POCOのパイプライン構造

パイプライン化した pocoのモジュール名を pocopとした。入出力は今までと同じ
なので省略した。

9.4.1 IFステージ

まずは命令フェッチ (IF)のステージである。

// Instruction Fetch //

reg [‘DATA_W-1:0] pc;

reg [‘DATA_W-1:0] ir;

assign iaddr = pc;

always @(posedge clk or negedge rst_n)

begin

if(!rst_n) pc <= 0;

else

pc <= pc+1;

end

always @(posedge clk or negedge rst_n)

begin

if(!rst_n) ir <= 0;

else ir <= idatain;

end

これは、毎クロック、pcの指し示す命令メモリから命令をフェッチして irに格納す
る。今の所、分岐命令は付けてないので、大変簡単である。

9.4.2 IDステージ

次は、命令デコード (ID)のステージである。このステージでは制御信号を生成す
ると共に、ir中のレジスタ番号に従ってレジスタファイルからレジスタを読み出す。

// Instruction Decode & Register Fetch //

wire st_op, addi_op, ld_op, alu_op;

wire ldi_op, ldiu_op, ldhi_op, addiu_op;

wire [‘SEL_W-1:0] com;

wire [‘OPCODE_W-1:0] opcode;

wire [‘OPCODE_W-1:0] func;

wire [‘REG_W-1:0] rs, rd;

wire [‘IMM_W-1:0] imm;

wire [‘DATA_W-1:0] rf_a, rf_b, rf_c, alu_b;

reg [‘SEL_W-1:0] com_id;

reg st_op_id, ld_op_id;

wire rwe;

reg rwe_id;

reg [‘REG_W-1:0] rd_id;

reg [‘DATA_W-1:0] areg, breg;

assign {opcode, rd, rs, func} = ir;

assign imm = ir[‘IMM_W-1:0];

assign st_op = (opcode == ‘OP_REG) & (func == ‘F_ST);

assign ld_op = (opcode == ‘OP_REG) & (func == ‘F_LD);

assign alu_op = (opcode == ‘OP_REG) & (func[4:3] == 2’b00);

assign ldi_op = (opcode == ‘OP_LDI);

assign ldiu_op = (opcode == ‘OP_LDIU);

assign addi_op = (opcode == ‘OP_ADDI);

assign addiu_op = (opcode == ‘OP_ADDIU);

assign ldhi_op = (opcode == ‘OP_LDHI);

assign alu_b = (addi_op | ldi_op) ? {{8{imm[7]}},imm} :

(addiu_op | ldiu_op) ? {8’b0,imm} :

(ldhi_op) ? {imm, 8’b0} : rf_b;

assign com = (addi_op | addiu_op ) ? ‘ALU_ADD:

(ldi_op | ldiu_op | ldhi_op) ? ‘ALU_THB: func[‘SEL_W-1:0];

assign rwe = ld_op | alu_op | ldi_op | ldiu_op | addi_op |

addiu_op | ldhi_op ;

rfile rfile_1(.clk(clk), .a(rf_a), .aadr(rd), .b(rf_b), .badr(rs),

.c(rf_c), .cadr(rd_ex), .we(rwe_ex ));

always @(posedge clk ) begin

st_op_id <= st_op; ld_op_id <= ld_op; rwe_id <= rwe;

areg <= rf_a; breg <= alu_b;

com_id <= com; rd_id <= rd; end

レジスタファイルから読み出した値はAポートの方はそのまま regaに格納して EX
ステージに送る。Bポートの方は、イミーディエイト命令では利用しないので、命令
に応じて ALUの B入力 (alu b)を生成してやる。さらに ALUの comも生成する。こ
の生成の仕方、命令デコードの仕方は、パイプライン化されていない判と同じであ
る。前回解説したように、生成した信号のうち、実行 (EX)ステージで利用するもの
についてはレジスタに格納して送ってやる。末尾に idを付けた信号線がこれに当た
る。レジスタファイルの書き込み部である rf c, ed ed, rwe exは前回述べたようにWB
ステージからの信号線である。

9.4.3 EXステージ

お膳立ては IDステージで終わっているので、EXステージは、ALUで演算を行う
と共に、データメモリをアクセスする。計算結果は cregに、メモリの読み出したデー
タは dregに入れて次のWBステージに送る。LD命令かどうか、レジスタファイル
に書き込むかどうか (rwe)、書き込む際の番号 (rd)は再びレジスタに格納してWBス
テージに送る。

// Execution

wire [‘DATA_W-1:0] alu_y;

reg [‘DATA_W-1:0] creg, dreg;

reg ld_op_ex ;

reg [‘REG_W-1:0] rd_ex;

reg rwe_ex;

alu alu_1(.a(areg), .b(breg), .s(com_id), .y(alu_y));

assign ddataout = areg;

assign daddr = breg;

assign we = st_op_id;

always @(posedge clk ) begin

creg <= alu_y;

dreg <= ddatain;

ld_op_ex <= ld_op_id;

rd_ex <= rd_id;

rwe_ex <= rwe_id; end

9.4.4 WBステージ

たった一文しかないが、これは、レジスタファイルに書き込むデータとして LD命
令ならば、dregをそうでなければ cregを選ぶための記述である。

// Write back

assign rf_c = ld_op_ex ? dreg : creg;

後の記述は、実は IDステージ内にあり、EXステージでレジスタに格納された信号
線に従って、この rf cがレジスタファイルに書き込まれる（ST命令などでは書き込
まない）。

// Instruction Decode and Register Fetch

...

rfile rfile_1(.clk(clk), .a(rf_a), .aadr(rd), .b(rf_b), .badr(rs),

.c(rf_c), .cadr(rd_ex), .we(rwe_ex ));

図に書くより Verilog記述を示した方が早いだろう。rfileの記述を以下のように変
更すれば良い。

assign a =

(aadr == cadr)&we ? c:

aadr == 0 ? r0:

aadr == 1 ? r1:

aadr == 2 ? r2:

aadr == 3 ? r3:

aadr == 4 ? r4:

aadr == 5 ? r5:

aadr == 6 ? r6: r7;

assign b =

(badr == cadr)&we ? c:

badr == 0 ? r0:

badr == 1 ? r1:

badr == 2 ? r2:

badr == 3 ? r3:

badr == 4 ? r4:

badr == 5 ? r5:

badr == 6 ? r6: r7;

これに対応する Verilog 記述は、ID ステージの記述を下記のように書き換える。
fwdataが図中には EXステージにあるマルチプレクサであり、（EXの所に書いてもよ
かったが分かりやすさを重視してこちらに書いた）フォワーディングすべきデータが
ALUからか、メモリからかを選択する。

fwdaは、aregの前に設けた rdに対するフォワーディングを行うためのマルチプレ
クサであり、先行命令の rd idと現在 IDステージに居る命令の rdを比較して等しく、
かつ先行命令が書き込みを行えば (rwe id)フォワーディングを行う。b入力 (alu b)の
方は、図とは異なり、マルチプレクサを一体化して記述している。b入力に対するフォ
ワーディングは、R型の命令だけに有効にする必要がある。マルチプレクサを一体化
することで、イミーディエイト命令等ファワーディングをしてはならない命令を最初
に選び出してやることで、設計を簡単化している。

assign fwddata = (ld_op_id) ? ddatain: alu_y;

assign alu_b = ((rd_id == rs)& rwe_id) ? fwddata :

(addi_op | ldi_op) ? {{8{imm[7]}},imm} :

(addiu_op | ldiu_op) ? {8’b0,imm} :

(ldhi_op) ? {imm, 8’b0} : rf_b;

assign fwda = ((rd_id == rd)& rwe_id) ? fwddata: rf_a;

...

always @(posedge clk or negedge rst_n ) begin

if(!rst_n)

rwe_id <= ‘DISABLE;

else begin

st_op_id <= st_op; ld_op_id <= ld_op; rwe_id <= rwe;

com_id <= com; rd_id <= rd;

areg <= fwda; breg <= alu_b;

end

end

他に初期化と NOP命令の識別を行ったコードを web上に示す。これで先に示した
プログラムが NOP無しに実行できることを確かめよう。

9.4.5 遅延分岐のVerilog記述

遅延分岐を使う場合、パイプラインに細工をする必要がないため、分岐命令を付け
加えるのは楽である。まず IFステージ内の pcの設定を以下のように書き直す。

always @(posedge clk or negedge rst_n)

begin

if(!rst_n) pc <= 0;

else if(pcsel)

pc <= pc +{{8{imm[7]}},imm} ;

else

pc <= pc+1;

end

次に IDステージに分岐命令のデコードと、pcselの制御を付け加える。

assign bez_op = (opcode == ‘OP_BEZ);

assign bnz_op = (opcode == ‘OP_BNZ);

...

assign pcsel = (bez_op & fwda == 16’b0 ) | (bnz_op & fwda != 16’b0) ;

ここで、fwdaはフォワーディング後の aregに蓄えるデータである。前回に付けた
マルチプレクサからのフォワーディングデータをそのまま利用している。

演習 9.3

JALR命令をパイプライン化された POCOに実装せよ。



161

第 10

記憶の階層とキャッシュ

10.1 記憶の階層

POCOは単純な構造を持っているので、最近の半導体プロセス (65nm)を用いて論
理合成すると 300MHz程度で動作可能となる。実際には配置配線の段階で配線遅延の
影響で動作周波数が低下するが、商用のマイクロプロセッサでは組み込み用途であっ
ても 200MHz以上の周波数で動作するものが多い。これがデスクトップ用、ラップ
トップ用ともなると動作周波数は 3GHzに達する。
一方、主記憶の容量は 1MIPS当り 1Mbyte必要であると言われており、現在の高性

能プロセッサの主記憶は数百Mbyteの容量を必要とする。しかし、これに追従するよ
うなアクセス速度と容量を持つメモリは存在しないか、あるいはひどく高価なものと
なる。そこで、高速／小容量のメモリと低速／大容量のメモリを組み合わせ、記憶の
階層を構成する。図 10.1に記憶の階層の構成例を示す。
コンピュータのメモリシステムは、CPUの近くから、順に高速なメモリを階層的に

接続し、高速なメモリ上にデータ（命令）が存在しなければ、階層が上の、低速だが
より大容量なメモリにデータを取りにいく。よく使うデータ（命令）がうまく高速小
容量なメモリ内に入れば、低速なメモリに取りにいく確率を小さくすることができる
ので、仮想的に、みかけは大容量で高速なメモリシステムを実現することができる。
これが記憶の階層である。
現在の多くのコンピュータは、以下の階層構造を持っている。

• L1キャッシュ　オンチップ SRAM　 4K-32K 1クロックでアクセス可能

• L2キャッシュ　オンチップ SRAM またはオンボード SRAM 　 64K-1M 10ク
ロック程度

• (L3キャッシュ　オンボード SRAM　 256K-4M 50-100クロック程度)

• メインメモリ　 DRAM 64M-4G 100クロック程度



162 第 10章 記憶の階層とキャッシュ

.....

Main Memory
10−100 clocks
64Mbyte−1Gbyte

On−board
Cache

On−chip
Cache

1−5Clocks
4Kbyte−32Kbyte

CPU

2ndary 
Memory

Operating
System

LSI
Chip

5−50Clocks
128Kbyte−1Mbyte

図 10.1: 記憶の階層

• 補助記憶　ディスク　 4G-1T 10msec程度

キャッシュは、記憶階層の最も CPU側に位置し、特に頻繁にアクセスする命令や
データを格納しておく高速小容量のメモリである。最近のプロセッサでは、図 10.1に
示すように、チップ内に L1キャッシュ(オンチップキャッシュ)、チップ外に L2キャッ
シュ(オンボードキャッシュ)を持ちキャッシュ自体が、階層構成になっている場合が
多い。

1次キャッシュはチップ内に置くため、容量は制限され、4K-32KByteであるが、命
令／データを分離した構成を取ることができる。小容量のオンチップキャッシュは 1
クロックでアクセスできるが、容量が大きいものだとチップ内でも数クロックから 10
クロック程度時間がかかる場合もある。一方、2次キャッシュは、チップ外に置くため、
64K-1MByteの容量の大きなメモリを用いることができるが、アクセス時間は大きく、
場合によっては数 10クロックになる。キャッシュは、高速で使い易い SRAM(Static
RAM)素子を用いる。SRAM素子は簡単な構造を持つ Flip Flopの 2次元配列状の記
憶領域を持ち、10nsecから 80nsec程度の時間で読み書き可能である。
これに対して主記憶（メインメモリ）は、チップ当たりの記憶容量の大きいDRAM(Dynamic

RAM)素子を用いる。DRAM素子は、チップ内にコンデンサを持ち、ここに電荷が
蓄えているかどうかで情報を記憶するため、1チップ当たり大きな情報量を記憶する
ことができる。最近の DRAM素子は 1チップ当たり 64Mbitの記憶が可能である。し
かし、電荷は情報をアクセスしたり、時間がたつと放電してしまうため、読み出し用
のプリチャージや、再充電のためのリフレッシュ操作が必要が必要で、使い難い。最

近の DRAM素子は同期型 DRAMや Rambus型 DRAMが用いられるが、これらの素
子は、最初のデータをアクセスするまでには時間がかかるが、連続読み書きを高速に
行うことができる。
通常、キャッシュはCPUから見て透過 (Transparent)である。すなわち、CPUはキャッ

シュ中にアクセスしたデータが存在しない場合に、プログラムで何かしなければなら
ないわけではなく、キャッシュ制御用ハードウェアが自動的にアクセスしたデータが
主記憶からキャッシュ中に転送されるまで CPUの命令実行は待ち状態になる。した
がって、CPUはキャッシュの存在を意識しないで動くことができる。このうち、メイ
ンメモリまでは、ハードウェアで管理され、ソフトウェアからは透過（トランスペア
レント）である。つまり、プログラムにとっては、メモリのどの階層から取ってくる
かは気にしなくてよい。
一方、補助記憶は、OS(Operating System)によって制御される。ディスクに代表さ

れる補助記憶の利用法は二つある。

• ファイルシステム

• 仮想記憶のスワップ領域

ファイルは、ユーザに対しても見える形で、管理するメモリ領域であり、純粋に OS
の範囲である。一方、仮想記憶 (Virtual Memory)は、メインメモリに載らないような
大きなプログラムや、多数のジョブを動作させるための機構であり、ハードウェアは
論理アドレスと物理アドレスの変換およびメインメモリに載っているかどうかの判定
のみを行う。
このような記憶の階層がうまく働くためには以下の２つの性質がなければならない。

• 時間的局所性: 一度アクセスした場所は、近いうちに再びアクセスされる可能
性が高い。

• 空間的局所性: 一度アクセスされた番地に近い番地が再びアクセスされる可能
性が高い。

命令フェッチにおけるアクセスは、この両者の局所性を強烈に持っている。すなわち、
多くのプログラムでは、多くの時間帯で、比較的小さなループを回っているのだ。一般
に、プログラムの実行時間の 90%は、コードの 10%が占めているという 10%-90%の
法則があるが、これは局所性を表したものといえる。データに対するアクセスは命令
程強烈ではないが、巨大な行列を扱う科学技術演算以外は、かなり存在する。

10.2 キャッシュの基本

10.2.1 キャッシュのマッピング

主記憶のアドレスが CPUから出力された時に、それがキャッシュのどこに対応す
るかを決めてやる必要がある。この番地の対応付けのことをマッピングと呼ぶ。プロ
セッサからのアクセスが発生した時に、その番地からアクセスされたデータがキャッ
シュ上に存在するかどうかを判別し、存在する（ヒットする）場合はキャッシュから
アクセスし、存在しない（ミスヒットまたはミスする）場合は主記憶からアクセスし、
キャッシュの内容を更新する等を行なう必要がある。この判断は高速に行なわなけれ
ばならないので、テーブルを順に検索したりするわけにはいかない。
さて、キャッシュは、管理上のオーバヘッドを少なくするため、一定サイズの、連

続した番地のデータを一括管理する。これをキャッシュブロックまたはラインと呼ぶ。
ちなみに 8章で導入したバイトアドレスで説明すべきだが、簡単のため、16bitアド
レスで説明する。ここで、POCOは 16ビットマシンなので、ここでは 16ビットのこ
とをワードと呼ぶ。(世の中は 32ビットマシンが多いので 32ビットをワードと呼ぶ
ことも多いので注意されたい。)
キャッシュのブロックサイズは 8から 64ワード程度が多い。ここでは 1ブロック

が 8ワードと考えよう。キャッシュも主記憶もブロック単位で区切られている。つま
り、主記憶のアドレスの中で、下位 3bitは、ブロック内に 8つあるワードのどの位置
を表すかを示すのに使われる。この部分をブロック内アドレスと呼ぶ。
ここで、非常に小規模の主記憶 (1k× 16ビット)、キャッシュ(64× 16ビット)間

の割りつけを考えてみる。主記憶は 1k(210)ワードすなわちアドレス 10bit分、キャッ
シュは全体として 64(26)× 16ビットとする。ブロックサイズは 8なので、キャッシュ
は 64/8=8ブロック分を保持することになり、主記憶中には 1k/8=128ブロックを保持
することになる。この例はあくまで説明のためのもので、前節に述べた現在の主記憶、
キャッシュの容量を考えると、まったく小規模なものである点に注意されたい。しか
し、原理は全く同じで実際の CPUではビット数が長くなるだけである。

ダイレクトマップ方式

主記憶は順にブロック単位に分割されていると考える。これに図 10.2に示すように
0(0000000) から 127(1111111)まで番号を降り、10bitのアドレスの上位 7bit で表す
ことにする。下位 3bitはブロック内アドレスである。一方、キャッシュ内のブロック
は 8個であり、同様に 0から 7まで番号を降る。そこで、まず主記憶の 0(0000000)
から 7(0000111)までをキャッシュの 0から 7まで順に割りつける。次に 8(00001000)
から 15(00001111)までを同様に 0 から 7 までに割りつけ、さらに 16(00010000) か
ら 23(00010111) までを 0 から 7 までに割りつけ、以下最後に割りつけて、最後の
120(11111000)から 127(11111111)を 0から 7までに割りつける。この方式だと、ア

ドレスの上位 7bitのなかで、下の 3ビットがキャッシュブロックのどの場所に格納さ
れるかを示すことになる。この下位 3ビットをインデックスと呼ぶ。インデックスで
表される 8つのキャッシュに対して、それぞれ主記憶上のどのブロックが入っている
かを示すためには、残った 4ビットを記憶すれば良い。これをタグあるいはキーと呼
ぶことにする。

0000000000

0000001000

0000010000

0000011000

0000100000

0001010000

Block:0000000

Block:0000001

0000101000

0000110000

0000111000
Block:0000111

0001000000
Block:0001000

0001001000
Block:0001001

1111111000

Block:1111111

.....

Block:000

Block:001

Block:010

Block:011

Block:100

Block:101

Block:110

Block:111.....

IndexTag

Main Memory 128-Block

Cache 8-Block

図 10.2: ダイレクトマップ方式

ダイレクトマップでは、図 10.3に示すように、インデックスで表されるキャッシュ
のそれぞれのブロックに対して 4ビットのタグメモリを設けて、そこにどの主記憶上
のブロックが入っているかを記憶させておく。

CPUから読み出しアクセスが行われた時、キャッシュは以下のように働く。(1)タ
グメモリをインデックス (010)をアドレスとして参照し、(2)テーブルから得られたタ
グ (0000)と CPUからアクセスされたアドレスのタグ部を比較する。等しければキャッ
シュ中にアクセスされたブロックが存在 (ヒット)し、等しくなければ存在しない（ミ
スヒット）ことがわかる。(3)ヒットすれば、キャッシュからデータを読み出し、(3’)
ミスヒットすれば、主記憶からキャッシュを転送して、インデックスで示す位置 (この
場合 010)に格納して、タグメモリのインデックスで示される番地 (この場合 010)に
アドレスのタグ部 (0000)を格納してから、アクセスを行う。
この方法では、キャッシュ中の格納されるブロックの総数に等しいエントリのタグ

メモリを持てば良い。インデックスを使ってタグメモリと同時にキャッシュをアクセ

図 10.3: ダイレクトマップ方式のキャッシュ

スできる。タグメモリから読み出した値と比較してミスヒットした場合は、読み出し
た値を捨ててしまえば良い。
しかし、この方法では、キャッシュ中に格納される位置が一意に決まってしまうた

め、インデックスが等しい 2つのブロックは、たとえキャッシュの他の位置が空いて
いたとしても、同時にキャッシュ中に格納することができなくなる。この例では、ブ
ロック 0000010が格納された状態では、ブロック 0010010は、インデックス 010が一
致してしまうために、格納することができない。したがって、ブロック 0010010をア
クセスした場合は、ミスヒットとなり、主記憶からブロックを取ってきて入れ替える
必要がある。このようにインデックスが重なったことによって起きるミスをインデッ
クス衝突ミス (conflict miss)と呼ぶ。インデックス衝突ミスは、特にキャッシュサイズ
が小さく、キャッシュに格納可能なブロック数が少ない場合に発生し、性能を阻害す
る原因となる。

セットアソシアティブ方式

同じ容量のキャッシュでも、図 10.4に示すようにインデックスの bit数を 1bit減らせ
ば、同一のインデックスに対して格納可能なブロック位置を 2箇所用意することがで
きる。2箇所のペアで同じ番号を割り当てているものをセットと呼ぶ。
この場合、キャッシュは図 10.5に示すようにタグメモリを 2組用意し（インデック

スが 1bit減るのでエントリ数は半分になる）、それぞれ対応するキャッシュに格納さ
れたブロックのタグを格納する。このことにより、キャッシュ上半分のインデックス
10にはブロック 00100を、下半分のインデックス 10にはブロック 00000を格納する
ことが可能となる。このように分割したキャッシュの１まとまりを wayと呼ぶ。この

図 10.4: 2-wayセットアソシアティブ方式

例では 2-wayに分割したことになる。CPUがアクセスした場合、今度は同一のイン
デックスで同時に２つのタグメモリを参照し、得られたアドレスと、アドレスのタグ
部を同時に比較する。そして、一致した wayのキャッシュに対しデータの読み書きを
行う。両方共一致しなかった場合は、ミスとなり、主記憶からどちらかの wayにブ
ロックを転送する必要が生じる。
この方法は、キャッシュを 2-wayに分割することから 2-wayセットアソシアティブ

方式と呼ぶ。ダイレクトマップ方式に比べ、ブロックの格納可能な位置が倍になるた
めインデックス衝突ミスを減らすことができる。一方で、タグメモリと比較器が２組
必要で、アクセスされたキャッシュからデータを CPUに送り出すパスにもマルチプ
レクサが必要になる。２つのタグメモリは同時に参照可能であるが、構造が複雑なた
め、全体としてダイレクトマップ方式に比べてアクセスのための遅延が大きくなる傾
向にある。
ちなみに、2-wayにしたからといってタグメモリが 2倍必要なわけではない。エン

トリ数が半分なものが 2組必要になるわけで、メモリ要求量の増加はタグの増加分
1bit分で済む。この考え方を拡張し、キャッシュを n個に分割して、タグメモリと比
較器を n組持たせれば、n-wayセットアソシィアティブ方式を実現することができる。
図 10.6は 4-wayセットアソシアティブキャッシュを示す。この例では、インデックス
がさらに減り 1ビットになっている。ちなみに、ダイレクトマップは、1-wayセット
アソシアティブキャッシュである。
ダイレクトマップは 16人が 1つだけの椅子を争う椅子取りゲームである。2-way

図 10.5: 2-wayセットアソシアティブ方式のキャッシュ

は椅子が二つに増えるが、参加者の数も倍になって 32人になってしまう。これでは
確率が同じないのではないか、と思うかもしれないが、実はそうでもない。あなたが
椅子が欲しいタイミングにちょうど椅子を要求するしつこいライバルが一人居るとし
よう。椅子が一つしかなければ毎回熾烈な争いをしなければならないが、椅子が二つ
あれば多くの場合は平和共存できるのである。これがインデックス衝突ミスを減らす
効果である。一般にダイレクトマップを 2-wayにするとヒット率が、キャッシュの容
量を倍にしたのと同じ位の上がると言われている。もちろん動作させるプログラム、
キャッシュの容量にも依存する。一般にキャッシュの容量が小さいほど、way数を増
やす効果がある。nの数を増やすことによるインデックス衝突ミスを減らす効果は n
が大きくなる程頭打ちになり、十分な大きさのキャッシュサイズに対しては nは 4よ
り大きくしてもあまり効果がないことが知られている。

フルアソシアティブ方式

2-way set associative方式を同様にもう一回、つまり 4つのブロックでセットを組めば
4-way set associative方式にすることができる。この場合、インデックスは 1bitにな
り、キーが 6bitとなる。それでは、さらに折り畳むとどうなるか？
ここでは、キャッシュ容量が非常に小さいため、図 10.7に示すように、すべての

キャッシュ領域に対してブロックの格納が可能になる。このように、どこにでも格納
が可能な方式をフルアソシアティブ方式と呼ぶ。この方式を実現するには、大量の比
較器を用いるか、メモリ中のデータを並列に検索する連想メモリ (associative memory,
content addressable memory)が必要になる。したがって、このフルアソシアティブ方
式は、キャッシュ容量が非常に小さい場合にのみ現実的な方法で、キャッシュよりも、

図 10.6: 4-wayセットアソシアティブ方式

後に紹介する TLB(Translation Look aside Buffer)に用いられる。

演習 10.1 poco の全アドレス空間は 64Kwords であり、この空間すべてに対して
1Kwordsのキャッシュを設ける。キャッシュブロックを 8wordsとする時、以下の方式
でインデックス、タグのビット数はそれぞれどうなるか？

1. ダイレクトマップ方式

2. 2-wayセットアソシアティブ方式

3. 4-wayセットアソシアティブ方式

10.2.2 書き込みの制御

読み出しがヒットした場合は、単にキャッシュからデータを読み出して CPUに送れ
ば良い。読み出しがミスすると、通常、主記憶からキャッシュブロックをキャッシュ
へ転送して、今までのブロック上に上書きする。この操作を置き換え（リプレイス）
と呼ぶ。
問題は、キャッシュがミスした場合である。この時の制御がキャッシュの方式を分

けることになる。
キャッシュの制御方式は次の二つの選択肢がある。

• ライトスルー（Write Through）あるいはストアスルー (Store Through): 書き込
みデータをその都度主記憶に書き込む方法

図 10.7: フルアソシアティブ方式

• ライトバック（Write Back）あるいはストアイン (Store In): 書き込みデータは
キャッシュのみに格納し、その場では主記憶には書き込まない方法

ライトスルー（Write Through）方式: ライトスルー方式は、図 10.9に示すように、
キャッシュに書いたデータが常に主記憶に反映されるので、キャッシュブロックが追
い出される時に主記憶に対して書き戻してやる操作は不要だが、常に書き込みを行な
うため、キャッシュ／主記憶間のバスが混雑する。

...

Main Memory

Cache Memory

Write data

図 10.8: ライトスルー方式

CPUはデータの書き込みに関しては、終了を待たずに次の命令に進むことができ
るので、書き込みデータを主記憶に書き込む際に仮に格納してやる簡単なメモリ（ラ

イトバッファ）を設け、連続したアドレスをまとめて書き込むマージ機構をきちんと
作ってやれば、CPUの実行時間に悪影響を与えることは防ぐことができる。ただし、
連続したアドレスに多数のデータを書き込んで、ライトバッファが溢れる場合には、
書き込みが行えなくなって、CPUが待ち状態になる場合がある。
ライトスルー方式は、書き込み時にミスした場合については、直接主記憶のみを更

新する方式 (ダイレクトライト：Direct Write)と、そのブロックを一度キャッシュ上に
取ってきてから書き込む方法（フェッチオンライト：Fetch on Write)がある。前者は
実装が簡単であるが、次に同じ番地を Readする場合にミスヒットしてしまう。これ
に対し、後者はまずブロックを読み込んでリプレイスしてから書き込むので制御が複
雑になるが、次の Readがヒットするためヒット率を高めることができる。

ライトバック（Write Back）方式: ライトバック方式は、キャッシュに書き込んだ
データをその都度主記憶に反映せず、キャッシュのみを更新し、そのブロックが追い
出される際に主記憶に書き戻して一致を取る。このため、一定期間、主記憶上のデー
タとキャッシュの一致が取れなくなる。そこで、タグメモリ上にダーティ(Dirty) bitを
設け、主記憶上のデータと一致しているかどうかを管理する。一致している状態を
Clean、そうでない状態を Dirtyと呼ぶ。
読み出しミスが起き、ブロックを主記憶から取り込んでリプレイスすると、最初は

ブロックの状態を Cleanとする。次にそのブロックに書き込みを行うと、その書き込
みは主記憶に反映せず、状態を Dirtyに変更する。この先何度書いてもキャッシュの
みにデータが書き込まれる。これが図 10.9 a)に当たる。次に、Dirtyのブロックが、
リプレイスの対象となって、キャッシュから追い出される場合、単純に上書きすると、
今まで書き込んだデータが消失してしまう。そこで、図 10.9 b)に示すように、主記憶
に書き戻し (Write Back)を行って一致を取ってから、ブロックを上書きする必要があ
る。Cleanなブロックを追い出す場合には書き戻しの必要はなく、直接新しいブロッ
クを上書きする。書き込みミスに際しては、常にフェッチオンライト方式を取る。す
なわち、一度キャッシュ内にブロックを読み込んでから、書き込みを行う。
この方法は、ライトバッファを設けなくてもキャッシュが高速ならば性能が落ちる

ことがなく、データの連続書き込みでライトバッファが溢れる心配も不要である。し
かし、一方で、追い出しの際の書き戻しのロスに加え、主記憶とキャッシュの一致が
取れないことから、I/Oが主記憶から DMA(Direct Memory Access)転送 (以前やった
が覚えているか？)を行う場合などには、問題が生じることがある。この場合、I/Oか
ら書き戻し要求を出して強制的に書き戻しを行う必要がある。普通のプロセッサの場
合は、きちんと作ればどちらの性能がいいとも言えないが、複数のプロセッサを用い
るマルチプロセッサ方式の場合は、バスの混雑が性能に致命的な影響を与えるためラ
イトバック方式が優れている。

Main Memory

b) Miss/Write Back 

図 10.9: ライトバック方式

例題 10.2: 前回のダイレクトマップキャッシュを想定して、以下のアクセスが順に行
われたとする。ダイレクトライトのライトスルーキャッシュ、フェッチオンライトの
ライトスルーキャッシュ、ライトバックキャッシュ、はそれぞれヒットするか？

(1) 0100001000番地から読み出し　 (2) 1100001001番地に書き込み　 (3) 1100001001
番地から読み出し　

答: (1)は最初なのですべてのキャッシュでミスする。これを初期化 (Conpulsory:必
須)ミスと呼ぶ。(2)はインデックス部分の 001が重なるために、やはりすべてのキャッ
シュがミスをする。これをインデックス衝突 (Conflict)ミスと呼ぶ。この時に、ダイ
レクトライトのライトスルーキャッシュは、主記憶にデータを直接送りつけて更新す
るためキャッシュの置き換え (リプレイス)を行わない。しかし、フェッチオンライト
のダイレクトマップキャッシュとライトバックキャッシュはここで、リプレイスを行
う。したがって、(3)で同じ番地を読み込む場合、フェッチオンライトとライトバック
はヒットし、ダイレクトライトのみミスする。
この例は、フェッチオンライトとライトバックに有利な場合であるが、逆の場合も

考えられる。これが以下の演習問題である。

演習 10.2 例題と同様に場合に以下のアクセスが順に行われたとする。それぞれの
キャッシュはヒットするか？

(1) 0100001000番地から読み出し　 (2) 1100001001番地に書き込み　 (3) 0100001001
番地から読み出し　

追い出しの制御

ダイレクトライトを採用したライトスルー方式における Write ミスを除き、ミス
ヒットの際は、どこかのキャッシュのブロックをリプレイス（置き換え）するために、
領域を確保する必要がある。ダイレクトマップ方式は１つしか可能性がないので問題
ないが、セットアソシアティブ方式ではどの wayのブロックを追い出すかを決める必
要がある。この時用いられるのが、LRU(Least Recently Used)方式である。この方法
は、最近アクセスされなかった方の wayを選ぶ方式で、主記憶-ディスク間の管理に
も用いられる。2-wayを越えると、厳密にもっとも最近用いられなかった wayを選ぶ
のは難しく、最近用いられなかった wayのうちどれかを選ぶ擬似的な方法を採る場合
も多い。ちなみに、先にキャッシュに格納した方の wayを選ぶ FIFO(First In First Out)
方式は、時としてランダムに追い出す方式より性能が悪いことが知られている。

10.3 キャッシュの性能評価

メモリアクセスのロスは、今回の設計の場合、以下の式で示すように CPIを増加さ
せる。ここで、CPI(ideal)は、全メモリがクロック内でアクセスできると考えた際の
CPI(Clock cycles Per Instruction)のことである。

平均CPI = CPI(ideal)+命令キャッシュのミス率×リードミスペナルティ+LD命令の確率×データキャッシュのミス率×ミスペナルティ

読み出しミスと書き込みミスは、本来ミスペナルティが違う。書き込みミス時は、
ライトスルー方式、ライトバック方式共に簡単なライトバッファを備えることで CPU
は処理完了を待たずに先に進むことができる。したがって、簡単なモデルでは書き込
みミスを無視する場合もある。しかし、書き込みミスが引き起こすリプレイス処理を
キャッシュが行っている際に、先に進んだ CPUが次のデータを読み書きすると、ここ
で CPUを待たせなければならない事態が出てきて、この辺は複雑である。本当に正
確に性能を評価する場合は命令レベルのシミュレータを用いなければならない。

演習 10.3 ミスペナルティを命令、データ共に 20クロック、命令キャッシュのミス率
を 2%、データキャッシュのミス率を 4%、LD命令の生起確率を 0.2として、平均 CPI
を計算せよ。
さて、キャッシュを装備したメモリシステムの性能は、ミス率とミス時のペナルティ

で決まることがわかる。これらを小さくする方法を簡単に紹介する。

10.3.1 ミス率を減らす

ミスの原因

キャッシュミスは、以下の 3つの原因で生じる1。

• 容量ミス (Capacity miss): キャッシュ容量の絶対的な不足によって起きる

• インデックス衝突ミス (Conflict miss):インデックス番号の重なるブロックは、ダ
イレクトマップキャッシュや way数が不足すると、キャッシュ全体の容量が不
足していなくても、格納することができない。これに起因するミス。

• 初期化ミス (Cold start miss/Compulsory miss):ジョブのスタート時、あるいは、
切り替え時には、キャッシュ上に有効データが存在しないことにより起きるミ
ス。ちなみに前節の実行例で生じたミスは全てこのタイプのミスである。

13 つの C と呼ぶ

キャッシュの全体容量を増やせば、ミス率が減るのは当然だが、その分のコスト増は
避けられない。したがって、設計時には全体容量は決まってしまっており、その中で
少ないコストでミス率を抑える構成を取ることが要求される。設計の目安は以下の通
りである。

ブロックサイズ

アクセス局所性により、ブロックサイズを大きくするとミス率は下がる傾向にあるが、
大きなブロックを主記憶から持ってくるための転送時間がかかるようになる。また、
キャッシュの容量が小さい時に大きなブロックサイズを選択すると、キャッシュ中に
格納可能なブロック数自体が少なくなってしまい、インデックス衝突ミスが増加して
ミス率が増えてしまう。このため、16byte-64byteの範囲内で、キャッシュサイズが大
きい場合やや大きめに設定するのが有利である。

way数

増やすことにより、インデックス衝突ミスを減らす効果があるが、数が大きくなる程、
その効果は頭打ちになり、4より大きくてもあまり効果がない。ブロックサイズが大
きい場合は、インデックス衝突ミス自体が少ないため、way数の効果がさらに頭打ち
になる。一般的に 1つまりダイレクトマップから 4-wayまでの構成を取る。オンチッ
プキャッシュ等、キャッシュ自体の動作を高速化したい場合はダイレクトマップ方式
が有利な場合もある。

その他のテクニック

ミス率を抑えるためのテクニックとしては以下の方法が用いられる。

• プリフェッチ: アクセスしそうなブロックを先取りしておくことでミス率を減ら
す方法。ハードウェアによる方法とソフトウェアによる方法がある。ハードウェ
アによる方法は、命令のプリフェッチが代表的で、最近のほとんどのプロセッ
サで用いられている。この方法では、命令フェッチの際に、次のいくつかのブ
ロックまで先行してフェッチしてしまう。フェッチされたブロックはキャッシュ
に格納せず、ストリームバッファとよばれる小規模高速の専用メモリに格納し
ておく。命令フェッチ時にキャッシュがミスすると、ストリームバッファをアク
セスし、存在すればこちらから命令を持って来ると共に、キャッシュ上にも転
送する。一方、ソフトウェアによるプリフェッチは、コンパイラまたはプログ
ラマがあらかじめコード中にプリフェッチ命令を埋め込んでおく。

• データ構造の変更: データについてのキャッシュは、データの配置を変えること
によってミス率を低減することができる。同一配列を扱うループ構造をまとめ

てしまったり、配列の添字を入れ替えてなるべく番地順にアクセスしたり、配
列自体をキャッシュに入るようにブロックしたりする方法が用いられる。

10.3.2 ミスペナルティの削減

Early restartと Critical word first

前節に示した実装では、キャッシュミスの際、メインメモリからキャッシュブロック
を転送し、それが終わるまで CPUを待たせておいた。この方法を取ると、ミス時の
ペナルティはブロックのサイズに対応して大きくなってしまう。そこで、全部キャッ
シュに転送するまで待たず、CPUのアクセスしたデータが到着したらそれを CPUに
渡し、CPUは次の命令を開始する方法が用いられる。これを Early restartと呼ぶ。さ
らに、この方法を取る際に、CPUが要求したワードを先に転送し、あとは、サイク
リックに１ブロック分を転送すれば、さらに CPUの待ち時間を短縮することができ
る。この方法を Critical word firstと呼ぶ。ただし、転送を開始するアドレスに制限が
ある DRAMもあり、このような場合は Critical word firstは使えない。これらの方法
によりキャッシュコントローラは複雑になるが、メモリ構成を変更せずに、ミスペナ
ルティを削減することができる。

階層キャッシュ

キャッシュの階層を増やすのも、ミスペナルティの削減を目的としていると考えるこ
とができる。前節で紹介した通り、最近のプロセッサは前回プリントの図に示すよう
にオンチップの１次キャッシュ(Level-1: L1キャッシュと呼ぶ)と、オンボードの２次
キャッシュ(Level-2: L2キャッシュと呼ぶ)の２階層構造が一般的である。しかし、プ
ロセッサの動作周波数が上がるにつれて、チップ内でも階層構造を持たせ、全体とし
て 3階層以上になる場合も増えている。基本的に 1次キャッシュに入るデータは必ず
2次キャッシュにも入っているように追い出し方法等を工夫する。1次キャッシュと 2
次キャッシュではブロックサイズや way数を変える場合もある。
キャッシュについては、性能に与える影響が大きいことから、ここでは紹介されて

いない様々な高速化技法が提案され、用いられている。

10.4 仮想記憶方式

キャッシュで行なった記憶階層に関する手法は、主記憶と 2次記憶間でも行なわれ
る。例えば 32bitマイクロプロセッサでは論理アドレスは 4GB取ることができるが、
実際に搭載するメモリは例えば 16M-64MBなどこれより小さい場合が多い。この場
合、プログラムは論理的なアドレス空間を全て利用できると考えて、論理アドレス空

間 (仮想アドレス空間)で動作させ、一方、実際の記憶の実体は物理アドレス空間上の
メモリにマップする。これを仮想記憶方式と呼ぶ。この際、データ管理の単位はペー
ジと呼び、キャッシュブロックよりも大きく 4KB程度に設定する。
通常、論理アドレスと物理アドレスの変換は CPUからアドレスが出た直後に行わ

れる。これは、複数のプログラムを同時並行的に処理する場合、論理アドレスは重複
する可能性があり、物理アドレスに変換しないでこれを解決するのが難しいためであ
る (シノニム問題と呼ばれる)。プログラムが出力した論理アドレスは、アドレス変換
用のテーブルによって物理アドレスに変換されるのだが、対象のアドレス幅が大きい
ため、このためのテーブルは巨大になってしまう。この変換を一次キャッシュの手前
で高速に行うことが仮想記憶方式の問題である。
このための機構が TLB(Translation look-aside buffer) と呼ばれる小容量のバッファ

で、変換テーブルのキャッシュとして働き、高速にアドレス変換を行なう。アドレス
変換テーブルの本体は、主記憶上に置かれるが、まともに作ると巨大なものとなるた
め、ハッシュ等の技術を用いて、OSの記憶管理プログラムがソフトウェアで管理す
る。TLBによるアドレス変換の様子を図 10.10に示す。この例では論理アドレス空間
を 4GB(32bit)とし、物理アドレス空間は 64MB(24bit)、ページは 4KBとした。この
場合、アドレスの下位 12bitはオフセットとなり、論理／物理アドレス共に共通とな
り、変換の対象は、31bit目から 12bit目までの 20bitの論理ページ番号となる。

Logical address space (4GB)

20bit 12bit

12bit12bit

Physical address (16MB)

TLB

Logical Page M

Phisical Page m

図 10.10: 仮想記憶のアドレス変換

TLBは、この 20bitの論理ページ番号から、12bitの物理ページ番号を生成するテー
ブルである。キャッシュ同様ページ番号を二つに分割して下位アドレスで参照して上
位アドレスで照合するダイレクトマップ方式も考えられるが、高速性と conflict miss

をなくすために、小容量の連想メモリ (Content Addressable Memory)あるいは多入力
のマルチプレクサを用いてフルアソシアティブ方式とする場合もある。
図 10.11に示すように、表の各ビットは有効かどうかを示すビット (V)、プロテク

ション情報、Dirty bitより構成される。

=

=

=

=

=

=

=

=

Logical Address

Logical Address

Match

20bit
Inner Page Address

12bit

Dirty Priority

Phisical Address

Page Address line address
5bit

Index of 
Cache
7bit

Key of 
Cache

Phisical
Address

図 10.11: TLB

論理アドレスが TLB上に存在すれば、物理アドレスとプロテクション情報が読み
出され、存在しなければページフォルト (TLBミス)として例外処理が生じ、主記憶
上のテーブルが OSのメモリ管理プログラムによって参照される。そして、アクセス
されたページが主記憶上に存在しなければ、2次記憶との間にページの入れ替えが行
なわれる。2次記憶にページを書き戻すことをスワップアウト、2次記憶からページ
を読み込むことをスワップインと呼ぶ。スワップアウトするページはキャッシュ同様
LRUで選ばれる。

TLBがヒットした場合は、プロテクション情報をチェックする。ページには、OS
などの特権モードでのみアクセスできるページと、ユーザがアクセス可能なページが
あり、それぞれに Read/Write等の保護を行なっている場合が多い。特権モードのペー
ジに対してアクセスしようとすると、Protection Violationの errorが生じ、これも例外
処理となる。
さらに、始めて当該ページに対して書き込みを行なう場合、主記憶の方のテーブル

を更新する必要があるので、これも例外処理となる。更新後はそのことを示すため、
Dirty bitがセットされる。

TLBは、キャッシュの前段に置かれなければならないため、高速性が必要とされる。
最も容易な実装法は、キャッシュのアドレス変換はページ内のアドレスのみを用い、
TLBによる変換とキャッシュのタグ引きを同時行なう方法である。この方法は小規模
なシステムには有効だが、キャッシュの容量がページサイズ X way数に制限される。

CPUのさらなる高速化技術

本書の目的である「コンピュータアーキテクチャの入門から一通りパイプライン化さ
れたプロセッサの設計ができる」は、今までの章で概ね記述が終わった。以下、最近
のプロセッサに使われている技術をざっと紹介し、今後の学習への繋ぎとしたい。

11.1 基本的な高速化技術

11.1.1 命令の動的スケジューリング

POCOのパイプラインは通常ストールを生じないが、4ステージしかないので動作
周波数をあまり高くすることができない。CPUの性能向上のために、最も直接的な
方法は動作周波数を上げることであり、このため 90年代以降、パイプラインの段数
はどんどん多くなり、10-20ステージに及んだ。長いパイプラインは種々の要因でス
トールするために、これを防ぐ技術が発達した。

アウトオブオーダ実行 データハザードにより、ある命令が結果を待ってストールし
ている時に、その結果を用いない後続の命令が依存関係がなければ、待っている命令
を追い越して先に進んでしまう方法。命令の非順序実行 (out-of order execution)と呼
ぶ。コンパイラでは、解析仕切れない場合でも動的に条件が満足されれば、命令の追
い越しができるため、効率的であるが、追い越す命令の条件、フォワーディングを管
理するハードウェアの構造が複雑になる。効率的な実行をシステマチックに管理する
ために、トーマスローの方法などが提案され、利用されている。

分岐予測 パイプラインが長くなると分岐命令による制御ハザードが深刻な問題と
なった。このため、分岐が成立するかどうかを予測する分岐予測の様々な方式が提案
されている。まず、分岐命令自体を識別するため、実行した分岐命令をキャッシュの

形で CPU内に持つ分岐予測バッファを持たせる。このバッファの上でそれぞれの分
岐命令に対して分岐が行われるかどうか予測を行う。
最も簡単な分岐予測は「以前実行した命令が分岐したから次も分岐するだろう」と

いう 1bit分岐法だが、これはループの構造が少しでも複雑になると役に立たなくな
る。このため、図 11.1に示すように、2bit設けて 4状態とし、予測が２度はずれた
ら、次の予測を変更する 2bit分岐法を導入すると、予測精度はさらに改善される。

Taken

Taken

Not
Taken

Not
Taken

Taken

Not
Taken Taken

Not Taken

Not 
Taken

Taken
Not
Taken

Taken

図 11.1: 2bit分岐予測方式

さらに精度を上げるために、その前に実行された他の分岐命令が分岐したかどうか
で予想を変える方法なども提案されている。また、分岐した場合の分岐先アドレスを
高速に得るために、分岐予測バッファに分岐先アドレスを記憶しておく方法も用いら
れる。最近の高速 CPUでは、いくつかの予測法を選択的に用いるなどさらに高度な
分岐予測が用いているものもある。

投機的実行: 高度な分岐予測を用いれば多くの場合に予測は的中するため、当ると見込
んで既に処理を実行してしまう場合もある。このような方法を投機的実行 (Specurative
Exection)と呼ぶ。投機的実行は、予測がはずれた場合に、実行をキャンセルしなけれ
ばならないため、とりあえず実行した結果を取っておき、その後、予測が当たった場
合にはこれを確定させる機構が必要となる。

11.1.2 命令レベル並列処理

スーパスカラ

長いパイプラインのストールを分岐予測、アウトオブオーダ実行、投機的実行を組み
合わせて抑える方法にも限界がある。これらはすべて、CPIを 1に近づけるための方
法であり、これ以上、性能を向上させるために、CPIを 1以下にすることを考える。
このためには、複数の命令を同時にフェッチし、実行する機構を備える必要がある。

このような機構をスーパスカラ方式と呼ぶ。図 11.2にスーパスカラ方式の基本概念を
示す。メモリ中の命令は、種類毎に分かれてキャッシュから同時にCPUに送られ、同時
にパイプラインに投入されて実行される。当然、キャッシュから CPUには二セットの
バスが必要だが、同一チップ上のオンチップキャッシュならばこのことは十分可能であ
る。この方式は、複数の命令を同時に発行することから命令レベル並列性 (Instruction
Level Parallellism: ILP)を利用した方式と呼ばれる。

Integer

Floating Point
Pipeline

IF

RF

EX

WB

IF

RF

EX

WB

Instructions

Cache

Memory

integer
Instructions

Floating Point
Instructions

図 11.2: スーパスカラ方式（2命令同時発行）

図 11.2では、二命令同時発行のスーパスカラを示す。同時に発行される命令は、な
るべく独立性が高いことが望ましいので、二命令同時発行の場合は、整数パイプライ
ンを用いる整数演算命令と、浮動章演算パイプラインを用いる浮動小数演算命令に分
けるなどする。発行された命令間にはもちろん依存性がある場合が考えられるので、

実行パイプライン間のフォワーディング、待ち合わせ等は、パイプラインの動的実行
と同様にハードウェアで管理する必要がある。三命令以上を同時に発行する場合、(1)
浮動小数演算命令を複数発行する。(2)メモリアクセス命令を独立に発行する。等の
方法がある。最近の高性能マイクロプロセッサの多くは、三命令あるいは四命令同時
発行のスーパスカラ構成を取っている。このうちどの程度の命令が同時に発行可能か
は、実行するプログラムによって異なるが、平均して二を越えるようにするのは困難
である。
スーパスカラ方式には、アウトオブオーダ実行および投機実行を行うものとこれら

を行わないものがある。後者はハードウェアが簡単ですむため、組み込み用の CPU
で用いられる。

VLIW(Very Long Instruction Word)方式

スーパスカラ方式は、複数命令を同時に発行するためのハードウェア制御が複雑で、
全体としてプロセッサの制御部が非常に複雑になる。このため、実行周波数を落とさ
ずに多数の命令を同時発行するのは困難である。そこで、図 11.3に示すように、一
つのフィールドが、普通のコンピュータの命令に相当するような非常に長い命令を設
け、コンパイラにより同時に実行可能な命令を見つけてフィールドを埋めることによっ
て、複数命令同時発行と同様な性能向上を実現する方法をVLIW(Very Long Instruciton
Word)方式と呼ぶ。

Memory

Cache

IF
RF

EX EX EX EX

WB

Long
Instructions

Processor
Pipeline

図 11.3: VLIW方式（4命令分のフィールドを持つ）

VLIW方式は、命令実行の依存関係、データの依存関係の調整を全てコンパイラに
任せているため、制御部はスーパスカラに比べて単純化することができる。このため、

同時実行可能なフィールドは 5以上設けることが可能であるが、長い命令のそれぞ
れのフィールドに対応する演算装置、バス等のデータパスは同時実行を構造化ハザー
ド無しに行えるように、きちんとフィールドに対応して用意する必要がある。また、
フィールドを埋めるために、トレーススケジューリング等、高度なコンパイル技術が
必要となり、既存の命令コードとの互換性が保たれない問題もある。このため、VLIW
方式は、主に信号処理向けプロセッサ、携帯用プロセッサなどで用いられている。

VLIW 方式とスーパスカラ方式の中間的な方式として EPIC(Explicitly Parallel In-
struction Computer)が Intel社の 64bitマイクロプロセッサ Itaniumに用いられた。し
かし、この方法は両者の良い点と共に悪い点も合わせることになり、Itaniumでは期
待されたほど効果を発揮しなかった。このため、VLIW、EPIC方式はやや特殊な方式
として位置づけられている。

11.1.3 より大きなレベルの並列処理

長いパイプラインを用いて周波数を上げ、命令レベル並列性 (Instruction Level Paral-
lellism: ILP)を利用して高速化する手法は 2000年代の始めころまでは有効であった。
しかし、2004年以降、この方式は以下の原因で限界を迎えた。(1)チップの発熱がひ
どくなり、これ以上周波数を上げることが難しくなった。(2)利用できる命令レベル
の並列性が限界に達した。(3)キャッシュを含めたメモリの遅延により、周波数を上げ
てもさほど性能が向上しなくなった。
そこで、PCを複数持つことで、並列に複数のスレッド (Thread)を実行することので

きるプロセッサの開発が盛んになった。ILPにとらわれず、もっと大きな処理の単位で
並列に実行すれば、もっと多くの処理を同時に実行することができ、性能を向上でき
る可能性がある。これがいわゆるマルチコアプロセッサであり、既に IBMの Power5、
Intelの Core2Duoなど１チップ上に 2-4程度のプロセッサを実装する方式が普及して
いる。さらに、もっと多く、16-256のプロセッサを 1チップに搭載したメニーコアと
呼ばれるプロセッサも開発されている。しかし、これらのプロセッサでは、単純に今
までのプログラムを動かしただけでは高速化されず、プログラマが明示的に並列に処
理を記述するか、コンパイラが並列化を行わなければならず、ソフトウェアの開発が
問題点となっている。

